

K
POWER

COMPUTER MONSTERS

STEPHEN MANES AND PAUL SOMERSON

For the Commodore 64, VIC 20; Apple II, II Plus, IIe; IBM PC, PCjr;
Atari 400, 800, XL Series; Radio Shack Color Computer; and TI 99/4A.





COMPUTER MONSTERS

STEPHEN MANES AND PAUL SOMERSON

A HARD/SOFT PRESS BOOK



SCHOLASTIC INC.
New York Toronto London Auckland Sydney Tokyo

No part of this publication may be reproduced in whole or in part, or stored in a retrieval system, or transmitted in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without written permission of the publisher. For information regarding permission write to Scholastic Inc., 730 Broadway, New York, NY 10003.

ISBN 0-590-33177-9

Copyright © 1984 by Hard/Soft Inc. All rights reserved.
Published by Scholastic Inc.

Program adaptations by McMullen & McMullen, Inc.

Designed by Gene Siegel

12 11 10 9 8 7 6 5 4 3 2 1


6

4 5 6 7 8/8

Printed in U.S.A.

14

To Jean, paragon of patience



Welcome, Human!

Did you ever create your own personal monster?

Would you like to match wits with obnoxious ogres and terrible trolls?

Did you ever see vampires and werewolves celebrate Halloween?

Have you ever battled a dragon?

With this book and your home computer you'll be able to do all these things—and a whole lot more!

Just be sure to read our MONSTROUS TIPS AND TRICKS before you begin!

MONSTROUS TIPS AND TRICKS: The programming fiends answer your questions

You may be tempted to skip this section and go right on to the programs. You know what we say to that? **DON'T!!!!**

Running BASIC programs is fun. So is typing them in. But there are times when programming can turn *you* into a monster. In this section, we'll give you lots of tips to help keep you from howling at the moon.

Will these programs run on my computer?

If you have an IBM Personal Computer or IBM PCjr, the answer is YES! All you have to do is type in the Program Listing. You shouldn't need to make any changes at all.

If you have any of the following computers, the answer is still YES!

- Apple II
- Apple II Plus
- Apple IIe
- Atari 400, 800 or XL Series
- Coleco ADAM
- Commodore 64 or VIC-20
- Radio Shack TRS-80 Color Computer
- Texas Instruments 99/4A

With these computers, you'll usually have to change the Program Listing slightly. Most of the time only a line or two will be different.

Just be sure to read the "If You Have . . ." section that appears after every Program Listing. It always tells you *exactly* what changes you'll need to make for your computer. (The ADAM needs special changes. Keep reading to learn how to make them.)

It's a good idea to read the "If You Have . . ." information *before* you start typing in your program. If this is your own copy of *Computer Monsters*, you may want to use a pen or pencil to mark the Program Listing with the changes you'll need.

There's more information about each of these computers at the end of this section. Be sure to read it before starting out!

If you have any other computer, the answer is still YES! But you may have to experiment until you find the exact changes your machine needs to run each program. That should be simple. We've tried hard to use only the parts of BASIC that are available on most machines.

Will my program run as soon as I type it in?

If a program runs as soon as you type it in, you're an absolutely amazing typist — and a very lucky one! Everybody makes little typing mistakes — even us experts! But in a computer program, little mistakes can cause BIG trouble.

If a program doesn't seem to run right, or if you get an error message — something like "Syntax Error" or "SN?" or "You Dummy!" — you'll have to go hunting for your mistakes. Programmers call them bugs. Remember, you put them there, and your program won't run right until you squish out every last one! The official word for that is "debugging."

How do I get the bugs out of my program?

When you type in a program, you must type each line exactly as it appears in the listing. If you miss even one space or one comma, your program probably won't run the way it should.

First, check your program line by line against the listing in the book. It's a good idea to LIST only a few lines at a time, using commands like LIST 100-150. That way, the whole LIST won't scroll up your screen before you get a chance to read it.

You'll probably slap your head when you discover the idiotic goofs you made. With some computers, you'll have to type the whole line in again to make a correction. With others, you can just type in the changes. Learn which way your computer works: It'll save you lots of typing.

Second, be sure you've made the changes your particular machine needs. *Always* look at the "If You Have . . ." section for your particular computer.

If your program still doesn't run, don't give up! There are plenty of things you may have done wrong. How do we know? Because we do them all the time ourselves! Here are some things to watch out for:

Rotten Leftovers

When you start typing in a program, be sure to type NEW and hit the ENTER (or RETURN) key. If you don't, you may find leftover lines from an old program popping up in your new one. It can be a real mess to get them out!

The Curse of the Missing Line Number

It's easy to skip a line by accident when you're typing a long program. Remember: Every program in this book includes a REM statement on line 10 with the name of the program. The next line in every program is line 100, and the lines count up by tens from there, like this:

```
10 REM USELESS PROGRAM
100 INPUT X
110 INPUT Y
120 PRINT Z
```

The only line numbers that won't end in 0 will be lines you add from the "If You Have . . ." section. And the only time the lines will skip — say from 240 to 260 — is when the "If You Have . . ." section tells you to remove a line for your machine.

One Out of Two

Some things have to come in pairs, or your computer will get terribly upset. What kinds of things? Mainly quotation marks (") and left and right parentheses (()). Leaving one out is just about guaranteed to get you some sort of error message. That's why it's important to check a statement like

```
300 PRINT CHR$(INT(RND(1)*(3+2)/(4+3)))
```

very carefully. If you're not positive, count the parentheses and quotation marks on your *other* personal computer — your fingers!

A Letter's Not a Number

Your computer is very, very fussy about letters and numbers. Remember, a one (1) is not a small L (l) or a capital i (I), even though they may look

almost the same. And as far as the computer is concerned, a capital letter O (O) is not the same as the number zero (0).

In our Program Listings, you'll always find a slash through the number zero (0) to keep you from getting confused. But your fingers may do something your brain never had in mind!

Here are some program lines that will cause your computer to scratch its head and give up. We stole them from our own wastebaskets. Each line contains just one error. Can you spot the goofs?

```
100 GOT0 240
200 IF I=2 THEN 500
300 N=J
400 R=22I
500 P=4I6
```

A Space Isn't Nothing!

Especially in graphics programs — programs that draw pictures — spaces are terribly important. If the Program Listing shows a space, be sure you don't leave it out. For example, in

```
100 F$="MURBLEHEAD"
110 PRINT F$;" IS A JERK!"
```

if you leave out the space between the first quotation mark and the word IS in line 110, you'll get this:

```
MURBLEHEADIS A JERK!
```

You can probably guess what prints out if you leave the space in.

But don't add spaces where there are none in the listing. That may cause problems that are even worse!

Semicolons and Commas

Semicolons and commas have special jobs in BASIC programs. Semicolons let you PRINT more than one thing on each line. In PRINT statements, commas do the same thing a little differently, but the programs in this book hardly ever use them that way.

In DATA statements, commas separate each item of data. If you leave one out or put an extra one in, it will confuse your computer. If you get a message like "Out of Data" when everything in your program looks right, chances are you left out a comma — or perhaps a whole DATA statement.

Mysterious Letters

Sometimes the letters and names used for variables in a program can get confusing. It's easy to type an E when you meant to type a D. They're right next to each other on the keyboard. So watch out! One mistake like that can keep an almost perfect program from running at all.

A Real Live Bug?

If you've done everything you can think of, and you're positive we've missed a bug in our own program, please write and describe the problem. We'll try to come up with a solution!

Why did you include the Sample Runs?

The Sample Runs are here for two reasons. First, they give you an idea of what the programs will do. That way, you can decide whether or not you want to type in a particular program.

The Sample Runs are also here to help you make sure your program is running the way it should. But remember: There are some things you can do on a computer that you can't do on a printed page. Many of these programs create pictures that move across your screen. There's no way we can show that here.

And some of the programs do different things every time you run them. So when you run your program, it may not do *exactly* what the Sample Run does — but it should come close.

How come the program listings are in ALL CAPITALS?

If you've already peeked at the Program Listings, you'll notice they're entirely in capital letters. That's because some computers only have capitals. If your machine is one of them, you won't have any problems with any of these programs.

But if you've got a computer that can give you both capitals *and* lower-case letters, you may run into some problems when you RUN these programs. Why? Let's say we ask a question that has a yes or no answer. You type in "yes" — all lowercase. But the program will be looking for "YES" — all capitals.

The solution? It's easy. Just press the key called CAPS LOCK or ALPHA LOCK or just plain LOCK on your computer. Do it when you type in your programs and when you run them. That way, all the letters will appear in CAPITALS!

What's an ENTER key? My computer doesn't have one!

One of the strangest and most important keys on the keyboard is the one you use to INPUT data. The strange part is that computer companies

can't agree on what to call it. Some call it the ENTER key. Some call it the RETURN key. And some put a weird symbol on it and no lettering at all!

It's always in roughly the same place — to the right of the letters on the keyboard. Our Program Listings always call it the ENTER key. If you've got a RETURN key, you'll need to type that word into your PRINT statements wherever we've used the word ENTER.

That brings up another rule we've used in this book. Whenever a program asks you to answer a question or make a choice, you have to hit the ENTER (or RETURN) key before anything will happen. That means you can use the BACKSPACE key to correct any mistakes before you ENTER your answer into the computer.

If you're a good programmer, you'll also notice that we've tried hard to keep you from entering an answer the computer isn't expecting. If we ask for a number between 1 and 4, you won't be able to enter 5 — or 0!

My machine doesn't have a BREAK key. Or does it?

The BREAK key is something else that's different on every machine. Sometimes it's just a key marked BREAK. Sometimes it's really two keys — CTRL and BREAK or some other combination that you have to hold down at the same time. We'll tell you exactly which combination works for your computer in a second. Be sure you know how to use it!

Why? Well, some of our programs will run forever unless you BREAK out of them — or turn off your machine! Sometimes a typing error will make the same thing happen when it's not supposed to. And when you're debugging a program, you may only want to run a small part of it and then quit. It's the same with computers as it is with cars — knowing how to get your machine to stop when you want it to is a really lucky BREAK.

How can I keep my programs from disappearing when I turn off the computer?

Many of the programs in this book will take you only a few minutes to type in. The long ones may take a few hours. And once you turn your computer off, all your hard work will disappear.

But if you have a cassette recorder or disk drive, you can SAVE your programs and RUN them again and again. SAVES are different on almost every machine. The instruction manual that came with your computer or disk drive should tell you how to SAVE your work — and LOAD it back into your computer again.

You don't have to wait until a program is working to SAVE it. Especially with longer programs, you should SAVE now and then as you type the program in. That way, you'll be able to LOAD everything you've done right back into your machine if your dog or cat or little sister accidentally pulls the plug — or if you've had enough of programming for one day and decide to go to the movies.

If you've got a friend with the same kind of computer as yours, you can each type in different programs and SAVE them on disk or cassette. Then you can swap your disks or cassettes. It's another great way to save yourself some typing!

I've got a printer. What about me?

You're really in luck! With a little thought, you'll be able to print out your very own personalized program runs. Do computer-makers agree on the way to get information to your printer? Not any more than they agree on the name of the ENTER (RETURN?) key!

With some machines, it's easy to print out your program runs. All you do is type something like PR#1, and everything you see on the screen will magically turn up on your printer, too!

With other machines it's a little trickier. You may have to change PRINT statements to LPRINT to get your printer to notice them. The best advice we can give is to check the manuals for your computer.

But remember this: If you try to run a program and absolutely *nothing* happens, it's a good bet that you tried to send something to your printer — and your printer wasn't turned on! On most machines, that's an excellent way to stop a program cold!

I can think of a better way to do some of the things you did. Why did you do them your way?

There are usually lots of different ways to get your computer to do the same thing. If you run this program:

```
10 PRINT "JESKLER LOVES PRUNES"  
20 PRINT "JESKLER LOVES PRUNES"
```

you'll get exactly the same result as:

```
10 FOR A=1 TO 2  
20 PRINT "JESKLER LOVES PRUNES"  
30 NEXT A
```

We can think of at least ten other ways to get the same two lines printed out on the screen.

Which way is best? It all depends. If we've done something that looks silly to you, it may be because we've tried to get it to work with machines that can't do some things yours can.

We've tried to write our programs so they'll run on as many machines as possible. If we've left out your favorite BASIC command or done something in

a roundabout way, we apologize. But if we've done something *really* stupid, let us know about it!

Will I mess up my computer if I make changes in your programs?

No! In fact, we *hope* you'll try to improve on these programs — and use them as a starting point for your own. But it's probably a good idea to get them running in their original versions first.

How do I run a program once I've typed it in?

Just type RUN and hit the ENTER or RETURN key!

What else do I have to read in this chapter?

Just the section on your own computer. After that — happy programming!

What if I have an IBM Personal Computer or PCjr?

You're in luck! These programs will run on your machine without any changes at all!

You should be sure to start off in BASIC by typing the command KEY OFF (and hitting the ENTER key). If you have a color monitor or TV set, you should next type WIDTH 40 (and the ENTER key again). You can choose the colors that you see on the screen by giving COLOR commands before you begin — or by writing them into the beginning of the program. The manual that comes with your computer will give you the details.

Remember to set the CAPS LOCK key so that you only get CAPITAL letters. To BREAK out of a program on a PC, hold down the CTRL key and press the BREAK key (it's also called SCROLL LOCK). To BREAK out on a PCjr, press the Fn key and then the letter B (for break). The PC's ENTER key has no lettering on it — just a funny arrow with a bent tail.

What if I have an Apple II Computer?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one RANDOMIZE statement and change the way the program clears the screen. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Be sure you run these programs in Applesoft BASIC — Apple's Integer BASIC just won't work. When you need to BREAK out of a program, hold down the CONTROL key and press the letter C. Your machine has a RETURN key instead of an ENTER key. If you have an Apple IIe, you should set the CAPS LOCK key so that you only get CAPITAL letters.

What if I have an Atari Computer?

Atari BASIC is very unusual. It doesn't allow string arrays like N\$(3). It makes you DIMENSION all your string variables before you can use them. It requires a special way of entering string commands. It lacks features such as the TAB function.

Most of these programs will need simple changes and additions. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has a BREAK key to let you break out of a program. It has a RETURN key instead of an ENTER key. And you can change the colors that appear on your screen—check your manual for the details.

What if I have a Coleco ADAM?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one RANDOMIZE statement and change the way the program clears the screen. Because the version of BASIC in ADAM is almost exactly like the one for the Apple, you'll usually be able to follow the directions for Apple computers in the "If You Have . . ." section after each Program Listing.

But since the ADAM does not display as many characters across the screen as Apples do, you will have to make changes to some programs—especially the ones that make pictures. The following steps will help you get the programs running on your ADAM:

1. Enter the program, making the changes needed for the Apple Computer.
2. Now look at the changes for the TRS-80 Color Computer. *Don't* change any of the lines you already changed for the Apple. *Don't* change any lines that include the words CLS, LINE INPUT, RND or SOUND. Make all other changes needed for the Color Computer.

When you need to BREAK out of a program, hold down the CONTROL key and press the letter C. Your machine has a RETURN key instead of an ENTER key. And you should set the LOCK key so that you only get CAPITAL letters.

What if I have a Commodore 64 or VIC-20?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove a RANDOMIZE statement. In some, you'll also need to change one number in RND statements and change the way the program clears the screen. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has a RETURN key instead of an ENTER key. To BREAK out of a program, press the RUN/STOP key. You may want to change the col-

ors that appear on your screen. Your computer's manual will tell you exactly how to do it.

Since the VIC prints fewer characters (letters and numbers) on the screen than most other home computers, you may want to adjust the PRINT statements so that words don't "wrap" around the ends of lines. The easiest way is to divide one PRINT statement into two. If you do this, be sure the first PRINT statement keeps its original line number.

What if I have a Texas Instruments 99/4A Computer?

TI BASIC is unusual in many ways. It uses special characters instead of AND and OR. It uses the term SEG\$ instead of the more common MID\$. It's very fussy about what you can put after the word THEN.

Most of these programs will need simple changes and additions. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has an ENTER key. To BREAK out of a program, hold down the FCTN key and press the number 4 key. You should set the ALPHA LOCK key so that you only get CAPITAL letters. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

What if I have a TRS-80 Color Computer?

Many of these programs will run on your machine without any changes at all. In some, you'll have to take out one RANDOMIZE statement and change the way RND works. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has an ENTER key and a BREAK key. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

NOW, ON TO THE FUN!

Halloween Night

It's time for all the goblins, ghosts, and spooks to come out of hiding!
Type in this program, and you'll discover some hair-raising things you've
never seen on your computer screen before!

Sample Run

IT'S HALLOWEEN NIGHT--
THE SPOOKS HAVE YOU IN SIGHT!
TURN OUT THE LIGHT
AND GET READY FOR FRIGHT!!!

HIT THE ENTER KEY TO WAKE UP THE
BATS AND SHAKE UP THE GOBLINS?

V V
O

V V
O

V V
O

```
*****
*****
***** O O *****
***** " *****
***** = *****
*****
*****
***** O O *****
***** " *****
***** = *****
*****
*****
** *****
** O O *****
** " *****
** = *****
*****
*****
***** O O *****
***** " *****
***** = *****
*****
*** *****
*** O O *****
*** " *****
*** = *****
*****
```

Break

□ Program Listing

```
10 REM HALLOWEEN NIGHT
100 PRINT CHR$(12)
110 PRINT "IT'S HALLOWEEN NIGHT--"
120 PRINT "THE SPOOKS HAVE YOU IN SIGHT!"
130 PRINT "TURN OUT THE LIGHT"
140 PRINT "AND GET READY FOR FRIGHT!!!"
150 PRINT
160 PRINT "HIT THE ENTER KEY TO WAKE UP THE"
170 PRINT "BATS AND SHAKE UP THE GOBLINS";
180 INPUT D$
190 FOR A=1 TO 20
200 PRINT CHR$(12)
210 FOR B=1 TO A
220 PRINT
230 NEXT B
240 PRINT TAB(5+A);"V V"
250 PRINT TAB(6+A);"O"
260 FOR C=1 TO 100
270 NEXT C
280 PRINT CHR$(12)
290 FOR D=1 TO A
300 PRINT
310 NEXT D
320 PRINT TAB(6+A);"O"
330 PRINT TAB(5+A);"A A"
340 FOR E=1 TO 100
350 NEXT E
360 NEXT A
370 PRINT CHR$(12)
380 R=INT(RND(1)*32)+2
390 GOSUB 750
400 ON RND(1)*6 GOTO 380,380,380,380,380
410 FOR F=1 TO R-1
420 PRINT "*";
430 NEXT F
440 PRINT " ";
450 FOR G=1 TO 34-R
460 PRINT "*";
470 NEXT G
480 PRINT
490 FOR H=1 TO R-1
500 PRINT "*";
510 NEXT H
520 PRINT " O O ";
530 FOR J=1 TO 34-R
540 PRINT "*";
550 NEXT J
560 PRINT
570 FOR K=1 TO R
580 PRINT "*";
590 NEXT K
600 PRINT " ";CHR$(34);" ";
```

```

610 FOR L=1 TO 35-R
620 PRINT "*";
630 NEXT L
640 PRINT
650 FOR M=1 TO R
660 PRINT "*";
670 NEXT M
680 PRINT " = ";
690 FOR N=1 TO 35-R
700 PRINT "*";
710 NEXT N
720 PRINT
730 GOTO 380
740 REM --- STRAIGHT LINE SUBROUTINE ---
750 FOR P=1 TO 38
760 PRINT "*";
770 NEXT P
780 PRINT
790 RETURN

```

If You Have . . .

APPLE II

```

Change: 100 HOME
        160 PRINT "HIT THE RETURN KEY TO WAKE UP THE"
        200 HOME
        280 HOME
        370 HOME

```

ATARI

```

Add: 174 DIM S$(20), D$(1)
     176 S$=" "

```

```

Change: 100 PRINT CHR$(125)
        160 PRINT "HIT THE RETURN KEY TO WAKE UP THE"
        200 PRINT CHR$(125)
        240 PRINT " ";S$(1,A);"V V"
        250 PRINT " ";S$(1,A);"O"
        280 PRINT CHR$(125)
        320 PRINT " ";S$(1,A);"O"
        330 PRINT " ";S$(1,A);"A A"
        370 PRINT CHR$(125)
        380 R=INT(RND(0)*30)+2
        400 ON RND(0)*6 GOTO 380,380,380,380,380
        450 FOR G=1 TO 32-R
        530 FOR J=1 TO 32-R
        610 FOR L=1 TO 33-R
        690 FOR N=1 TO 33-R
        750 FOR P=1 TO 36

```

(continued)

COMMODORE 64

Change: 100 PRINT CHR\$(147)
160 PRINT "HIT THE RETURN KEY TO WAKE UP THE"
200 PRINT CHR\$(147)
280 PRINT CHR\$(147)
370 PRINT CHR\$(147)
380 R=INT(RND(0)*32)+2
400 ON RND(0)*6 GOTO 380,380,380,380,380

COMMODORE VIC-20

Change: 100 PRINT CHR\$(147)
160 PRINT "HIT THE RETURN KEY TO WAKE UP THE"
190 FOR A=1 TO 14
200 PRINT CHR\$(147)
280 PRINT CHR\$(147)
370 PRINT CHR\$(147)
380 R=INT(RND(0)*14)+2
400 ON RND(0)*6 GOTO 380,380,380,380,380
450 FOR G=1 TO 17-R
530 FOR J=1 TO 17-R
610 FOR L=1 TO 18-R
690 FOR N=1 TO 18-R
750 FOR P=1 TO 21

TEXAS INSTRUMENTS 99/4A

Change: 100 CALL CLEAR
200 CALL CLEAR
280 CALL CLEAR
370 CALL CLEAR
380 R=INT(RND*22)+2
400 IF (RND>0.49) * (RND<5.5) THEN 380
450 FOR G=1 TO 24-R
530 FOR J=1 TO 24-R
610 FOR L=1 TO 25-R
690 FOR N=1 TO 25-R
750 FOR P=1 TO 28

TRS-80 COLOR COMPUTER

100 CLS
200 CLS
280 CLS
370 CLS
380 R=INT(RND(24))+2
400 ON RND(6) GOTO 380,380,380,380,380
450 FOR G=1 TO 26-R
530 FOR J=1 TO 26-R
610 FOR L=1 TO 27-R
690 FOR N=1 TO 27-R
750 FOR P=1 TO 31

Revenge of the Bats

How do you feel about bats? How will you feel about a whole screenful of them?

Sample Run

THOSE BATS ARE GETTING AWFULLY BOLD!
NOW THEY'RE STARING AT YOU FROM ALL
OVER THE SCREEN!

HIT THE ENTER KEY TO SEE?

```
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/  
\00/\00/\00/\00/\00/\00/\00/\00/\00/\00/
```

Break

Program Listing

```
10 REM REVENGE OF THE BATS  
100 PRINT CHR$(12)  
110 PRINT "THOSE BATS ARE GETTING AWFULLY BOLD!"  
120 PRINT "NOW THEY'RE STARING AT YOU FROM ALL  
130 PRINT "OVER THE SCREEN!"  
140 PRINT  
150 PRINT "HIT THE ENTER KEY TO SEE";  
160 INPUT X$  
170 PRINT CHR$(12)  
180 PRINT CHR$(92);"00/";  
190 PRINT CHR$(7);  
200 FOR I=1 TO 200  
210 NEXT I  
220 GOTO 180
```

If You Have . . .

APPLE II

Change: 100 HOME
150 PRINT "HIT THE RETURN KEY TO SEE";
170 HOME

ATARI

Add: 91 DIM X\$(1)

Change: 100 PRINT CHR\$(125)
120 PRINT "NOW THEY'RE STARING AT YOU FROM ALL"
150 PRINT "HIT THE RETURN KEY TO SEE";
170 PRINT CHR\$(125)
190 PRINT CHR\$(253)

COMMODORE 64

Change: 100 PRINT CHR\$(147)
150 PRINT "HIT THE RETURN KEY TO SEE";
170 PRINT CHR\$(147)
180 PRINT CHR\$(109);"OO/";

COMMODORE VIC-20

Change: 100 PRINT CHR\$(147)
150 PRINT "HIT THE RETURN KEY TO SEE"
170 PRINT CHR\$(147)
180 PRINT CHR\$(109);"OO/";

TEXAS INSTRUMENTS 99/4A

Change: 100 CALL CLEAR
110 PRINT "THOSE BATS ARE GETTING AWFULLY BOLD!"
170 CALL CLEAR
190 CALL SOUND(100,440,2)

TRS-80 COLOR COMPUTER

Change: 100 CLS
170 CLS
190 SOUND 173,3

Horrible Hydra

The Hydra is a famous monster of Greek myth. When Hercules destroyed her, his charioteer helped. But now the Hydra's back! If you don't have a charioteer, you'll have to slay the Hydra all by yourself.

Sample Run

THE HYDRA IS A CLASSIC MONSTER.
SHE HAS SNAKES INSTEAD OF HAIR.
AND SHE HAS MANY HEADS.
YOU MUST STAB THE RIGHT ONE
TO KILL HER AND ESCAPE HER LAIR.

BUT WATCH OUT!
IF YOU ATTACK AND MISS,
EVERY ONE OF THE HYDRA'S HEADS
SPROUTS A NEW ONE.

HOW MANY TIMES HAVE YOU DEFEATED
THE HYDRA? 10

WHICH HEAD WILL YOU ATTACK--
CHOOSE A NUMBER FROM 1 TO 3? 2

WRONG ONE!
THE HYDRA NOW HAS 4 HEADS.

WHICH HEAD WILL YOU ATTACK--
CHOOSE A NUMBER FROM 1 TO 4? 4

WRONG ONE!
THE HYDRA NOW HAS 6 HEADS.

WHICH HEAD WILL YOU ATTACK--
CHOOSE A NUMBER FROM 1 TO 6? 3
GOOD WORK!
YOU HAVE SLAIN THE HYDRA!

□ Program Listing

```
10 REM HORRIBLE HYDRA
100 PRINT
110 PRINT "THE HYDRA IS A CLASSIC MONSTER."
120 PRINT "SHE HAS SNAKES INSTEAD OF HAIR."
130 PRINT "AND SHE HAS MANY HEADS."
140 PRINT "YOU MUST STAB THE RIGHT ONE"
150 PRINT "TO KILL HER AND ESCAPE HER LAIR."
160 PRINT
170 PRINT "BUT WATCH OUT!"
180 PRINT "IF YOU ATTACK AND MISS,"
190 PRINT "EVERY ONE OF THE HYDRA'S HEADS"
200 PRINT "SPROUTS A NEW ONE."
210 PRINT
220 PRINT "HOW MANY TIMES HAVE YOU DEFEATED"
230 PRINT "THE HYDRA";
240 INPUT T
250 RANDOMIZE T
260 HEADS=3
270 PRINT
280 PRINT "WHICH HEAD WILL YOU ATTACK--"
290 PRINT "CHOOSE A NUMBER FROM 1 TO ";HEADS;
300 INPUT NUM
310 IF NUM<HEADS+1 AND NUM>0 THEN 340
320 PRINT "TRY AGAIN:"
330 GOTO 290
340 R=1+INT(RND(1)*HEADS)
350 IF NUM=R THEN 440
360 IF HEADS>60000000 THEN 470
370 HEADS=(HEADS-1)*2
380 PRINT
390 PRINT "WRONG ONE!"
400 PRINT "THE HYDRA NOW HAS ";HEADS;" HEADS."
410 PRINT
420 GOTO 280
430 PRINT
440 PRINT "GOOD WORK!"
450 PRINT "YOU HAVE SLAIN THE HYDRA!"
460 END
470 PRINT "TOO BAD!"
480 PRINT "YOU ARE DOOMED TO REMAIN"
490 PRINT "IN THE HYDRA'S LAIR FOREVER!"
```


If You Have . . .

APPLE II

Delete: 250

ATARI

Delete: 250

Change: 340 R=1+INT(RND(0)*HEADS)

COMMODORE 64

Delete: 250

Change: 340 R=1+INT(RND(0)*HEADS)

COMMODORE VIC-20

Delete: 250

Change: 230 PRINT "THE HYDRA"
290 PRINT "CHOOSE A NUMBER FROM 1 TO ";HEADS
340 R=1+INT(RND(0)*HEADS)

TEXAS INSTRUMENTS 99/4A

Add: 115 PRINT "MONSTER."
155 PRINT "LAIR."
195 PRINT "HEADS"

Change: 110 PRINT "THE HYDRA IS A CLASSIC"
150 PRINT "TO KILL HER AND ESCAPE HER"
190 PRINT "EVERY ONE OF THE HYDRA'S"
310 IF (NUM<HEADS+1) * (NUM>0) THEN 340
340 R=1+INT(RND*HEADS)

TRS-80 COLOR COMPUTER

Delete: 250

Change: 340 R=RND(HEADS)

Creepy Candle

Your flashlight batteries just ran out. Only one thing can protect you from the creepy monsters in this dim, dark cave — this special slow-burning candle!

Sample Run

WHEN YOU'RE IN A CREEPY CAVE
YOU NEED A CANDLE TO LIGHT
YOUR WAY. HIT THE ENTER KEY
FOR YOUR CANDLE--AND PRAY THAT
IT DOESN'T BURN OUT?

```
)  
)  
( )  
( ( )  
( )  
I  
I  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM
```

[There's more—when you get to the end of the program, you'll see something like this:]

```
(  
( )  
( ( )  
( ) ( )  
( ( )  
I  
I  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM  
MMMMM
```

(
()
(()
() ()
(()
I
I
MMMMM
MMMMM
MMMMM
MMMMM
MMMMM

)
))
())
(())
())
I
I
MMMMM
MMMMM
MMMMM
MMMMM

(
()
(()
() ()
(()
I
I
MMMMM
MMMMM
MMMMM

(
()
(()
() ()
(()
I
I
MMMMM
MMMMM

)
))
())
(())
())
I
I
MMMMM

(continued)


```

320 FOR A=1 TO 2
330 PRINT TAB(12);"I"
340 NEXT A
350 FOR B=1 TO 15-K
360 PRINT TAB(10);
370 FOR C=1 TO 5
380 PRINT "M";
390 NEXT C
400 PRINT
410 NEXT B
420 K=K+1
430 FOR D=1 TO K*33
440 NEXT D
450 IF K=15 THEN 470
460 GOTO 160
470 PRINT CHR$(12)
480 GOSUB 580
490 FOR Y=1 TO 25
500 ON INT(RND(1)*2)+1 GOTO 510,530
510 PRINT TAB(10);"("
520 GOTO 540
530 PRINT TAB(10);")"
540 NEXT Y
550 GOSUB 580
560 END
570 REM --- PRINT CLEAR SUBROUTINE ---
580 FOR X=1 TO 25
590 PRINT
600 NEXT X
610 RETURN

```

If You Have . . .

APPLE II

Change: 120 PRINT "YOUR WAY. HIT THE RETURN KEY"
160 HOME
470 HOME

ATARI

Add: 145 DIM X\$(1)

Change: 120 PRINT "YOUR WAY. HIT THE RETURN KEY"
160 PRINT CHR\$(125)
200 ON INT(RND(0)*2) + 1 GOTO 210,270
210 PRINT " ", "("
220 PRINT " ", "()"
230 PRINT " ", "()"
240 PRINT " " ; "(" ; "()" ; "()"
250 PRINT " " ; "(" ; "()" ; "()"

(continued)

```

270 PRINT " ",")"
280 PRINT " ",")"
290 PRINT " ";"( )"
300 PRINT " ";"( ( )"
310 PRINT " ";"( )"
330 PRINT " ", "I"
360 PRINT ,;
470 PRINT CHR$(125)
500 ON INT(RND(0)*2)+1 GOTO 510,530
510 PRINT ,)"
530 PRINT ,( "

```

COMMODORE 64

Change: 120 PRINT "YOUR WAY. HIT THE RETURN KEY"
160 PRINT CHR\$(147)
470 PRINT CHR\$(147)

COMMODORE VIC-20

Change: 120 PRINT "YOUR WAY. HIT THE RETURN KEY"
160 PRINT CHR\$(147)
470 PRINT CHR\$(147)

TEXAS INSTRUMENTS 99/4A

Change: 130 PRINT "FOR YOUR CANDLE--AND PRAY"
140 PRINT "THAT IT DOESN'T BURN OUT";
160 CALL CLEAR
200 ON INT(RND*2)+1 GOTO 210,270
470 CALL CLEAR
500 ON INT(RND*2)+1 GOTO 510,530
510 PRINT TAB(12);"("
530 PRINT TAB(12);")"

TRS-80 COLOR COMPUTER

Change: 160 CLS
200 ON INT(RND(2)) + 1 GOTO 210,270
350 FOR B=1 TO 9-K
450 IF K=9 THEN 470
470 CLS
500 ON INT(RND(2)) + 1 GOTO 510,530
510 PRINT TAB(12);"("
530 PRINT TAB(12);")"

Big Drac, Shake, and Flies

Take your human to Dracmonster's Restaurant for lunch. Just be sure
to bring along a monstrous appetite!

Sample Run

WELCOME TO DRACMONSTER'S!

GLAD TO SEE YOU BROUGHT
YOUR HUMAN ALONG.

HOW MANY POUNDS DOES YOUR HUMAN WEIGH? 95

HOW MUCH DO YOU WEIGH, MONSTER? 500

COME ON! I CAN SEE
THAT YOU WEIGH
MORE THAN TEN TIMES
AS MUCH AS THAT HUMAN!
HOW MUCH DO YOU WEIGH, MONSTER? 1000

HOW MANY BIG DRACS FOR YOUR HUMAN--
THEY'RE \$1.75 A PIECE? 2

HOW MANY ORDERS OF FLIES FOR YOUR HUMAN--
THEY'RE 69 CENTS EACH? 7
RIDICULOUS! YOUR HUMAN'LL GET SICK!
I'LL GIVE YOU 1.

AND HOW MANY BLOOD SHAKES--
THEY'RE 87 CENTS? 3

AND YOUR USUAL MEAL, OF COURSE--
20 BIG DRACS, 10 ORDERS OF FLIES
AND 30 SHAKES.

FOR THE TWO OF YOU, THAT'LL BE
\$38.50 FOR THE BIG DRACS,
\$7.59 FOR THE FLIES AND,
\$28.71 FOR THE SHAKES.

A GRAND TOTAL OF
\$74.80
ENJOY YOUR MEAL!

□ Program Listing

```
10 REM BIG DRAC, SHAKE AND FLIES
100 PRINT
110 PRINT "WELCOME TO DRACMONSTER'S!"
120 PRINT
130 PRINT "GLAD TO SEE YOU BROUGHT"
140 PRINT "YOUR HUMAN ALONG."
150 PRINT
160 PRINT "HOW MANY POUNDS DOES YOUR HUMAN WEIGH";
170 INPUT HW
180 PRINT
190 IF HW<1 THEN 160
200 PRINT "HOW MUCH DO YOU WEIGH, MONSTER";
210 INPUT MW
220 PRINT
230 IF MW>HW*10 THEN 290
240 PRINT "COME ON! I CAN SEE"
250 PRINT "THAT YOU WEIGH"
260 PRINT "MORE THAN TEN TIMES"
270 PRINT "AS MUCH AS THAT HUMAN!"
280 GOTO 200
290 C=INT(MW/HW)
300 PRINT "HOW MANY BIG DRACS FOR YOUR HUMAN--"
310 PRINT "THEY'RE $1.75 APIECE";
320 L=1+INT(RND(1)*5)
330 INPUT BD
340 IF BD<1 THEN 300
350 IF BD>L THEN GOSUB 750
360 IF BD>L THEN BD=L
370 GOSUB 780
380 PRINT
390 PRINT "HOW MANY ORDERS OF FLIES FOR YOUR HUMAN--"
400 PRINT "THEY'RE 69 CENTS EACH";
410 INPUT FR
420 IF FR<1 THEN 390
430 IF FR>L THEN GOSUB 750
440 IF FR>L THEN FR=L
450 GOSUB 780
460 PRINT
470 PRINT "AND HOW MANY BLOOD SHAKES--"
480 PRINT "THEY'RE 87 CENTS";
490 INPUT SH
500 IF SH<1 THEN 470
510 IF SH>L THEN GOSUB 750
520 IF SH>L THEN SH=L
530 PRINT
540 PRINT "AND YOUR USUAL MEAL, OF COURSE--"
550 PRINT BD*C; " BIG DRACS,";
560 PRINT FR*C;" ORDERS OF FLIES"
570 PRINT "AND ";SH*C; " SHAKES."
580 PRINT
590 PRINT "FOR THE TWO OF YOU, THAT'LL BE "
```

```

600 FD=INT((BD*(C+1)*175)+.5)/100
610 FF=INT((FR*(C+1)*69)+.5)/100
620 FS=INT((SH*(C+1)*87)+.5)/100
630 T=FD+FF+FS
640 PRINT "$";FD;
650 PRINT " FOR THE BIG DRACS,"
660 PRINT "$";FF;
670 PRINT " FOR THE FLIES AND,"
680 PRINT "$";FS;
690 PRINT " FOR THE SHAKES."
700 PRINT
710 PRINT "A GRAND TOTAL OF"
720 PRINT "      $";T
730 PRINT "ENJOY YOUR MEAL!"
740 END
750 PRINT "RIDICULOUS! YOUR HUMAN'LL GET SICK!"
760 PRINT "I'LL GIVE YOU ";L;". "
770 RETURN
780 L=1+INT(RND(1)*10)
790 RETURN

```

If You Have . . .

APPLE II No Changes Required

ATARI

Change: 320 L=1+INT(RND(0)*5)
780 L=1+INT(RND(0)*10)

COMMODORE 64

Change: 320 L=1+INT(RND(0)*5)
390 PRINT "HOW MANY ORDERS OF FLIES FOR YOUR HUMAN"

COMMODORE VIC-20

Change: 160 PRINT "HOW MANY POUNDS DOES YOUR HUMAN WEIGH"
200 PRINT "HOW MUCH DO YOU WEIGH, MONSTER"
310 PRINT "THEY'RE \$1.75 APIECE"
320 L=1+INT(RND(0)*5)
390 PRINT "HOW MANY ORDERS OF FLIES FOR YOUR HUMAN"
400 PRINT "THEY'RE 69 CENTS EACH"
480 PRINT "THEY'RE 87 CENTS"

TEXAS INSTRUMENTS 99/4A

Add: 165 PRINT "HUMAN WEIGH";
205 PRINT "MONSTER";

(continued)

```

355 GOSUB 750
435 GOSUB 750
515 GOSUB 750
545 PRINT "OF COURSE--"
605 FD=INT(A)/100
615 FF=INT(B)/100
625 FS=INT(C)/100
755 PRINT "GET SICK!"

```

```

Change: 160 PRINT "HOW MANY POUNDS DOES YOUR"
200 PRINT "HOW MUCH DO YOU WEIGH,"
320 L=INT(5*RND)+1
350 IF BD<=L THEN 370
360 BD=L
430 IF FR<=L THEN 450
440 FR=L
510 IF SH<=L THEN 530
520 SH=L
540 PRINT "AND YOUR USUAL MEAL,"
550 PRINT BD*C; " BIG DRACS,"
600 A=(BD*(C+1)*175)+.5
610 B=(FR*(C+1)*69)+.5
620 C=(SH*(C+1)*87)+.5
750 PRINT "RIDICULOUS! YOUR HUMAN'LL"
780 L=INT(10*RND)+1

```

TRS-80 COLOR COMPUTER

```

Change: 320 L=RND(5)
780 L=RND(10)

```


□ Program Listing

```
10 REM MONSTER CONSTRUCTION COMPANY
100 PRINT "NEED A BRICK WALL TO KEEP OUT"
110 PRINT "HUMANS? MONSTER CONSTRUCTION"
120 PRINT "COMPANY AT YOUR SERVICE!"
130 PRINT
140 PRINT "HIT THE ENTER KEY WHEN YOU'D"
150 PRINT "LIKE US TO START BUILDING";
160 INPUT X$
170 PRINT
180 B$="*****"
190 L$="****"
200 GOSUB 270
210 GOSUB 340
220 GOSUB 270
230 GOSUB 370
240 GOSUB 470
250 GOSUB 370
260 GOTO 200
270 FOR A=1 TO 4
280 IF A<4 THEN 310
290 PRINT B$
300 GOTO 320
310 PRINT B$;" ";
320 NEXT A
330 RETURN
340 PRINT "*" ;TAB(9);"* *";TAB(19);"* *";
350 PRINT TAB(29);"* *";TAB(39);"*"
360 RETURN
370 FOR A=1 TO 5
380 IF A<>5 THEN 410
390 PRINT L$
400 GOTO 450
410 IF A>1 THEN 440
420 PRINT L$;" ";
430 GOTO 450
440 PRINT B$;" ";
450 NEXT A
460 RETURN
470 PRINT TAB(4);"* *";TAB(14);"* *";
480 PRINT TAB(24);"* *";TAB(34);"* *"
490 RETURN
```

□ If You Have . . .

APPLE II

Change: 140 PRINT "HIT THE RETURN KEY WHEN YOU'D"

ATARI

Add: 155 DIM X\$(1),B\$(11),L\$(6)
395 PRINT "*"

Change: 140 PRINT "HIT THE RETURN KEY WHEN YOU'D"
180 B\$="*****"
190 L\$="***"
340 PRINT " ";"* *";";" * *"
350 PRINT " ", "* *";",,,," *"
390 PRINT L\$;
470 PRINT " * *", " * *"
480 PRINT " ", "* *";",,,," * *"

COMMODORE 64

Change: 140 PRINT "HIT THE RETURN KEY WHEN YOU'D"
340 PRINT "*" ; TAB(8) ; "* *"; TAB(18) ; "* *";
350 PRINT TAB(28) ; "* *"; TAB(38) ; "*"
470 PRINT TAB(3) ; "* *"; TAB(13) ; "* *";
480 PRINT TAB(23) ; "* *"; TAB(33) ; "* *"

COMMODORE VIC-20

Add: 225 PRINT
255 PRINT

Change: 140 PRINT "HIT THE RETURN KEY WHEN YOU'D"
180 B\$="*****"
190 L\$="***"
290 PRINT B\$; "*"
340 PRINT "*" ; TAB(3) ; "* *"; TAB(8) ; "* *";
350 PRINT TAB(13) ; "* *"; TAB(19) ; "*"
470 PRINT TAB(1) ; "* *"; TAB(6) ; "* *";
480 PRINT TAB(11) ; "* *"; TAB(16) ; "* *"

TEXAS INSTRUMENTS 99/4A

Delete: 480

Add: 225 PRINT
255 PRINT

Change: 180 B\$="*****"

(continued)

```
190 L$="***"  
340 PRINT "*";TAB(6);"* *";TAB(13);"* *";  
350 PRINT TAB(20);"* *";TAB(27);"*"  
470 PRINT TAB(3);"* *";TAB(10);"* *";TAB(17);  
    "* *";TAB(24);"* *"
```

TRS-80 COLOR COMPUTER

Delete: 480

```
Change: 180 B$="*****"  
190 L$="***"  
340 PRINT "*";TAB(6);"* *";TAB(14);"* *";  
350 PRINT TAB(22);"* *";TAB(30);"*"  
470 PRINT TAB(1);"* *";TAB(9);"* *";TAB(17);  
    "* *";TAB(25);"* *"
```


Blood Bank

Count Dracula has an account at the First National Blood Bank — but he likes to keep some refreshments handy around the house. Help him keep track of his private cellar.

Sample Run

DRACULA NEEDS BLOOD--
LOTS OF IT!
HE CAN BUY IT IN PINT
BOTTLES FROM A BLOOD BANK.
BUT HE LIKES TO KEEP
IT IN BIGGER BOTTLES.

PLEASE HELP THE POOR VAMPIRE.
HE'S NOT VERY GOOD AT MATH.

DRACULA HAS 152 PINTS OF BLOOD.
HOW MANY GALLON BOTTLES DOES HE NEED? 34

WRONG!
TRY AGAIN:
DRACULA HAS 152 PINTS OF BLOOD.
HOW MANY GALLON BOTTLES DOES HE NEED? 19

RIGHT! NOW TRY THIS ONE:

DRACULA HAS 688 PINTS OF BLOOD.
HOW MANY GALLON BOTTLES DOES HE NEED? 86

RIGHT! NOW TRY THIS ONE:

DRACULA HAS 672 PINTS OF BLOOD.
HOW MANY GALLON BOTTLES DOES HE NEED? 84

RIGHT THREE TIMES!
DRACULA THANKS YOU--
AND SPARES YOU (THIS TIME).

□ Program Listing

```
10 REM BLOOD BANK
100 PRINT
110 PRINT "DRACULA NEEDS BLOOD--"
120 PRINT "LOTS OF IT!"
130 PRINT "HE CAN BUY IT IN PINT"
140 PRINT "BOTTLES FROM A BLOOD BANK."
150 PRINT "BUT HE LIKES TO KEEP"
160 PRINT "IT IN BIGGER BOTTLES."
170 PRINT
180 PRINT "PLEASE HELP THE POOR VAMPIRE."
190 PRINT "HE'S NOT VERY GOOD AT MATH."
200 PRINT
210 S=1+INT(RND(1)*2)
220 IF S=1 THEN C=2
230 IF S=1 THEN Q$="QUART"
240 IF S=2 THEN C=8
250 IF S=2 THEN Q$="GALLON"
260 A=8*(1+INT(RND(1)*100))
270 PRINT "DRACULA HAS ";A;" PINTS OF BLOOD."
280 PRINT "HOW MANY ";Q$;" BOTTLES DOES HE NEED";
290 INPUT ANSWER
300 IF ANSWER=0 THEN 280
310 PRINT
320 IF ANSWER<>A/C THEN 430
330 R=R+1
340 IF R<>3 THEN 390
350 PRINT "RIGHT THREE TIMES!"
360 PRINT "DRACULA THANKS YOU--"
370 PRINT "AND SPARES YOU (THIS TIME)."
```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 205 DIM Q\$(8)
Change: 210 S=1+INT(RND(0)*2)
260 A=8*(1+INT(RND(0)*100))

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 280 PRINT "HOW MANY ";Q\$;" BOTTLES DOES HE NEED"

TEXAS INSTRUMENTS 99/4A

Delete: 230, 250
Add: 225 C=2
235 Q\$="QUART"
245 C=8
255 Q\$="GALLON"
465 PRINT "THAT'S THE SECOND TIME YOU GOOFED."
Change: 210 S=INT(RND*2)+1
220 IF S=1 THEN 225 ELSE 240
240 IF S=2 THEN 245 ELSE 260
260 A=8*(INT(RND*100))
460 IF F=2 THEN 465 ELSE 480
470 PRINT "VAMPIRES ARE VERY IMPATIENT!"

TRS-80 COLOR COMPUTER

Add: 285 PRINT "DOES HE NEED";
465 IF F=2 THEN PRINT "GOOFED."
Change: 210 S=RND(2)
260 A=8*RND(100)
280 PRINT "HOW MANY "; Q\$;" BOTTLES"
460 IF F=2 THEN PRINT "THAT'S THE SECOND TIME YOU"
520 PRINT "DRACULA WILL JUST HAVE TO MAKE"
530 PRINT "DO WITH YOUR BLOOD INSTEAD!"

Obnoxious Ogre

Some ogres are extremely friendly. This one isn't. In fact, you have to be very careful what you say to him (or her — we're not quite sure). Take care — this sensitive ogre has been known to get very nasty indeed!

Sample Run

I AM THE OGRE!
I AM VERY TOUCHY.
INSULT ME AT YOUR PERIL!

WHAT DO YOU SAY TO THAT? YOU'RE A WIMP!
TAKES ONE TO KNOW ONE!

WHAT DO YOU SAY TO THAT? YOU'RE A JERK!
YOU'RE A BIGGER ONE!

WHAT DO YOU SAY TO THAT? DROP DEAD!
HOW DARE YOU SAY THAT TO ME?

WHAT DO YOU SAY TO THAT? YOU STINK!
ME? YOU DO, ONLY TWICE AS MUCH!

WHAT DO YOU SAY TO THAT? GO JUMP IN A LAKE!
I DARE YOU TO SAY THAT AGAIN!

WHAT DO YOU SAY TO THAT? GO JUMP IN A LAKE!

MUNCH.

CRUNCH.

BURP.

YOU TASTE AWFUL!
BUT AT LEAST I DON'T HAVE TO LISTEN
TO YOUR INSULTS ANY MORE.
DON'T SAY I DIDN'T WARN YOU!

□ Program Listing

```
100 REM OBNOXIOUS OGRE
110 PRINT "I AM THE OGRE!"
120 PRINT "I AM VERY TOUCHY."
130 PRINT "INSULT ME AT YOUR PERIL!"
140 PRINT
150 PRINT "WHAT DO YOU SAY TO THAT?"
160 INPUT I$
170 IF I$="" THEN 150
180 C=1+INT(RND(1)*5)
190 IF MID$(I$,1,3)<>"YOU" THEN 370
200 IF MID$(I$,5,2)="RE" OR MID$(I$,6,2)="RE" THEN 230
210 IF MID$(I$,4,2)="R " THEN 450
220 GOTO 300
230 IF C=1 THEN PRINT "YOU'RE A BIGGER ONE!"
240 IF C=2 THEN PRINT "YOU'RE ONE, TOO!"
250 IF C=3 THEN PRINT "HA! YOU'RE THE BIGGEST ONE I'VE
EVER SEEN!"
260 IF C=4 THEN PRINT "ME? YOU SHOULD TALK!"
270 IF C>4 THEN PRINT "TAKES ONE TO KNOW ONE!"
280 PRINT
290 GOTO 150
300 IF C=1 THEN PRINT "ME? YOU DO, ONLY TWICE AS MUCH!"
310 IF C=2 THEN PRINT "SO DO YOU!"
320 IF C=3 THEN PRINT "YOU TAUGHT ME HOW!"
330 IF C=4 THEN PRINT "YOU DO THAT EVERY DAY AND TWICE
ON SUNDAYS!"
340 IF C>4 THEN PRINT "WELL, YOU STINK!"
350 PRINT
360 GOTO 150
370 IF I$=R$ THEN 480
380 IF C=1 OR C=2 THEN PRINT "I DARE YOU TO SAY THAT AGAIN!"
390 IF C=1 OR C=2 THEN R$=I$
400 IF C=3 THEN PRINT "HOW DARE YOU SAY THAT TO ME?"
410 IF C=4 THEN PRINT "WANT TO FIGHT ABOUT IT?"
420 IF C>4 THEN PRINT "BETTER WATCH YOUR LANGUAGE, BUSTER!"
430 PRINT
440 GOTO 150
450 IF C<3 THEN PRINT "SO DOES YOURS!"
460 IF C>2 THEN PRINT "YOURS IS EVEN WORSE!"
470 GOTO 150
480 PRINT
490 PRINT
500 PRINT "MUNCH."
510 PRINT "          CRUNCH."
520 PRINT "          BURP."
530 PRINT
540 PRINT "YOU TASTE AWFUL!"
550 PRINT "BUT AT LEAST I DON'T HAVE TO LISTEN"
560 PRINT "TO YOUR INSULTS ANY MORE."
570 PRINT "DON'T SAY I DIDN'T WARN YOU!"
```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 145 DIM I\$(25),R\$(25)

Change: 180 C=1+INT(RND(0)*5)
190 IF I\$(1,3)<>"YOU" THEN 370
200 IF I\$(5,6)="RE" OR IF I\$(6,7)="RE" THEN 230
210 IF I\$(4,5)="R " THEN 450

COMMODORE 64 No Changes Required

COMMODORE VIC-20 No Changes Required

TEXAS INSTRUMENTS 99/4A

Add: 205 IF SEG\$(I\$,6,2)="RE" THEN 230
235 PRINT "YOU'RE A BIGGER ONE!"
245 PRINT "YOU'RE ONE, TOO!"
255 PRINT "HA! YOU'RE THE BIGGEST ONE I'VE EVER
SEEN!"
265 PRINT "ME? YOU SHOULD TALK!"
275 PRINT "TAKES ONE TO KNOW ONE!"
305 PRINT "ME? YOU DO, ONLY TWICE AS MUCH!"
315 PRINT "SO DO YOU!"
325 PRINT "YOU TAUGHT ME HOW!"
335 PRINT "YOU DO THAT EVERY DAY AND TWICE ON
SUNDAYS!"
345 PRINT "WELL, YOU STINK!"
385 PRINT "I DARE YOU TO SAY THAT AGAIN!"
405 PRINT "HOW DARE YOU SAY THAT TO ME?"
415 PRINT "WANT TO FIGHT ABOUT IT?"
425 PRINT "BETTER WATCH YOUR LANGUAGE, BUSTER!"
455 PRINT "SO DOES YOURS!"
465 PRINT "YOURS IS EVEN WORSE!"

Change: 180 C=1+INT(RND*5)
190 IF SEG\$(I\$,1,3)<>"YOU" THEN 370
200 IF SEG\$(I\$,5,2)="RE" THEN 230
210 IF SEG\$(I\$,4,2)="R " THEN 450
230 IF C<>1 THEN 240
240 IF C<>2 THEN 250
250 IF C<>3 THEN 260
260 IF C<>4 THEN 270
270 IF C>4 THEN 275 ELSE 280
300 IF C<>1 THEN 310
310 IF C<>2 THEN 320

```
320 IF C<>3 THEN 330
330 IF C<>4 THEN 340
340 IF C>4 THEN 345 ELSE 350
380 IF (C=1)+(C=2) THEN 385 ELSE 400
390 R$=I$
400 IF C<>3 THEN 410
410 IF C<>4 THEN 420
420 IF C>4 THEN 425 ELSE 430
450 IF C>=3 THEN 460
460 IF C<=2 THEN 470
```

TRS-80 COLOR COMPUTER

Change: 180 C=RND(5)
550 PRINT "BUT AT LEAST I DON'T HAVE TO"
560 PRINT "LISTEN TO YOUR INSULTS ANYMORE."

Monster Snapshots

When you take your monster around the world with you, you'll both need passports. Run this program, and you'll get your monster's official photo delivered instantly. Say! You and your monster look so much alike, you could pass for twins!

Sample Run

WELCOME TO THE CREEPIX STUDIO!
IT'S TIME TO TAKE YOUR MONSTER'S
OFFICIAL PHOTO.

HOW MANY EYES DOES YOUR
MONSTER HAVE: 2, 4, 6, OR 8 ? 8

HOW MANY NOSES DOES YOUR
MONSTER HAVE: 1 OR 2 ? 2

IS YOUR MONSTER HAPPY OR SAD ? HAPPY

```

I I I I I I I I I I I I I I
=====
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
I          <(O)>      <(O)>      I
XXXXXI          -M-          IXXXXX
X   I          MMM          I   X
XXXXXI          (M-M)        IXXXXX
I
XXXXXI          -M-          IXXXXX
X   I          MMM          I   X
XXXXXI          (M-M)        IXXXXX
I
I          M          M          I
I          MMMMMM          I
I
=====

```

WANT ANOTHER PICTURE ? YES

HOW MANY EYES DOES YOUR
MONSTER HAVE: 2, 4, 6, OR 8 ? 4

HOW MANY NOSES DOES YOUR
MONSTER HAVE: 1 OR 2 ? 1

IS YOUR MONSTER HAPPY OR SAD ? SAD

```

      I I I I I I I I I I I I I I I
      =====
      I                                I
      I      <(O)>          <(O)>      I
      I                                I
      I      <(O)>          <(O)>      I
XXXXXXI                                IXXXXXX
X  I      -M-                                I  X
XXXXXXI      MMM                                IXXXXXX
      (M-M)
      I                                I
      I      MMMMMMMM                          I
      I      M          M                          I
      I                                I
      =====

```

WANT ANOTHER PICTURE ? NO

Program Listing

```

10 REM MONSTER SNAPSHOTS
100 PRINT CHR$(12)
110 PRINT "WELCOME TO THE CREEPIX STUDIO!"
120 PRINT "IT'S TIME TO TAKE YOUR MONSTER'S"
130 PRINT "OFFICIAL PHOTO."
140 PRINT
150 PRINT "HOW MANY EYES DOES YOUR"
160 PRINT "MONSTER HAVE: 2, 4, 6, OR 8";
170 INPUT E
180 E=E/2
190 IF E<>1 AND E<>2 AND E<>3 AND E<>4 THEN 140
200 PRINT
210 PRINT "HOW MANY NOSES DOES YOUR"
220 PRINT "MONSTER HAVE: 1 OR 2";
230 INPUT N
240 IF N<1 OR N>2 THEN 200
250 PRINT
260 PRINT "IS YOUR MONSTER HAPPY OR SAD";
270 INPUT H$
280 T$=MID$(H$,1,1)
290 IF T$<>"H" AND T$<>"S" THEN 250
300 PRINT CHR$(12)
310 PRINT TAB(7);

```

(continued)

```

320 FOR A=1 TO 27 STEP 2
330 PRINT "I";CHR$(32);
340 NEXT A
350 GOSUB 540
360 FOR J=1 TO E
370 GOSUB 610
380 NEXT J
390 FOR K=1 TO N
400 GOSUB 650
410 NEXT K
420 IF T$="H" THEN 450
430 GOSUB 750
440 GOTO 460
450 GOSUB 710
460 PRINT TAB(7);"I";TAB(33);"I";
470 GOSUB 540
480 PRINT
490 PRINT
500 PRINT TAB(9);"WANT ANOTHER PICTURE";
510 INPUT Y$
520 IF MID$(Y$,1,1)<>"N" THEN 100
530 END
540 REM --- LINE SUBROUTINE ---
550 PRINT TAB(7);
560 FOR B=1 TO 27
570 PRINT "=";
580 NEXT B
590 RETURN
600 REM --- EYES SUBROUTINE ---
610 PRINT TAB(7);"I";TAB(33);"I"
620 PRINT TAB(7);"I";TAB(13);"<(O)>";TAB(23);"<(O)>";
TAB(33);"I"
630 RETURN
640 REM --- NOSE SUBROUTINE ---
650 PRINT TAB(2);"XXXXX";TAB(7);"I";TAB(18);" -M- ";
TAB(33);"I";"XXXXX"
660 PRINT TAB(2);"X";TAB(7);"I";TAB(18);" MMM ";TAB(33);
"I";TAB(38);"X"
670 PRINT TAB(2);"XXXXX";TAB(7);"I";TAB(18);"(M-M)";
TAB(33);"I";"XXXXX"
680 PRINT TAB(7);"I";TAB(33);"I"
690 RETURN
700 REM --- HAPPY MOUTH SUBROUTINE ---
710 PRINT TAB(7);"I";TAB(16);"M M";TAB(33);"I"
720 PRINT TAB(7);"I";TAB(16);" MMMMMM ";TAB(33);"I"
730 RETURN
740 REM --- SAD MOUTH SUBROUTINE ---
750 PRINT TAB(7);"I";TAB(16);" MMMMMM ";TAB(33);"I"
760 PRINT TAB(7);"I";TAB(16);"M M";TAB(33);"I"
770 RETURN

```

If You Have . . .

APPLE II

Add: 345 PRINT
585 PRINT

Change: 100 HOME
300 HOME
460 PRINT TAB(7);"I";TAB(33);"I"

ATARI

Add: 92 DIM H\$(7),T\$(7),Y\$(7)
311 GOSUB 780
355 PRINT
461 GOSUB 780
462 PRINT "I";
463 R=25
464 GOSUB 780
465 PRINT "I";
501 GOSUB 780
502 PRINT "WANT ANOTHER PICTURE";
551 GOSUB 780
611 GOSUB 780
612 PRINT "I";
613 R=25
614 GOSUB 780
615 PRINT "I";
621 GOSUB 780
622 PRINT "I";
623 R=5
624 GOSUB 780
625 PRINT "<(O)>";
626 GOSUB 780
627 PRINT "<(O)>";
628 GOSUB 780
629 PRINT "I"
651 PRINT "I";
652 R=9
653 GOSUB 780
654 PRINT " -M- ";
655 GOSUB 780
656 PRINT "IXXXXX"
661 R=4
662 GOSUB 780
663 PRINT "I";
664 R=9
665 GOSUB 780
666 PRINT " MMM ";
667 GOSUB 780
668 PRINT "I";
669 R=4

(continued)

```

671 PRINT "X"
672 PRINT "  XXXXXI";
673 R=9
674 GOSUB 780
675 PRINT "(M-M)";
676 GOSUB 780
677 PRINT "IXXXX"
780 FOR B=1 TO R
781 PRINT " ";
782 NEXT B
783 RETURN

```

```

Change: 100 PRINT CHR$(125)
        280 T$=H$(1,1)
        300 PRINT CHR$(125)
        310 R=7
        460 R=7
        500 R=9
        520 IF Y$(1,1)<>"N" THEN 100
        550 R=7
        610 R=7
        620 R=7
        650 PRINT "  XXXXX";
        660 PRINT "  X";
        670 GOSUB 780

```

COMMODORE 64

```

Add: 345 PRINT
     465 PRINT
     585 PRINT

```

```

Change: 100 PRINT CHR$(147)
        300 PRINT CHR$(147)

```

COMMODORE VIC-20

```

Delete: 480, 490

```

```

Add: 345 PRINT
     465 PRINT
     585 PRINT

```

```

Change: 100 PRINT CHR$(147)
        160 PRINT "MONSTER HAVE: 2, 4, 6, OR 8"
        220 PRINT "MONSTER HAVE: 1 OR 2"
        260 PRINT "IS YOUR MONSTER HAPPY OR SAD"
        300 PRINT CHR$(147)
        310 PRINT TAB(3);
        320 FOR A=1 TO 15 STEP 2
        460 PRINT TAB(3);"I";TAB(12);"I"
        500 PRINT TAB(9);"WANT ANOTHER PICTURE"
        550 PRINT TAB(3)

```

```

610 PRINT TAB(3);"I";TAB(17);"I"
620 PRINT TAB(3);"I";TAB(5);"<(O)> <(O)>";TAB(17);
    "I"
650 PRINT "XXX";TAB(3);"I";TAB(8);" -M- ";TAB(17);
    "I";"XXX"
660 PRINT "X";TAB(3);"I";TAB(8);" MMM ";TAB(17);"I";
    TAB(20);"X"
670 PRINT "XXX";TAB(3); "I";TAB(8);"(M-M);TAB(17);
    "I"
680 PRINT TAB(3);"I";TAB(17);"I"
710 PRINT TAB(3);"I";TAB(6);"M           M";TAB(17);"I"
720 PRINT TAB(3);"I";TAB(7);"MMMMMMM";TAB(17);"I"
750 PRINT TAB(3);"I";TAB(7);"MMMMMMM";TAB(17);"I"
760 PRINT TAB(3);"I";TAB(6);"M           M";TAB(17);"I"

```

TEXAS INSTRUMENTS 99/4A

Add: 661 PRINT TAB(27);"I";TAB(28);"X"

Change:

```

100 CALL CLEAR
120 PRINT "IT'S TIME TO TAKE YOUR"
130 PRINT "MONSTER'S OFFICIAL PHOTO."
160 PRINT "MONSTER HAVE: 2, 4, OR 6";
190 IF (E<>1) * (E<>2) * (E<>3) THEN 140
280 T$=SEG$(H$,1,1)
290 IF (T$<>"H") * (T$<>"S") THEN 250
300 CALL CLEAR
310 PRINT TAB(5)
320 FOR A=1 TO 23 STEP 2
460 PRINT TAB(7);"I";TAB(27);"I";
500 PRINT TAB(6);"WANT ANOTHER PICTURE";
520 IF SEG$(Y$,1,1)<>"N" THEN 100
550 PRINT TAB(5)
560 FOR B=1 TO 23
610 PRINT TAB(5);"I";TAB(27);"I"
620 PRINT TAB(5);"I";TAB(11);"<(O)>";TAB(19);
    "<(O)>";TAB(27);"I"
650 PRINT TAB(4);"X";TAB(5);"I";TAB(14);" -M- ";
    TAB(27);"I";"X"
660 PRINT TAB(4);"X";TAB(5);"I";TAB(14);" MMM ";
670 PRINT TAB(4);"X";TAB(5);"I";TAB(14);"(M-M)";
    TAB(27);"I";"X"
680 PRINT TAB(5);"I";TAB(27);"I"
710 PRINT TAB(5);"I";TAB(12);"M           M";TAB(27);"I"
720 PRINT TAB(5);"I";TAB(12);"MMMMMMM ";TAB(27);"I"
750 PRINT TAB(5);"I";TAB(12);"MMMMMMM ";TAB(27);"I"
760 PRINT TAB(5);"I";TAB(12);"M           M";TAB(27);"I"

```

TRS-80 COLOR COMPUTER

Delete: 210, 220, 230, 240, 250, 610

Add: 345 PRINT
465 PRINT

(continued)

```

Change: 100 CLS
120 PRINT "IT'S TIME TO TAKE YOUR"
130 PRINT "MONSTER'S OFFICAL PHOTO."
160 PRINT "MONSTER HAVE: 2, 4, OR 6";
190 IF E<>1 AND E<>2 AND E<>3 THEN 140
300 CLS
310 PRINT TAB(3);
320 FOR A=1 TO 25 STEP 2
370 GOSUB 620
460 PRINT TAB(3);"I";TAB(27);"I";
550 PRINT TAB(3);
560 FOR B = 1 TO 25
620 PRINT TAB(3);"I";TAB(8);"<O>";TAB(17);"<O>";
TAB(27);"I"
650 PRINT "XXXI";TAB(14);"-M-";TAB(27);"IXXX"
660 PRINT "X";TAB(3);"I";TAB(14);"MMM";TAB(27);
"I";TAB(30);"X"
670 PRINT "XXXI";TAB(13);"M-M";TAB(27);"IXXX"
680 PRINT TAB(3);"I";TAB(27);"I"
710 PRINT TAB(3);"I";TAB(11);"M M";TAB(27);"I"
720 PRINT TAB(3);"I";TAB(11);" M M M M M ";TAB(27);"I"
750 PRINT TAB(3);"I";TAB(11);" M M M M M ";TAB(27);"I"
760 PRINT TAB(3);"I";TAB(11);"M M";TAB(27);"I"

```

Monstrous Numbers

What's the biggest number you can think of? Chances are it's nowhere near as big as these truly monstrous ones!

Sample Run

I AM THE KEEPER OF THE MONSTROUS
NUMBERS. IF YOU HIT THE ENTER KEY,
I WILL SHOW THEM TO YOU?

A BILLION
B TRILLION
C QUADRILLION
D QUINTILLION
E SEXTILLION
F SEPTILLION
G OCTILLION
H NONILLION
I DECILLION
J UNDECILLION
K DUODECILLION
L TREDECILLION
M QUATTUORDECILLION
N QUINDECILLION
O SEXDECILLION
P SEPTENDECILLION
Q OCTODECILLION
R NOVEMDECILLION
S VIGINTILLION
T CENTILLION
U GOOGOL

ENTER THE LETTER BESIDE THE NUMBER
YOU WISH TO SEE? R

AT YOUR SERVICE, MASTER!
ONE NOVEMDECILLION COMING RIGHT UP!

1,000,000,000,000,000,000,000,000,000,
000,000,000,000,000,000,000,000,000,
000,000

THERE! ONE NOVEMDECILLION!

MAY I SHOW YOU ANOTHER MONSTROUS NUMBER? NO

□ Program Listing

```
10 REM MONSTROUS NUMBERS
100 PRINT
110 PRINT "I AM THE KEEPER OF THE MONSTROUS"
120 PRINT "NUMBERS. IF YOU HIT THE ENTER KEY,"
130 PRINT "I WILL SHOW THEM TO YOU";
140 INPUT X$
150 PRINT
160 L$="ILLION"
170 Z$="000,"
180 FOR I=1 TO 20
190 READ P$
200 PRINT TAB(5);CHR$(I+64);TAB(8);P$;L$
210 NEXT I
220 PRINT TAB(5);"U";TAB(8);"GOOGOL"
230 PRINT "ENTER THE LETTER BESIDE THE NUMBER"
240 PRINT "YOU WISH TO SEE";
250 INPUT R$
260 IF R$="" THEN 300
270 R=ASC(R$)-64
280 IF R>0 AND R<21 THEN 370
290 IF R=21 THEN 320
300 RESTORE
310 GOTO 180
320 GOSUB 600
330 PRINT "10,000,";
340 R=31
350 K=6
360 GOTO 410
370 GOSUB 650
380 PRINT "1,000,";
390 K=5
400 IF R=20 THEN R=99
410 FOR I=1 TO R
420 FOR J=1 TO 600
430 NEXT J
440 PRINT Z$;
450 K=K+4
460 IF K>35 THEN PRINT
470 IF K>35 THEN K=0
480 NEXT I
490 FOR J=1 TO 600
500 NEXT J
510 PRINT "000"
520 PRINT
530 IF R<>31 THEN PRINT "THERE! ONE ";Q$;"ILLION!"
540 PRINT
550 PRINT "MAY I SHOW YOU ANOTHER MONSTROUS NUMBER";
560 INPUT Y$
570 IF MID$(Y$,1,1)="N" THEN END
580 RESTORE
590 GOTO 150
```

```

600 PRINT
610 PRINT "AT YOUR SERVICE, MASTER!"
620 PRINT "ONE GOOGOL COMING RIGHT UP!"
630 PRINT
640 RETURN
650 RESTORE
660 PRINT
670 PRINT
680 FOR I=1 TO R
690 READ Q$
700 NEXT I
710 PRINT "AT YOUR SERVICE, MASTER!"
720 PRINT "ONE ";Q$;"ILLION COMING RIGHT UP!"
730 PRINT
740 RETURN
750 DATA B,TR,QUADR,QUINT,SEXT,SEPT
760 DATA OCT,NON,DEC,UNDEC,DUODEC,TREDEC
770 DATA QUATTUORDEC,QUINDEC,SEXDEC
780 DATA SEPTENDEC,OCTODEC,NOVEMDEC
790 DATA VIGINT,CENT

```

If You Have . . .

APPLE II

Change: 120 PRINT "NUMBERS. IF YOU HIT THE RETURN KEY,"

ATARI

Add: 91 DIM L\$(6),Z\$(4),P\$(11),R\$(1),Y\$(11),X\$(11),
Q\$(11)

Change: 120 PRINT "NUMBERS. IF YOU HIT THE RETURN KEY,"
200 PRINT " ";CHR\$(I+64);" ";P\$;L\$
220 PRINT " ";"U GOOGOL"
460 IF K>31 THEN PRINT
470 IF K>31 THEN K=0
570 IF Y\$(1,1)="N" THEN END

COMMODORE 64

Add: 555 PRINT "NUMBER";

Change: 120 PRINT "NUMBERS. IF YOU HIT THE RETURN KEY,"
550 PRINT "MAY I SHOW YOU ANOTHER MONSTROUS"

COMMODORE VIC-20

Add: 205 IF I<>7 THEN 210
206 PRINT

(continued)

```
207 PRINT "FOR REST OF LIST HIT RETURN"
208 INPUT X$
555 PRINT "NUMBER"
```

```
Change: 120 PRINT "NUMBERS. IF YOU HIT THE RETURN KEY,"
200 PRINT TAB(2);CHR$(I+64);TAB(4);P$;L$
220 PRINT TAB(2);"U";TAB(4);"GOOGOL"
240 PRINT "YOU WISH TO SEE"
550 PRINT "MAY I SHOW YOU ANOTHER MONSTROUS NUMBER"
```

TEXAS INSTRUMENTS 99/4A

```
Add: 112 PRINT "MONSTROUS NUMBERS."
232 PRINT "BESIDE THE NUMBER"
402 R=99
462 PRINT
472 K=0
535 PRINT "THERE! ONE ";Q$;"ILLION!"
552 PRINT "MONSTROUS NUMBER";
722 PRINT "RIGHT UP!"
800 END
```

```
Change: 110 PRINT "I AM THE KEEPER OF THE"
120 PRINT "IF YOU HIT THE ENTER KEY,"
230 PRINT "ENTER THE LETTER"
280 IF (R>0) * (R<21) THEN 370
400 IF R<>20 THEN 410
460 IF K<24 THEN 470
470 IF K<24 THEN 480
530 IF R=31 THEN 540
550 PRINT "MAY I SHOW YOU ANOTHER"
570 IF SEG$(Y$,1,1)="N" THEN 800
720 PRINT "ONE ";Q$;"ILLION COMING"
```

TRS-80 COLOR COMPUTER

```
Add: 205 IF I=7 THEN GOTO 206 ELSE 210
206 PRINT
207 PRINT "FOR REST OF LIST, HIT ENTER.";
208 INPUT X$
```

```
Change: 120 PRINT "NUMBERS. IF YOU HIT THE ENTER"
130 PRINT "KEY, I WILL SHOW THEM TO YOU."
230 PRINT "ENTER THE LETTER BESIDE THE"
240 PRINT "NUMBER YOU WISH TO SEE.";
350 K=8
390 K=7
460 IF K>27 THEN PRINT
470 IF K>27 THEN K=0
710 PRINT "AT YOUR SERVICE, MASTER! ONE"
720 PRINT Q$;"ILLION COMING RIGHT UP!"
```


Monster Jungle

Talk about creepy places! In this jungle, plants grow right before your eyes! It's a perfect breeding ground for monsters, too — just run this program and see!

Sample Run

IN THE DEEPEST VALLEYS OF
RETUPMOC, THE BAMBOO GROWS
SO FAST YOU CAN SEE IT RISE.
IF YOU LOOK CLOSELY, YOU
MAY EVEN SEE A RARE CHIP-
MONSTER HIDING AMONG THE
BASES OF THE STALKS. BUT
FIRST YOU HAVE TO WAIT FOR
THE VALLEY'S MONSOON RAINS...

HOW MANY INCHES OF RAIN ARE FORECAST? 5



(continued)


```

370 PRINT
380 NEXT A
390 FOR B=1 TO 38 STEP 2
400 T(B)=INT(RND(1)*10)+1
410 NEXT B
420 REM --- GROWING STALKS ---
430 FOR C=1 TO 10
440 FOR D=1 TO 2
450 FOR E=1 TO 38
460 IF T(E)=0 THEN 540
470 IF T(E)>C THEN 540
480 IF C<10 THEN 520
490 IF D=1 THEN 520
500 IF E<>Q OR E=Q+1 THEN 520
510 PRINT "00";
520 PRINT "!";
530 GOTO 550
540 PRINT " ";
550 NEXT E
560 PRINT " "
570 NEXT D
580 NEXT C
590 PRINT

```

If You Have . . .

APPLE II

Delete: 240

Change: 100 HOME

ATARI

Delete: 240

Change: 100 PRINT CHR\$(125)
250 Q=(((INT(RND(0)*13))+3)*2)+1
280 FOR Y=1 TO 36
290 IF INT(RND(0)*2)=1 THEN 320
390 FOR B=1 TO 36 STEP 2
400 T(B)=INT(RND(0)*10)+1
450 FOR E=1 TO 36

COMMODORE 64

Delete: 240

Change: 100 PRINT CHR\$(147)
250 Q=(((INT(RND(0)*13))+3)*2)+1

(continued)

```
290 IF INT(RND(0)*2)=1 THEN 320
400 T(B)=INT(RND(0)*10)+1
```

COMMODORE VIC-20

Delete: 240

```
Change: 100 PRINT CHR$(147)
210 PRINT "HOW MANY INCHES OF RAIN ARE FORECAST"
250 Q=(((INT(RND(0)*13))+3)*2)+1
290 IF INT(RND(0)*2)=1 THEN 320
400 T(B)=INT(RND(0)*10)+1
450 FOR E=1 TO 21
```

TEXAS INSTRUMENTS 99/4A

Add: 212 PRINT "RAIN ARE FORECAST";

```
Change: 100 CALL CLEAR
210 PRINT "HOW MANY INCHES OF"
240 RANDOMIZE
250 Q=(((INT(RND*13))+3)*2)+1
280 FOR Y=1 TO 27
290 IF INT(RND*2)=1 THEN 320
390 FOR B=1 TO 27 STEP 2
400 T(B)=INT(RND*10)+1
450 FOR E=1 TO 27
500 IF (E<>Q) + (E=Q+1) THEN 520
```

TRS-80 COLOR COMPUTER

Delete: 240

```
Change: 100 CLS
250 Q=(RND(13)+3)*2+1
280 FOR Y=1 TO 30
290 IF RND(2)=1 THEN 320
390 FOR B=1 TO 31 STEP 2
400 T(B)=RND(10)+1
450 FOR E=1 TO 30
```

Monster Machine

You can create your own personal monster! With this program the horrible possibilities are endless.

Sample Run

WELCOME TO THE MONSTER MACHINE!
IT'S TIME TO CREATE YOUR OWN MONSTER!

HOW MANY HEADS DOES THE MONSTER HAVE? 123

THE HEADS MAY BE:

- 1 FUZZY
- 2 BLOODY
- 3 PIMPLY
- 4 SCABBY
- 5 WARTY
- 6 GNARLED
- 7 TWISTED
- 8 SLIMY
- 9 CREEPY
- 10 WEIRD

WHAT KIND OF HEADS ARE THEY? 6

THE HEADS MAY BE:

- 1 RED
- 2 ORANGE
- 3 YELLOW
- 4 GREEN
- 5 BLUE
- 6 PURPLE
- 7 PINK
- 8 SILVER
- 9 WHITE
- 10 BLACK

WHAT COLOR HEADS ARE THEY? 6

HOW MANY EYES DOES THE MONSTER HAVE? 36

THE EYES MAY BE:

- 1 FUZZY
- 2 BLOODY
- 3 PIMPLY
- 4 SCABBY

(continued)

- 5 WARTY
- 6 GNARLED
- 7 TWISTED
- 8 SLIMY
- 9 CREEPY
- 10 WEIRD

WHAT KIND OF EYES ARE THEY? 5

THE EYES MAY BE:

- 1 RED
- 2 ORANGE
- 3 YELLOW
- 4 GREEN
- 5 BLUE
- 6 PURPLE
- 7 PINK
- 8 SILVER
- 9 WHITE
- 10 BLACK

WHAT COLOR EYES ARE THEY? 7

[There's more—when you get to the end of the program, you'll see something like this:]

YOU HAVE CREATED A MONSTER!

THE MONSTER HAS

123 GNARLED PURPLE HEADS,
36 WARTY PINK EYES,
262 PIMPLY RED NOSES,
723 SLIMY GREEN MOUTHS,
87 CREEPY YELLOW ARMS,
3 GNARLED SILVER HANDS,
117 TWISTED ORANGE FINGERS,
1453 SCABBY WHITE LEGS,
6 WEIRD YELLOW FEET,
AND 6730 FUZZY BLUE TOES.

□ Program Listing

```
10 REM MONSTER MACHINE
100 PRINT "WELCOME TO THE MONSTER MACHINE!"
110 PRINT "IT'S TIME TO CREATE YOUR OWN MONSTER!"
120 PRINT
130 FOR I=1 TO 10
140 FOR Z=1 TO I
150 READ P$
160 NEXT Z
170 RESTORE
180 PRINT
190 PRINT "HOW MANY ";P$;" DOES THE MONSTER HAVE";
200 INPUT A(I)
210 IF A(I)>1 THEN 240
220 PRINT "THIS MONSTER HAS MORE THAN ONE!"
230 GOTO 190
240 PRINT
250 PRINT "THE ";P$;" MAY BE:"
260 PRINT
270 FOR J=1 TO 20
280 READ K$
290 IF J<20 AND J>10 THEN PRINT " ";
300 IF J>10 THEN PRINT J-10;" "; K$
310 NEXT J
320 RESTORE
330 PRINT
340 PRINT "WHAT KIND OF ";P$;" ARE THEY";
350 INPUT K(I)
360 IF K(I)<11 AND K(I)>0 THEN 390
370 PRINT "1 TO 10, PLEASE!"
380 GOTO 350
390 PRINT
400 PRINT "THE ";P$;" MAY BE:"
410 PRINT
420 FOR J=1 TO 30
430 READ C$
440 IF J<30 AND J>20 THEN PRINT " ";
450 IF J>20 THEN PRINT J-20;" ";C$
460 NEXT J
470 RESTORE
480 PRINT
490 PRINT "WHAT COLOR ";P$;" ARE THEY";
500 INPUT C(I)
510 IF C(I)<11 AND C(I)>0 THEN 540
520 PRINT "1 TO 10, PLEASE!"
530 GOTO 500
540 NEXT I
550 PRINT
560 PRINT "YOU HAVE CREATED A MONSTER!"
570 PRINT
580 PRINT "THE MONSTER HAS "
590 FOR I=1 TO 10
600 PRINT A(I); " ";
```

(continued)

```

610 FOR J=1 TO 10+K(I)
620 READ K$
630 NEXT J
640 RESTORE
650 PRINT K$;" ";
660 FOR J=1 TO 20+C(I)
670 READ C$
680 NEXT J
690 RESTORE
700 PRINT C$;" ";
710 FOR J=1 TO I
720 READ P$
730 NEXT J
740 RESTORE
750 PRINT P$;
760 IF I<10 THEN PRINT ","
770 IF I=9 THEN PRINT "AND ";
780 IF I=10 THEN PRINT "."
790 NEXT I
800 PRINT
810 DATA HEADS,EYES,NOSES,MOUTHS,ARMS
820 DATA HANDS,FINGERS,LEGS,FEET,TOES
830 DATA FUZZY,BLOODY,PIMPLY,SCABBY,WARTY
840 DATA GNARLED,TWISTED,SLIMY,CREEPY,WEIRD
850 DATA RED,ORANGE,YELLOW,GREEN,BLUE
860 DATA PURPLE,PINK,SILVER,WHITE,BLACK

```

If You Have . . .

APPLE II

Delete: 290

ATARI

Add: 91 DIM P\$(8),K\$(8),C\$(8),A(10),K(10),C(10)
201 A(I)=B
351 K(I)=E
501 C(I)=F

Change: 200 INPUT B
350 INPUT E
500 INPUT F

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Delete: 800

Change: 190 PRINT "HOW MANY ";P\$;" DOES THE MONSTER HAVE"
340 PRINT "WHAT KIND OF ";P\$;" ARE THEY"
490 PRINT "WHAT COLOR ";P\$;" ARE THEY"

TEXAS INSTRUMENTS 99/4A

Add: 112 PRINT "CREATE YOUR OWN MONSTER!"
192 PRINT "THE MONSTER HAVE";
222 PRINT "MORE THAN ONE!"
292 PRINT " ";
302 PRINT J-10;" ";K\$
342 PRINT "ARE THEY";
442 PRINT " " :
452 PRINT J-20;" ";C\$
492 PRINT "ARE THEY";
762 PRINT " , "
772 PRINT "AND " ;
782 PRINT " . "

Change: 100 PRINT "WELCOME TO THE MONSTER"
110 PRINT "MACHINE! IT'S TIME TO"
190 PRINT "HOW MANY ";P\$;" DOES"
220 PRINT "THIS MONSTER HAS"
290 IF (J<11) + (J>19) THEN 300
300 IF J<11 THEN 310
340 PRINT "WHAT KIND OF ";P\$
360 IF (K(I)<11) * (K(I)>0) THEN 390
440 IF (J<21) + (J>29) THEN 450
450 IF J<21 THEN 460
490 PRINT "WHAT COLOR ";P\$
510 IF (C(I)<11) * (C(I)>0) THEN 540
760 IF I>9 THEN 770
770 IF I<>9 THEN 780
780 IF I<>10 THEN 790

TRS-80 COLOR COMPUTER

No Changes Required

Abominable Snowman

Some people say there's no such thing as an Abominable Snowman. What do they know? The miracle of modern technology makes it possible for you to see this creature right on your own personal computer screen!

Sample Run

IT'S A STORMY NIGHT!
YOU CAN HARDLY SEE A THING!

(BUT IF YOU HIT THE ENTER KEY
EVERY TIME YOU SEE A QUESTION MARK
YOU MIGHT FIND THE ABOMINABLE
SNOWMAN LOOKING BACK AT YOU!)

?

o o

?

o o

?

o o

?

Break

Program Listing

```
10 REM ABOMINABLE SNOWMAN
100 PRINT "IT'S A STORMY NIGHT!"
110 PRINT "YOU CAN HARDLY SEE A THING!"
120 PRINT
130 PRINT "(BUT IF YOU HIT THE ENTER KEY"
140 PRINT "EVERY TIME YOU SEE A QUESTION MARK"
150 PRINT "YOU MIGHT FIND THE ABOMINABLE"
160 PRINT "SNOWMAN LOOKING BACK AT YOU!)"
170 PRINT
180 INPUT A$
190 FOR I=1 TO 25
200 PRINT
210 NEXT I
220 FOR I=1 TO 1+INT(RND(1)*34)
230 PRINT " ";
240 NEXT I
250 PRINT "O O"
260 FOR I=1 TO 1+INT(RND(1)*23)
270 PRINT
280 NEXT I
290 GOTO 180
```

If You Have . . .

APPLE II

Change: 130 PRINT "(BUT IF YOU HIT THE RETURN KEY"

ATARI

Add: 91 DIM A\$(1)
92 POKE 752,1

Change: 130 PRINT "(BUT IF YOU HIT THE RETURN KEY"
220 FOR I=1 TO 1+INT(RND(0)*34)
260 FOR I=1 TO 1+INT(RND(0)*23)

COMMODORE 64

Change: 130 PRINT "(BUT IF YOU HIT THE RETURN KEY"

COMMODORE VIC-20

Change: 130 PRINT "(BUT IF YOU HIT THE RETURN KEY"
220 FOR I=1 TO 1+INT(RND(1)*18)
260 FOR I=1 TO 1+INT(RND(1)*21)

(continued)

TEXAS INSTRUMENTS 99/4A

Add: 152 PRINT "ABOMINABLE SNOWMAN"

Change: 150 PRINT "YOU MIGHT FIND THE"
160 PRINT "LOOKING BACK AT YOU!"
220 FOR I=1 TO 1+INT(RND*24)
260 FOR I=1 TO 1+INT(RND*23)

TRS-80 COLOR COMPUTER

Change: 190 FOR I=1 TO 15
220 FOR I=1 TO RND(27)
260 FOR I=1 TO RND(14)

Monster Halloween

It's the annual Halloween parade in Monsterburg! The creatures are dressed in all sorts of disgusting costumes. But wait till you discover the scariest one of all!

Sample Run

WHAT'S YOUR NAME? CHIP MONGER

TELL ME THE NAME OF A FRIEND? SAM KEYS

TELL ME THE NAME OF ANOTHER FRIEND? NANCY DISK

TELL ME THE NAME OF SOMEBODY FAMOUS? JULIUS CAESAR

NAME SOMEONE ELSE WHO'S FAMOUS? BABE RUTH

TELL ME A TEACHER'S NAME? MRS. RODGERS

NOW A RELATIVE'S NAME, PLEASE? UNCLE IRVING

IT'S THE HALLOWEEN PARADE!

THERE'S FRANKENSTEIN'S MONSTER.
HE'S GOING AS MRS. RODGERS.

WOW! WHAT A SCARY COSTUME DRACULA HAS!
HE'S DRESSED AS SAM KEYS.

BOY, THAT FACE IS THE UGLIEST
I'VE EVER SEEN. THE OGRE'S
GOT A NANCY DISK MASK!

YUCK! WHAT A CREEPY WITCH!
IT'S JULIUS CAESAR IN DISGUISE!

NOW, THERE'S A HILARIOUS OUTFIT,
ALL RIGHT. IT'S THE BLOB, DRESSED
IN AN UNCLE IRVING SUIT.

AND THAT HORRIBLE BABE RUTH COSTUME
DOESN'T SCARE ME. I KNOW THE MUMMY'S
INSIDE.

BUT THE FIRST PRIZE FOR SCARINESS GOES TO
THE ABOMINABLE SNOWMAN. HE'S DRESSED UP
AS CHIP MONGER!

□ Program Listing

```
10 REM MONSTER HALLOWEEN
100 GOSUB 700
110 PRINT "WHAT'S YOUR NAME?"
120 INPUT Y$
130 IF Y$="" THEN 110
140 PRINT
150 PRINT "TELL ME THE NAME OF A FRIEND:"
160 INPUT F1$
170 IF F1$="" THEN 150
180 PRINT
190 PRINT "TELL ME THE NAME OF ANOTHER FRIEND:"
200 INPUT F2$
210 IF F2$="" THEN 190
220 PRINT
230 PRINT "TELL ME THE NAME OF SOMEBODY FAMOUS:"
240 INPUT C1$
250 IF C1$="" THEN 230
260 PRINT
270 PRINT "NAME SOMEONE ELSE WHO'S FAMOUS:"
280 INPUT C2$
290 IF C2$="" THEN 270
300 PRINT
310 PRINT "TELL ME A TEACHER'S NAME:"
320 INPUT T$
330 IF T$="" THEN 310
340 PRINT
350 PRINT "NOW A RELATIVE'S NAME, PLEASE:"
360 INPUT R$
370 IF R$="" THEN 350
380 GOSUB 700
390 PRINT "IT'S THE HALLOWEEN PARADE!"
400 PRINT
410 PRINT "THERE'S FRANKENSTEIN'S MONSTER."
420 PRINT "HE'S GOING AS ";T$;"."
430 GOSUB 660
440 PRINT "WOW! WHAT A SCARY COSTUME DRACULA HAS!"
450 PRINT "HE'S DRESSED AS ";F1$;"."
460 GOSUB 660
470 PRINT "BOY, THAT FACE IS THE UGLIEST"
480 PRINT "I'VE EVER SEEN. THE OGRE'S"
490 PRINT "GOT A ";F2$;" MASK!"
500 GOSUB 660
510 PRINT "YUCK! WHAT A CREEPY WITCH!"
520 PRINT "IT'S ";C1$;" IN DISGUISE!"
530 GOSUB 660
540 PRINT "NOW, THERE'S A HILARIOUS OUTFIT,"
550 PRINT "ALL RIGHT. IT'S THE BLOB, DRESSED"
560 PRINT "IN A ";R$;" SUIT."
570 GOSUB 660
580 PRINT "AND THAT HORRIBLE ";C2$
590 PRINT "COSTUME DOESN'T SCARE ME. I KNOW"
600 PRINT "THE MUMMY'S INSIDE."
```

```

610 GOSUB 660
620 PRINT "BUT THE FIRST PRIZE FOR SCARINESS"
630 PRINT "GOES TO THE ABOMINABLE SNOWMAN."
640 PRINT "HE'S DRESSED UP AS ";Y$;"!"
650 END
660 FOR D=1 TO 1200
670 NEXT D
680 PRINT
690 RETURN
700 FOR I=1 TO 25
710 PRINT
720 NEXT I
730 RETURN

```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM Y\$(40),F1\$(40),F2\$(40),C1\$(40),C2\$(40),
T\$(40),R\$(40)

COMMODORE 64 No Changes Required

COMMODORE VIC-20 No Changes Required

TEXAS INSTRUMENTS 99/4A

Add: 192 PRINT "ANOTHER FRIEND"
232 PRINT "OF SOMEBODY FAMOUS"
272 PRINT "WHO IS FAMOUS"
412 PRINT "FRANKENSTEIN'S MONSTER."
442 PRINT "COSTUME DRACULA HAS!"
542 PRINT "HILARIOUS OUTFIT, ALL RIGHT."
622 PRINT "SCARINESS GOES TO THE"

Change: 190 PRINT "TELL ME THE NAME OF"
230 PRINT "TELL ME THE NAME"
270 PRINT "NAME SOMEONE ELSE"
410 PRINT "THERE'S"
440 PRINT "WOW! WHAT A SCARY"
540 PRINT "NOW, THERE'S A"
550 PRINT "IT'S THE BLOB, DRESSED IN"
560 PRINT "A ";R\$;" SUIT."
580 PRINT "AND THAT HORRIBLE ";C2\$

(continued)

```
590 PRINT "COSTUME DOESN'T SCARE ME. I"  
600 PRINT "KNOW THE MUMMY'S INSIDE."  
620 PRINT "BUT THE FIRST PRIZE FOR"  
630 PRINT "ABOMINABLE SNOWMAN. HE'S"  
640 PRINT "DRESSED UP AS ";Y$;"!"  
660 FOR D=1 TO 800
```

TRS-80 COLOR COMPUTER No Changes Required

Dragonslayer

A ferocious dragon is terrorizing the countryside with its foul breath. You are the only person in the kingdom who can possibly slay it. Pick up your sword, shield, bow and arrow, and hope that your training serves you well.

Sample Run

YOU'RE THE DRAGONSLAYER.

HOW MANY DRAGONS HAVE YOU DEFEATED? 10

YOU HAVE A FEARSOME SWORD.
BUT YOU MUST BE 1 PACE AWAY
FROM THE DRAGON TO USE IT.

YOU HAVE A BOW, BUT ONLY 2 ARROWS.

YOU HAVE A SHIELD, BUT IT CAN ONLY
TAKE 5 HITS OF DRAGONFIRE.

THE SWORD OR THE ARROW WILL KILL
THE DRAGON--IF YOUR AIM IS TRUE.

YOU ARE 6 PACES FROM THE DRAGON.
WHAT NOW?

1: SHOOT ARROW
2: STAB DRAGON
3: MOVE CLOSER
4: MOVE AWAY
? 3

A CURL OF FIRE SHOOTS FROM THE
DRAGON'S NOSTRILS!

LUCKY YOU! IT MISSED!

YOU ARE 5 PACES FROM THE DRAGON.
WHAT NOW?

1: SHOOT ARROW
2: STAB DRAGON
3: MOVE CLOSER
4: MOVE AWAY
? 1

POOR AIM! YOU MISSED!

(continued)

A CURL OF FIRE SHOOTS FROM THE
DRAGON'S NOSTRILS!

YOUR SHIELD PROTECTS YOU FROM
THE BLAST.

*[There's more—when you get to the end of the program, you'll see
something like this:]*

YOU ARE 1 PACE FROM THE DRAGON.
WHAT NOW?

1: SHOOT ARROW
2: STAB DRAGON
3: MOVE CLOSER
4: MOVE AWAY
? 2

YOU'VE DONE IT!
YOU'VE SLAIN THE DRAGON!

□ Program Listing

```
10 REM DRAGONSLAYER
100 PRINT "YOU'RE THE DRAGONSLAYER."
110 PRINT
120 PRINT "HOW MANY DRAGONS HAVE YOU DEFEATED";
130 INPUT D
140 IF D<32767 THEN 170
150 PRINT "TOO MANY!"
160 GOTO 120
170 RANDOMIZE D
180 AR=2+INT(RND(1)*5)
190 P=2+INT(RND(1)*19)
200 SH=1+P-INT(RND(1)*P)
210 PRINT
220 PRINT "YOU HAVE A FEARSOME SWORD."
230 PRINT "BUT YOU MUST BE 1 PACE AWAY"
240 PRINT "FROM THE DRAGON TO USE IT."
250 PRINT
260 PRINT "YOU HAVE A BOW, BUT ONLY ";AR;" ARROWS."
270 PRINT
280 PRINT "YOU HAVE A SHIELD, BUT IT CAN ONLY"
290 PRINT "TAKE ";SH;"HITS OF DRAGONFIRE."
300 PRINT
310 PRINT "THE SWORD OR THE ARROW WILL KILL"
320 PRINT "THE DRAGON--IF YOUR AIM IS TRUE."
330 PRINT
340 PRINT "YOU ARE ";P;" PACES FROM THE DRAGON."
350 PRINT "WHAT NOW?"
```

```

360 PRINT
370 PRINT "1: SHOOT ARROW"
380 PRINT "2: STAB DRAGON"
390 PRINT "3: MOVE CLOSER"
400 PRINT "4: MOVE AWAY"
410 INPUT C
420 PRINT
430 IF C<1 OR C>4 THEN 450
440 ON C GOTO 520,570,650,480
450 PRINT "SINCE YOU CAN'T COUNT,"
460 PRINT "YOU DO ABSOLUTELY NOTHING!"
470 GOTO 680
480 P=P+1
490 PRINT "CHICKEN!"
500 PRINT "COME BACK AND FIGHT LIKE A KNIGHT!"
510 GOTO 680
520 IF AR=0 THEN 670
530 IF INT(RND(1)*(P+1))=P THEN 890
540 AR=AR-1
550 PRINT "POOR AIM! YOU MISSED!"
560 GOTO 680
570 IF P>1 THEN 610
580 IF INT(RND(1)*2)=1 THEN 890
590 PRINT "MISSED!"
600 GOTO 680
610 PRINT "THE ONLY THING YOU STABBED WAS THE AIR!"
620 PRINT "YOU HAVE TO BE ONE PACE AWAY"
630 PRINT "TO USE YOUR SWORD!"
640 GOTO 680
650 P=P-1
660 IF P=0 THEN 840
670 IF AR=0 THEN PRINT "YOU'RE OUT OF ARROWS!"
680 PRINT "A CURL OF FIRE SHOOTS FROM THE"
690 PRINT "DRAGON'S NOSTRILS!"
700 PRINT
710 IF 1+INT(RND(1)*20)<P THEN 780
720 SH=SH-1
730 IF SH<0 THEN 800
740 PRINT "YOUR SHIELD PROTECTS YOU FROM"
750 PRINT "THE BLAST."
760 PRINT
770 GOTO 330
780 PRINT "LUCKY YOU! IT MISSED!"
790 GOTO 330
800 PRINT
810 PRINT "YOUR SHIELD CAN'T PROTECT YOU."
820 PRINT "YOU ARE ROASTED ALIVE!"
830 END
840 PRINT "SILLY YOU!"
850 PRINT "YOU WALKED RIGHT INTO THE"
860 PRINT "DRAGON'S CLAWS! YOU WON'T"
870 PRINT "LIVE TO SEE ANOTHER ONE!"
880 END
890 PRINT "YOU'VE DONE IT!"
900 PRINT "YOU'VE SLAIN THE DRAGON!"

```

If You Have . . .

APPLE II

Change: 170 REM
290 PRINT "TAKE ";SH;" HITS OF DRAGONFIRE."

ATARI

Delete: 170

Change: 140 IF D<32767 THEN 180
180 AR=2+INT(RND(0)*5)
190 P=2+INT(RND(0)*19)
200 SH=1+P-INT(RND(0)*P)
530 IF INT(RND(0)*(P+1))=P THEN 890
580 IF INT(RND(0)*2)=1 THEN 890
710 IF 1+INT(RND(0)*20)<P THEN 780

COMMODORE 64

Change: 170 REM
180 AR=2+INT(RND(0)*5)
190 P=2+INT(RND(0)*19)
200 SH=1+P-INT(RND(0)*P)
530 IF INT(RND(0)*(P+1))=P THEN 890
580 IF INT(RND(0)*2)=1 THEN 890
710 IF INT(RND(0)*20)<P THEN 780

COMMODORE VIC-20

Add: 334 PRINT "HIT RETURN TO CONTINUE"
335 INPUT X\$

Change: 120 PRINT "HOW MANY DRAGONS HAVE YOU DEFEATED"
170 REM
180 AR=2+INT(RND(0)*5)
190 P=2+INT(RND(0)*19)
200 SH=1+P-INT(RND(0)*P)
530 IF INT(RND(0)*(P+1))=P THEN 890
580 IF INT(RND(0)*2)=1 THEN 890
710 IF INT(RND(0)*20)<P THEN 780

TEXAS INSTRUMENTS 99/4A

Add: 122 PRINT "HAVE YOU DEFEATED";
262 PRINT "ONLY ";AR;" ARROWS."
282 PRINT "IT CAN ONLY TAKE"
312 PRINT "KILL THE DRAGON--IF"
502 PRINT "LIKE A KNIGHT!"

```
612 PRINT "STABBED WAS THE AIR!"
675 PRINT "YOU'RE OUT OF ARROWS!"
```

```
Change: 120 PRINT "HOW MANY DRAGONS"
170 RANDOMIZE
180 AR=2+INT(RND*5)
190 P=2+INT(RND*19)
200 SH=1+P-INT(RND*P)
260 PRINT "YOU HAVE A BOW, BUT"
280 PRINT "YOU HAVE A SHIELD, BUT"
290 PRINT SH;"HITS OF DRAGONFIRE."
310 PRINT "THE SWORD OR THE ARROW WILL"
320 PRINT "YOUR AIM IS TRUE."
340 PRINT "YOU ARE ";P;" PACES FROM"
350 PRINT "THE DRAGON. WHAT NOW?"
430 IF (C<1) + (C>4) THEN 450
500 PRINT "COME BACK AND FIGHT"
530 IF INT(RND*(P+1))=P THEN 890
580 IF INT(RND*2)=1 THEN 890
610 PRINT "THE ONLY THING YOU"
670 IF AR>0 THEN 680
680 PRINT "A CURL OF FIRE SHOOTS FROM"
690 PRINT "THE DRAGON'S NOSTRILS!"
710 IF 1+INT(RND*20)<P THEN 780
740 PRINT "YOUR SHIELD PROTECTS YOU"
750 PRINT "FROM THE BLAST."
810 PRINT "YOUR SHIELD CAN'T PROTECT"
820 PRINT "YOU. YOU ARE ROASTED ALIVE!"
```

TRS-80 COLOR COMPUTER

Delete: 210, 250, 270, 300, 360, 700, 760

```
Add: 345 IF A=1 THEN 350 ELSE 346
346 PRINT "HIT ENTER TO START";
347 A=1
348 INPUT X$
```

```
Change: 140 IF D<32767 THEN 180
180 AR=RND(5)
190 P=RND(19)
200 SM=1+P-RND(P)
260 PRINT "YOU HAVE A BOW,BUT ONLY";AR;"ARROWS";
280 PRINT "YOU HAVE A SHIELD, BUT IT CAN"
290 PRINT "ONLY TAKE";SH;"HITS OF DRAGONFIRE."
310 PRINT "SWORD OR ARROW WILL KILL THE"
320 PRINT "DRAGON--IF YOUR AIM IS TRUE."
340 PRINT "YOU ARE";P;"PACES FROM THE DRAGON"
530 IF RND(P+1)=P THEN 890
580 IF RND(2)=1 THEN 890
710 IF RND(20)<P THEN 780
730 IF SH<0 THEN 810
770 GOTO 340
790 GOTO 340
```


Monster Cafe

The Monster Cafe likes to please its customers, and no wonder — when monsters don't like the food, they eat the employees instead! You're guaranteed to find something horribly wonderful on the menu!

Sample Run

WELCOME TO THE MONSTER CAFE!
WE AIM TO PLEASE!

PLEASE TAKE THE TIME
TO FILL OUT THIS SIMPLE
LIST OF QUESTIONS SO THAT
WE MAY SERVE YOU BETTER.

NAME SOMETHING SICKENING? GARBAGE
EXCELLENT!
NAME SOMETHING SLIMY? WORMS
HOW TASTEFUL!
WHAT DO YOU FIND UNPLEASANT? CIGARS
UNUSUAL . . .
WHAT'S REALLY AWFUL? OLD SOCKS
HMMM . . .
WHO'S MORE HORRIBLE THAN FRANKENSTEIN? DRACULA
I THINK WE'RE ALL OUT OF THAT--
NAME SOMEONE ELSE HORRIBLE? THE MUMMY
THANKS. WHAT'S TRULY UGLY? JOHNSON SCHOOL
I AGREE.
NOW NAME SOMETHING REPULSIVE? GOPHER GUTS
SURE IS. AND FINALLY,
NAME SOMETHING DISGUSTING? UNCLE IRVING

MONSTER CAFE MENU

APPETIZERS

WORMS COCKTAIL	\$ 15.99
GOPHER GUTS JUICE	\$ 50.99

ENTREES

FILET OF JOHNSON SCHOOL	\$ 10.99
ROAST DRACULA	\$ 31.99
FRENCH FRIED UNCLE IRVING	\$ 62.99
STEWED CIGARS	\$ 67.99

DESSERTS

OLD SOCKS CAKE	\$ 71.99
GARBAGE PIE	\$ 13.99

ENJOY YOUR MEAL!

□ Program Listing

```
10 REM MONSTER CAFE
100 PRINT
110 DIM P(8)
120 PRINT "WELCOME TO THE MONSTER CAFE!"
130 PRINT "WE AIM TO PLEASE!"
140 PRINT
150 PRINT "PLEASE TAKE THE TIME"
160 PRINT "TO FILL OUT THIS SIMPLE"
170 PRINT "LIST OF QUESTIONS SO THAT"
180 PRINT "WE MAY SERVE YOU BETTER."
190 PRINT
200 PRINT "NAME SOMETHING SICKENING:"
210 INPUT SK$
220 IF SK$="" THEN 200
230 PRINT "EXCELLENT!"
240 PRINT "NAME SOMETHING SLIMY:"
250 INPUT SL$
260 IF SL$="" THEN 240
270 PRINT "HOW TASTEFUL!"
280 PRINT "WHAT DO YOU FIND UNPLEASANT?"
290 INPUT UN$
300 IF UN$="" THEN 280
310 PRINT "UNUSUAL . . . "
320 PRINT "WHAT'S REALLY AWFUL?"
330 INPUT AWFUL$
340 IF AWFUL$="" THEN 320
350 PRINT "HMMM . . . "
360 PRINT "WHO'S MORE HORRIBLE THAN FRANKENSTEIN?"
370 INPUT HR$
380 IF HR$="" THEN 360
390 PRINT "I THINK WE'RE ALL OUT OF THAT--"
400 PRINT "NAME SOMEONE ELSE HORRIBLE:"
410 IF RND(1)>.5 THEN 440
420 INPUT HR$
430 GOTO 450
440 INPUT X$
450 PRINT "THANKS. WHAT'S TRULY UGLY?"
460 INPUT UG$
470 IF UG$="" THEN 450
480 PRINT "I AGREE."
490 PRINT "NOW NAME SOMETHING REPULSIVE:"
500 INPUT RP$
510 IF RP$="" THEN 490
520 PRINT "SURE IS. AND FINALLY,"
530 PRINT "NAME SOMETHING DISGUSTING:"
540 INPUT DG$
550 IF DG$="" THEN 530
560 FOR I=1 TO 8
570 P(I)=10+INT(RND(1)*80)+.99
580 NEXT I
590 PRINT
600 PRINT TAB(14);"MONSTER CAFE"
```

(continued)

```

610 PRINT TAB(18);"MENU"
620 PRINT
630 PRINT TAB(15); "APPETIZERS"
640 PRINT SL$;" COCKTAIL";TAB(30);"$";P(1)
650 PRINT RP$;" JUICE";TAB(30);"$";P(2)
660 PRINT
670 PRINT TAB(16);"ENTREES"
680 PRINT "FILET OF ";UG$;TAB(30);"$";P(3)
690 PRINT "ROAST ";HR$;TAB(30);"$";P(4)
700 PRINT "FRENCH FRIED ";DG$;TAB(30);"$";P(5)
710 PRINT "STEWED ";UN$;TAB(30);"$";P(6)
720 PRINT
730 PRINT TAB(16);"DESSERTS"
740 PRINT AWFUL$;" CAKE";TAB(30);"$";P(7)
750 PRINT SK$;" PIE";TAB(30);"$";P(8)
760 PRINT
770 PRINT TAB(12);"ENJOY YOUR MEAL!"
780 PRINT

```

If You Have . . .

APPLE II No Changes Required

ATARI

```

Add: 111 DIM SK$(40),SL$(40),UN$(40),AWFUL$(40),
      HR$(40),X$(40)
      112 DIM UG$(40),RP$(40),DG$(40)

```

```

Change: 410 IF RND(0)>.5 THEN 440
        570 P(I)=10+INT(RND(0)*80)+.99
        600 PRINT "      ", "MONSTER CAFE"
        610 PRINT "      "; "MENU"
        630 PRINT "      ", "APPETIZERS"
        640 PRINT SL$;" COCKTAIL"; "      "; "$"; P(1)
        650 PRINT RP$;" JUICE"; "      "; "$"; P(2)
        670 PRINT "      "; "ENTREES"
        680 PRINT "FILET OF ";UG$;"      "; "$"; P(3)
        690 PRINT "ROAST ";HR$;"      "; "$"; P(4)
        700 PRINT "FRENCH FRIED ";DG$;"      "; "$"; P(5)
        710 PRINT "STEWED ";UN$;"      "; "$"; P(6)
        730 PRINT "      "; "DESSERTS"
        740 PRINT AWFUL$;" CAKE"; "      "; "$"; P(7)
        750 PRINT SK$;" PIE"; "      "; "$"; P(8)
        770 PRINT "      ", "ENJOY YOUR MEAL!"

```

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Add: 665 PRINT "HIT RETURN TO SEE ENTREES"
667 INPUT EN\$
725 PRINT "HIT RETURN TO SEE DESSERTS"
727 INPUT DS\$

Change: 600 PRINT TAB(7); "MONSTER CAFE"
610 PRINT TAB(8); "MENU"
630 PRINT TAB(6); "APPETIZERS"
640 PRINT SL\$; " COCKTAIL"; TAB(16); "\$"; P(1)
650 PRINT RP\$; " JUICE"; TAB(16); "\$"; P(2)
670 PRINT TAB(6); "ENTREES"
680 PRINT "FILET OF "; UG\$; TAB(16); "\$"; P(3)
690 PRINT "ROAST "; HR\$; TAB(16); "\$"; P(4)
700 PRINT "FRENCH FRIED "; DG\$; TAB(16); "\$"; P(5)
710 PRINT "STEWED "; UN\$; TAB(16); "\$"; P(6)
730 PRINT TAB(6); "DESSERTS"
740 PRINT AWFUL\$; " CAKE"; TAB(16); "\$"; P(7)
750 PRINT SK\$; " PIE"; TAB(16); "\$"; P(8)

TEXAS INSTRUMENTS 99/4A

Add: 365 PRINT "THAN FRANKENSTEIN"
395 PRINT "OUT OF THAT--"

Change: 360 PRINT "WHO'S MORE HORRIBLE"
390 PRINT "I THINK WE'RE ALL"
410 IF RND>0.5 THEN 440
570 P(I)=10+INT(RND*80)+0.99
640 PRINT SL\$; " COCKTAIL"; TAB(20); "\$"; P(1)
650 PRINT RP\$; " JUICE"; TAB(20); "\$"; P(2)
680 PRINT "FILET OF "; UG\$; TAB(20); "\$"; P(3)
690 PRINT "ROAST "; HR\$; TAB(20); "\$"; P(4)
700 PRINT "FRENCH FRIED "; DG\$; TAB(20); "\$"; P(5)
710 PRINT "STEWED "; UN\$; TAB(20); "\$"; P(6)
740 PRINT AWFUL\$; " CAKE"; TAB(20); "\$"; P(7)
750 PRINT SK\$; " PIE"; TAB(20); "\$"; P(8)

TRS-80 COLOR COMPUTER

Delete: 620, 660, 720, 760, 780

Change: 570 P(I)=10+RND(80)+.99
600 PRINT TAB(8); "MONSTER CAFE"
610 PRINT TAB(12); "MENU"
630 PRINT TAB(9); "APPETIZERS"
640 PRINT SL\$; " COCKTAIL"; TAB(23); "\$"; P(1)
650 PRINT RP\$; " JUICE"; TAB(23); "\$"; P(2)
670 PRINT TAB(10); "ENTREES"
680 PRINT "FILET OF "; UG\$; TAB(23); "\$"; P(3)

(continued)

```
690 PRINT "ROAST ";HR$;TAB(23);"$";P(4)
700 PRINT "FRENCH FRIED ";DG$;TAB(23);"$";P(5)
710 PRINT "STEWED ";UN$;TAB(23);"$";P(6)
730 PRINT TAB(10);"DESSERTS"
740 PRINT AWFUL$;" CAKE";TAB(23);"$";P(7)
750 PRINT SK$;" PIE";TAB(23);"$";P(8)
```

Loch Ness Monster Hunt

Does the Loch Ness monster really exist? Many Scots think so — and just as many think not! If you run this program, you may find “Nessie” living in your own computer.

Sample Run

CAN YOU FIND THE LOCH NESS MONSTER?
LOCH NESS IS IN SCOTLAND.
IT'S TEN MILES WIDE AND TEN MILES LONG.
YOUR SUBMARINE'S SONAR WILL BEEP ONCE
IF YOU ARE THREE MILES FROM THE MONSTER.
IT WILL BEEP TWICE IF YOU ARE TWO MILES
AWAY. AND IT WILL BEEP THREE TIMES
IF YOU GET WITHIN ONE MILE.
YOUR SUB WILL MOVE ONE MILE EACH TURN.
SO WILL THE MONSTER (UNLESS IT DOESN'T
MOVE AT ALL).

HOW OLD DO YOU THINK THE MONSTER IS? 129

YOU ARE 6 MILES SOUTH OF THE NORTH
SHORE OF THE LOCH AND 6 MILES EAST
OF THE WEST SHORE.

MONSTER OUT OF SONAR RANGE . . .

(N, E, S, W, NE, NW, SE, SW):
YOUR MOVE? S

MONSTER IN SONAR RANGE . . .

(N, E, S, W, NE, NW, SE, SW):
YOUR MOVE? E

MONSTER IN SONAR RANGE . . .

(N, E, S, W, NE, NW, SE, SW):
YOUR MOVE? SW

MONSTER IN SONAR RANGE . . .

(N, E, S, W, NE, NW, SE, SW):
YOUR MOVE? SE

(continued)

FOUND HIM!
 IT TOOK YOU 5 TRIES.
 GREAT WORK, EXPLORER!

```

      00      00      00      00
     0 0    0 0    0 0    0 0
    0  0 0  0  0 0  0  0  0  0
   0   0 0  0  0  0  0  0  0  0 0:<
        00      00      00      00
  
```

□ Program Listing

```

10 REM LOCH NESS MONSTER HUNT
100 PRINT
110 PRINT "CAN YOU FIND THE LOCH NESS MONSTER?"
120 PRINT "LOCH NESS IS IN SCOTLAND."
130 PRINT "IT'S TEN MILES WIDE AND TEN MILES LONG."
140 PRINT "YOUR SUBMARINE'S SONAR WILL BEEP ONCE"
150 PRINT "IF YOU ARE THREE MILES FROM THE MONSTER."
160 PRINT "IT WILL BEEP TWICE IF YOU ARE TWO MILES"
170 PRINT "AWAY. AND IT WILL BEEP THREE TIMES"
180 PRINT "IF YOU GET WITHIN ONE MILE."
190 PRINT "YOUR SUB WILL MOVE ONE MILE EACH TURN."
200 PRINT "SO WILL THE MONSTER (UNLESS IT DOESN'T"
210 PRINT "MOVE AT ALL)."
220 PRINT
230 PRINT "HOW OLD DO YOU THINK THE MONSTER IS";
240 INPUT R
250 PRINT
260 RANDOMIZE R
270 YA=INT(RND(1)*11)
280 YB=INT(RND(1)*11)
290 MA=INT(RND(1)*11)
300 MB=INT(RND(1)*11)
310 PRINT "YOU ARE ";YA;" MILES SOUTH OF THE NORTH"
320 PRINT "SHORE OF THE LOCH AND ";YB;" MILES EAST"
330 PRINT "OF THE WEST SHORE."
340 IF YA=MA AND YB=MB THEN 830
350 IF ABS(YA-MA)<4 AND ABS(YB-MB)<4 THEN 400
360 PRINT
370 PRINT "MONSTER OUT OF SONAR RANGE . . ."
380 PRINT
390 GOTO 500
400 PRINT
410 PRINT "MONSTER IN SONAR RANGE . . ."
420 FOR I=3 TO 1 STEP -1
430 IF ABS(YA-MA)>I THEN 480
440 IF ABS(YB-MB)>I THEN 480
450 PRINT CHR$(7);
460 FOR J=1 TO 200
470 NEXT J
480 NEXT I
490 PRINT
  
```

```

500 PRINT "(N, E, S, W, NE, NW, SE, SW):"
510 PRINT "YOUR MOVE";
520 INPUT MOVE$
530 IF MOVE$="N" THEN YA=YA-1
540 IF MOVE$="S" THEN YA=YA+1
550 IF MOVE$="W" THEN YB=YB-1
560 IF MOVE$="E" THEN YB=YB+1
570 IF MOVE$="NW" THEN YA=YA-1
580 IF MOVE$="NW" THEN YB=YB-1
590 IF MOVE$="NE" THEN YA=YA-1
600 IF MOVE$="NE" THEN YB=YB+1
610 IF MOVE$="SW" THEN YA=YA+1
620 IF MOVE$="SW" THEN YB=YB-1
630 IF MOVE$="SE" THEN YA=YA+1
640 IF MOVE$="SE" THEN YB=YB+1
650 IF YA<11 AND YB<11 AND YA>=0 AND YB>=0 THEN 750
660 PRINT "YOUR SUB WON'T WORK ON LAND!"
670 IF YA>10 THEN PRINT "YOU'RE AT THE SOUTH SHORE!"
680 IF YA<0 THEN PRINT "YOU'RE AT THE NORTH SHORE!"
690 IF YB>10 THEN PRINT "YOU'RE AT THE EAST SHORE!"
700 IF YB<0 THEN PRINT "YOU'RE AT THE WEST SHORE!"
710 IF YB>10 THEN YB=10
720 IF YB<1 THEN YB=0
730 IF YA<0 THEN YA=0
740 IF YA>10 THEN YA=10
750 MA=MA-1+INT(RND(1)*3)
760 MB=MB-1+INT(RND(1)*3)
770 IF MA>10 THEN MA=10
780 IF MA<0 THEN MA=0
790 IF MB>10 THEN MB=10
800 IF MB<0 THEN MB=0
810 COUNT=COUNT+1
820 GOTO 340
830 PRINT
840 PRINT "FOUND HIM!"
850 PRINT "IT TOOK YOU ";COUNT;" TRIES."
860 IF COUNT<10 THEN PRINT "GREAT WORK, EXPLORER!"
870 IF COUNT>25 THEN PRINT "UNFORTUNATELY, HE FOUND
YOU FIRST!"
880 PRINT
890 FOR Q=1 TO 25 STEP 8
900 PRINT TAB(Q+4);"00";
910 NEXT Q
920 FOR Q=1 TO 25 STEP 8
930 PRINT TAB(Q+3);"0 0";
940 NEXT Q
950 FOR Q=1 TO 25 STEP 8
960 PRINT TAB(Q+2);"0 0";
970 NEXT Q
980 FOR Q=1 TO 25 STEP 8
990 PRINT TAB(Q+2);"0 0";
1000 IF Q>24 THEN PRINT " ";";"0:<"
1010 NEXT Q
1020 FOR Q=1 TO 25 STEP 8
1030 PRINT TAB(Q+8);"00";
1040 NEXT Q

```


If You Have . . .

APPLE II

Delete: 260

ATARI

Delete: 260

Add: 91 DIM MOVE\$(2)
901 GOSUB 1055
902 PRINT "00";
931 GOSUB 1055
932 PRINT "0 0";
961 GOSUB 1055
962 PRINT "0 0";
991 GOSUB 1055
992 PRINT "0 0";
1031 GOSUB 1055
1032 PRINT "00";
1055 FOR A=1 TO Q
1056 PRINT " ";
1057 NEXT Q
1058 RETURN

Change: 270 YA=INT(RND(0)*11)
280 YB=INT(RND(0)*11)
290 MA=INT(RND(0)*11)
300 MB=INT(RND(0)*11)
450 PRINT CHR\$(253)
750 MA=MA-1+INT(RND(0)*3)
760 MB=MB-1+INT(RND(0)*3)
900 PRINT " ";
930 PRINT " ";
960 PRINT " ";
990 PRINT " ";
1030 PRINT "0000000";

COMMODORE 64

Add: 915 PRINT
945 PRINT
975 PRINT

Change: 260 REM
270 YA=INT(RND(0)*11)
280 YB=INT(RND(0)*11)
290 MA=INT(RND(0)*11)
300 MB=INT(RND(0)*11)
450 PRINT "BEEP"
750 MA=MA-1+INT(RND(0)*3)
760 MB=MB-1+INT(RND(0)*3)

COMMODORE VIC-20

```
Add: 905 Q=Q+1
      915 PRINT
      945 PRINT
      975 PRINT
      1035 Q=Q+1
```

```
Change: 230 PRINT "HOW OLD DO YOU THINK THE MONSTER IS"
        260 REM
        270 YA=INT(RND(0)*11)
        280 YB=INT(RND(0)*11)
        290 MA=INT(RND(0)*11)
        300 MB=INT(RND(0)*11)
        450 PRINT "BEEP"
        510 PRINT "YOUR MOVE"
        750 MA=MA-1+INT(RND(0)*3)
        760 MB=MB-1+INT(RND(0)*3)
        890 FOR Q=1 TO 12 STEP 3
        900 PRINT TAB(Q+2);"00";
        920 FOR Q=1 TO 12 STEP 4
        930 PRINT TAB(Q+1);"00";
        950 FOR Q=1 TO 12 STEP 4
        960 PRINT TAB(Q+1);"0 0";
        980 FOR Q=1 TO 12 STEP 4
        990 PRINT TAB(Q+1);"0 0";
        1000 IF Q>8 THEN PRINT " "; "0=<"
        1020 FOR Q=1 TO 12 STEP 3
        1030 PRINT TAB(Q+4);"00";
```

TEXAS INSTRUMENTS 99/4A

```
Add: 112 PRINT "LOCH NESS MONSTER?"
      192 PRINT "MILE EACH TURN."
      532 YA=YA-1
      542 YA=YA+1
      552 YB=YB-1
      562 YB=YB+1
      572 YA=YA-1
      582 YB=YB-1
      592 YA=YA-1
      602 YB=YB+1
      612 YA=YA+1
      622 YB=YB-1
      632 YA=YA+1
      642 YB=YB+1
      672 PRINT "YOU'RE AT THE SOUTH SHORE!"
      682 PRINT "YOU'RE AT THE NORTH SHORE!"
      692 PRINT "YOU'RE AT THE EAST SHORE!"
      702 PRINT "YOU'RE AT THE WEST SHORE!"
      712 YB=10
      722 YB=0
      732 YA=0
      742 YA=10
```

(continued)

```

772 MA=10
782 MA=0
792 MB=10
802 MB=0
862 PRINT "GREAT WORK, EXPLORER!"
872 PRINT "UNFORTUNATELY, HE FOUND YOU FIRST!"
1002 PRINT " "; "0:<"

```

```

Change: 110 PRINT "CAN YOU FIND THE"
160 PRINT "IT WILL BEEP TWICE IF YOU ARE TWO MILES"
190 PRINT "YOUR SUB WILL MOVE ONE"
200 PRINT "SO WILL THE MONSTER"
210 PRINT "(UNLESS IT DOESN'T MOVE AT ALL)."
230 PRINT "HOW OLD DO YOU THINK THE MONSTER IS";
260 RANDOMIZE
270 YA=INT(RND*11)
280 YB=INT(RND*11)
290 MA=INT(RND*11)
300 MB=INT(RND*11)
340 IF (YA=MA) * (YB=MB) THEN 830
350 IF (ABS(YA-MA)<4) * (ABS(YB-MB)<4) THEN 400
370 PRINT "MONSTER OUT OF SONAR RANGE . ."
450 CALL SOUND (100,440,2)
530 IF MOVE$<>"N" THEN 540
540 IF MOVE$<>"S" THEN 550
550 IF MOVE$<>"W" THEN 560
560 IF MOVE$<>"E" THEN 570
570 IF MOVE$<>"NW" THEN 580
580 IF MOVE$<>"NW" THEN 590
590 IF MOVE$<>"NE" THEN 600
600 IF MOVE$<>"NE" THEN 610
610 IF MOVE$<>"SW" THEN 620
620 IF MOVE$<>"SW" THEN 630
630 IF MOVE$<>"SE" THEN 640
640 IF MOVE$<>"SE" THEN 650
650 IF (YA<11) * (YB<11) * (YA>=0) * (YB>=0) THEN 750
670 IF YA<=10 THEN 680
680 IF YA>=0 THEN 690
690 IF YB<=10 THEN 700
700 IF YB>=0 THEN 710
710 IF YB<=10 THEN 720
720 IF YB>=1 THEN 730
730 IF YA>=0 THEN 740
740 IF YA<=10 THEN 750
750 MA=MA-1+INT(RND*3)
760 MB=MB-1+INT(RND*3)
770 IF MA<=10 THEN 780
780 IF MA>=0 THEN 790
790 IF MB<=10 THEN 800
800 IF MB>=0 THEN 810
860 IF COUNT>=10 THEN 870
870 IF COUNT<=25 THEN 880
890 FOR Q=1 TO 21 STEP 7
920 FOR Q=1 TO 21 STEP 7
950 FOR Q=1 TO 15 STEP 7
980 FOR Q=1 TO 15 STEP 7

```

```
1000 IF Q<=14 THEN 1010
1020 FOR Q=1 TO 15 STEP 7
1030 PRINT TAB(Q+8);"0";
```

TRS-80 COLOR COMPUTER

```
Add: 175 PRINT "TWO MILES AWAY. AND IT WILL"
      915 PRINT
      945 PRINT
      975 PRINT
```

```
Change: 110 PRINT "CAN YOU FIND THE LOCH NESS MON-"
        120 PRINT "STER? LOCH NESS IS IN SCOTLAND."
        130 PRINT "IT'S TEN MILES WIDE AND TEN"
        140 PRINT "MILES LONG. YOUR SUBMARINE'S"
        150 PRINT "SONAR WILL BEEP ONCE IF YOU ARE"
        160 PRINT "THREE MILES FROM THE MONSTER."
        170 PRINT "IT WILL BEEP TWICE IF YOU ARE"
        180 PRINT "BEEP THREE TIMES IF YOU GET WITHIN
        ONE MILE.";
        190 PRINT " YOUR SUB WILL MOVE ONE MILE EACH
        TURN.";
        200 PRINT " SO WILL THE MONSTER (UNLESS IT
        DOESN'T ";
        260 REM
        270 YA=RND(11)
        280 YB=RND(11)
        290 MA=RND(11)
        300 MB=RND(11)
        310 PRINT "YOU ARE ";YA;" MILES SOUTH OF THE NORTH";
        320 PRINT "SHORE OF THE LOCH AND ";YB;" MILES EAST";
        330 PRINT "OF THE WEST SHORE.";
        450 SOUND 173,3
        750 MA=MA-1+RND(3)
        760 MB=MB-1+RND(3)
        890 FOR Q=1 TO 17 STEP 8
        920 FOR Q=1 TO 17 STEP 8
        950 FOR Q=1 TO 17 STEP 8
        980 FOR Q=1 TO 17 STEP 8
        1000 IF Q>16 THEN PRINT " "; "0:<"
        1020 FOR Q=1 TO 17 STEP 8
```

Monster Wall

Do frightening humans still have you climbing the walls? Well, you won't climb this one — and neither will they!

Sample Run

STILL HAVING TROUBLE WITH THOSE
PESKY HUMANS? PERHAPS YOU NEED
A STRONGER WALL! OUR MONSTER
CONSTRUCTION COMPANY GUARANTEES
SATISFACTION!

HIT THE ENTER KEY TO BEGIN BUILDING?

```
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
*           *           *           *
*           *           *           *
*           *           *           *
*****
```

Break

Program Listing

```
10 REM MONSTER WALL
100 PRINT "STILL HAVING TROUBLE WITH THOSE"
110 PRINT "PESKY HUMANS? PERHAPS YOU NEED"
120 PRINT "A STRONGER WALL! OUR MONSTER"
130 PRINT "CONSTRUCTION COMPANY GUARANTEES"
140 PRINT "SATISFACTION!"
150 PRINT
160 PRINT "HIT THE ENTER KEY TO BEGIN BUILDING";
170 INPUT X$
180 PRINT
190 FOR A=1 TO 39
200 PRINT "*";
210 NEXT A
220 PRINT
230 FOR C=1 TO 3
240 PRINT TAB(6);"*";TAB(16);"*";
250 PRINT TAB(26);"*";TAB(36);"*"
260 NEXT C
270 FOR A=1 TO 39
280 PRINT "*";
290 NEXT A
300 PRINT
310 FOR C=1 TO 3
320 PRINT "*";TAB(11);"*";
330 PRINT TAB(21);"*";TAB(31);"*"
340 NEXT C
350 GOTO 190
```

If You Have . . .

APPLE II

Change: 160 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";

ATARI

Add: 91 DIM X\$(1)

```
Change: 160 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";
190 FOR A=1 TO 37
240 R=6: GOSUB 360: PRINT "*";: R=9: GOSUB 360:
PRINT "*";
250 GOSUB 360: PRINT "*";: GOSUB 360: PRINT "*"
270 FOR A=1 TO 37
320 PRINT "*";: R=10: GOSUB 360: PRINT "*";
330 GOSUB 360: PRINT "*";:GOSUB 360: PRINT "*"
360 FOR P=1 TO R
370 PRINT " ";: NEXT P: RETURN
```

(continued)

COMMODORE 64

Change: 160 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";

COMMODORE VIC-20

Change: 160 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING"
190 FOR A=1 TO 21
240 PRINT TAB(3);"*";TAB(8);"*";
250 PRINT TAB(13);"*";TAB(18);"*"
270 FOR A=1 TO 21
320 PRINT "*";TAB(6);"*";
330 PRINT TAB(11);"*";TAB(16);"*"

TEXAS INSTRUMENTS 99/4A

Change: 100 PRINT "STILL HAVING TROUBLE WITH"
110 PRINT "THOSE PESKY HUMANS? PERHAPS"
120 PRINT "YOU NEED A STRONGER WALL! OUR
 MONSTER CONSTRUCTION"
130 PRINT "COMPANY GUARANTEES"
160 PRINT "HIT THE ENTER KEY TO BEGIN BUILDING";
190 FOR A=1 TO 28
240 PRINT TAB(6);"*";TAB(13);"*";
250 PRINT TAB(20);"*";TAB(27);"*"
270 FOR A=1 TO 28
330 PRINT TAB(18);"*";TAB(31);"*"

TRS-80 COLOR COMPUTER

Change: 190 FOR A=1 TO 31
240 PRINT TAB(5);"*";TAB(15);"*";
250 PRINT TAB(26);"*"
270 FOR A=1 TO 31
320 PRINT "*";TAB(10);"*";
330 PRINT TAB(20);"*";TAB(30);"*"

Scramblanian Monster

The many-headed creatures of Scramblania speak a language a lot like English, but all those heads get them mixed up. To drive them away, all you have to do is unscramble their language and discover an English word — but that isn't as easy as it sounds!

Sample Run

THE CREATURE FROM SCRAMBLANIA
IS COMING TO GET YOU!

THE ONLY WAY TO ESCAPE
IS TO UNSCRAMBLE THE SECRET
SCRAMBLANIAN WORD THAT WILL KEEP
IT AWAY!

HOW MANY HEADS DOES THE CREATURE HAVE? 25

THE SCRAMBLANIAN WORD IS:
RDIWREO

WHAT IS THE WORD IN ENGLISH? WEIRDER

NO! THAT'S NOT IT!
THE MONSTER'S COMING CLOSER!

A TELEGRAM FROM YOUR FRIENDS
IN SCRAMBLANIA:
THE FIRST LETTER OF THE WORD IS W

THE SCRAMBLANIAN WORD IS:
RDIWREO

WHAT IS THE WORD IN ENGLISH? DORRIE

NO! THAT'S NOT IT!
THE SCRAMBLANIAN MONSTER
CAN SMELL ITS DINNER!

A TELEGRAM FROM YOUR FRIENDS
IN SCRAMBLANIA:
THE FIRST 2
LETTERS OF THE WORD ARE WO

THE SCRAMBLANIAN WORD IS:
RDIWREO

(continued)

WHAT IS THE WORD IN ENGLISH? WORRIED
YOU GOT IT!
THE MONSTER GOES HOME TO SCRAMBLANIA!

OH, NO!
HERE COMES ANOTHER SCRAMBLANIAN MONSTER!
WANT TO KEEP IT AWAY? NO

□ Program Listing

```
10 REM SCRAMBLANIAN MONSTER
100 PRINT
110 PRINT "THE CREATURE FROM SCRAMBLANIA"
120 PRINT "IS COMING TO GET YOU!"
130 PRINT
140 C=0
150 PRINT "THE ONLY WAY TO ESCAPE"
160 PRINT "IS TO UNSCRAMBLE THE SECRET"
170 PRINT "SCRAMBLANIAN WORD THAT WILL KEEP"
180 PRINT "IT AWAY!"
190 PRINT
200 PRINT "HOW MANY HEADS DOES THE CREATURE HAVE";
210 INPUT H
220 RANDOMIZE H
230 FOR I=1 TO 7
240 READ N(I)
250 NEXT I
260 FOR I=1 TO 2+INT(RND(1)*18)
270 READ W$
280 NEXT I
290 FOR I=1 TO 7
300 A$=A$+MID$(W$,N(I),1)
310 NEXT I
320 PRINT
330 PRINT "THE SCRAMBLANIAN WORD IS:"
340 PRINT W$
350 PRINT
360 PRINT "WHAT IS THE WORD IN ENGLISH";
370 INPUT P$
380 IF P$="" THEN 360
390 IF P$=A$ THEN 660
400 PRINT
410 PRINT "NO! THAT'S NOT IT!"
420 C=C+1
430 IF C=1 THEN PRINT "THE MONSTER'S COMING CLOSER!"
440 IF C=2 THEN PRINT "THE SCRAMBLANIAN MONSTER"
450 IF C=2 THEN PRINT "CAN SMELL ITS DINNER!"
460 IF C=3 THEN PRINT "THERE'S NO ESCAPE"
470 IF C=3 THEN PRINT "IF YOU DON'T GET IT RIGHT THIS TIME!"
480 IF C=3 THEN PRINT "QUICK! UNSCRAMBLE THE WORD!"
490 IF C=4 THEN 610
500 PRINT
```

```

510 PRINT "A TELEGRAM FROM YOUR FRIENDS"
520 PRINT "IN SCRAMBLANIA:"
530 IF C<>1 THEN 560
540 PRINT "THE FIRST LETTER OF THE WORD IS ";
550 GOTO 580
560 PRINT "THE FIRST ";C
570 PRINT "LETTERS OF THE WORD ARE ";
580 PRINT MID$(A$,1,C)
590 PRINT
600 GOTO 330
610 PRINT
620 PRINT "THE SCRAMBLANIAN MONSTER MADE YOU"
630 PRINT "EAT HIS WORD. THE WORD WAS"
640 PRINT A$;". "
650 GOTO 680
660 PRINT "YOU GOT IT!"
670 PRINT "THE MONSTER GOES HOME TO SCRAMBLANIA!"
680 PRINT
690 PRINT "OH, NO!"
700 PRINT "HERE COMES ANOTHER SCRAMBLANIAN MONSTER!"
710 PRINT "WANT TO KEEP IT AWAY";
720 INPUT Y$
730 IF MID$(Y$,1,1)="N" THEN END
740 RESTORE
750 A$=""
760 C=0
770 PRINT
780 GOTO 230
790 DATA 4,7,1,5,3,6,2
800 DATA BRTLSEO,ASODGNR,AAUDCLR,CSES RTE,WEENHRO
810 DATA RSOMRRI,ISHFGTR,RDIWREO,RSOHRRO,IRISMEL
820 DATA TEIOSDU,FYLAULW,VSRCENA,MSIMMEU,STUDGSI
830 DATA ONMSWAN,LEAVLGI,MEIVPRA,CSODTRO,SEAMSGE

```

If You Have . . .

APPLE II

Delete: 220

ATARI

Delete: 220, 310

Add: 91 DIM A\$(7),W\$(7),Y\$(3),P\$(15),N(7)
241 N(I)=X

Change: 240 READ X
260 FOR I=1 TO 2+INT(RND(0)*18)
300 A\$(I)=W\$(N(I),N(I))
580 PRINT A\$(1,C)
730 IF Y\$(1,1)="N" THEN END

(continued)

COMMODORE 64

Delete: 220

Change: 260 FOR I=1 TO 2+INT(RND(0)*18)

COMMODORE VIC-20

Delete: 220

Change: 200 PRINT "HOW MANY HEADS DOES THE CREATURE HAVE"
260 FOR I=1 TO 2+INT(RND(0)*18)
360 PRINT "WHAT IS THE WORD IN ENGLISH"
710 PRINT "WANT TO KEEP IT AWAY"

TEXAS INSTRUMENTS 99/4A

Add: 112 PRINT "SCRAMBLANIA"
202 PRINT "CREATURE HAVE"
432 PRINT "THE MONSTER'S COMING CLOSER!"
442 PRINT "THE SCRAMBLANIAN MONSTER"
452 PRINT "CAN SMELL ITS DINNER!"
462 PRINT "THERE'S NO ESCAPE"
472 PRINT "IF YOU DON'T GET IT RIGHT"
474 PRINT "THIS TIME!"
482 PRINT "QUICK! UNSCRAMBLE THE WORD!"
542 PRINT "THE WORD IS "
702 PRINT "SCRAMBLANIAN MONSTER"
840 END

Change: 110 PRINT "THE CREATURE FROM"
170 PRINT "SCRAMBLANIAN WORD THAT WILL"
180 PRINT "KEEP IT AWAY!"
200 PRINT "HOW MANY HEADS DOES THE"
220 RANDOMIZE
260 FOR I=1 TO 2+INT(RND*18)
300 A\$=A\$ & SEG\$(W\$,N(I),1)
430 IF C<>1 THEN 440
440 IF C<>2 THEN 450
450 IF C<>2 THEN 460
460 IF C<>3 THEN 470
470 IF C<>3 THEN 480
480 IF C<>3 THEN 490
540 PRINT "THE FIRST LETTER OF"
580 PRINT SEG\$(A\$,1,C)
670 PRINT "THE MONSTER GOES HOME TO SCRAMBLANIA!"
700 PRINT "HERE COMES ANOTHER"
730 IF SEG\$(Y\$,1,1)="N" THEN 840

TRS-80 COLOR COMPUTER

Change: 220 REM
260 FOR I=1 TO 2+RND(18)

Monster Memory

Terrible monsters are coming to get you. There's just one thing they can't stand — humans who remember exactly what they look like. If you've got a monstrous memory, you'll survive. If not . . . well, don't say we didn't warn you!

Sample Run

HERE COMES ONE OF THE WORST
MONSTERS OF ALL TIME.
YOU'D BETTER REMEMBER EXACTLY
WHAT HE LOOKS LIKE OR YOU
WILL BE EATEN!

THE HIGHER THE NUMBER,
THE HARDER THE GAME--
ENTER A NUMBER FROM 1 TO 5? 2

THE MONSTER HAS 45 HEADS.
PRESS THE ENTER KEY WHEN READY?

HOW MANY HEADS DOES THE MONSTER HAVE? 45
RIGHT!
YOU SURVIVED--FOR NOW!

THE MONSTER HAS 5 LEGS.
PRESS THE ENTER KEY WHEN READY?

HOW MANY LEGS DOES THE MONSTER HAVE? 5
RIGHT!
YOU SURVIVED--FOR NOW!

THE MONSTER HAS 27 ARMS.
PRESS THE ENTER KEY WHEN READY?

HOW MANY ARMS DOES THE MONSTER HAVE? 27
RIGHT!
YOU SURVIVED--FOR NOW!

THE MONSTER HAS 2 EYES.
PRESS THE ENTER KEY WHEN READY?

HOW MANY EYES DOES THE MONSTER HAVE? 2
RIGHT!
YOU SURVIVED--FOR NOW!

(continued)

[There's more—when you get to the end of the program, you'll see something like this:]

HOW MANY ARMS DOES THE MONSTER HAVE? 23
NO! THE MONSTER HAS 27 ARMS--
AND THEY ALL JUST ATTACKED YOU!
TOO BAD . . .

□ Program Listing

```
100 REM MONSTER MEMORY
110 DIM N(10)
120 PRINT "HERE COMES ONE OF THE WORST"
130 PRINT "MONSTERS OF ALL TIME."
140 PRINT "YOU'D BETTER REMEMBER EXACTLY"
150 PRINT "WHAT HE LOOKS LIKE OR YOU"
160 PRINT "WILL BE EATEN!"
170 PRINT
180 PRINT "THE HIGHER THE NUMBER,"
190 PRINT "THE HARDER THE GAME--"
200 PRINT "ENTER A NUMBER FROM 1 TO 5";
210 INPUT N
220 IF N<1 OR N>5 THEN 180
230 FOR I=1 TO N+5
240 N(I)=2+INT(RND(1)*50)
250 NEXT I
260 FOR I=1 TO N+5
270 FOR K=1 TO I
280 READ P$
290 NEXT K
300 RESTORE
310 PRINT "THE MONSTER HAS ";N(I); " "; P$; "."
320 PRINT "PRESS THE ENTER KEY WHEN READY";
330 INPUT R$
340 GOSUB 590
350 FOR J=I TO 1 STEP -1
360 FOR K=1 TO J
370 READ P$
380 NEXT K
390 RESTORE
400 PRINT "HOW MANY ";P$; " DOES THE MONSTER HAVE";
410 INPUT A
420 IF A=0 THEN 400
430 IF A<>N(J) THEN 550
440 PRINT "RIGHT!"
450 PRINT "YOU SURVIVED--FOR NOW!"
460 PRINT
470 NEXT J
480 PRINT
```

```

490 NEXT I
500 PRINT CHR$(7)
510 PRINT "YOU DID IT!"
520 PRINT "YOU'VE GOT A REALLY"
530 PRINT "MONSTROUS MEMORY!"
540 END
550 PRINT "NO! THE MONSTER HAS ";N(J);" ";P$;"--"
560 PRINT "AND THEY ALL JUST ATTACKED YOU!"
570 PRINT "TOO BAD . . ."
580 END
590 FOR K=1 TO 25
600 PRINT
610 NEXT K
620 RETURN
630 DATA HEADS,LEGS,ARMS,EYES,NOSES
640 DATA EARS,FINGERS,FANGS,LIPS,CLAWS

```

If You Have . . .

APPLE II

Change: 320 PRINT "PRESS THE RETURN KEY WHEN READY";

ATARI

Change: 110 DIM N(10),P\$(7),R\$(1)
240 N(I)=2+INT(RND(0)*50)
320 PRINT "PRESS THE RETURN KEY WHEN READY";
500 PRINT CHR\$(253)

COMMODORE 64

Change: 320 PRINT "PRESS THE RETURN KEY WHEN READY";

COMMODORE VIC-20

Change: 200 PRINT "ENTER A NUMBER FROM 1 TO 5"
320 PRINT "PRESS THE RETURN KEY WHEN READY"
400 PRINT "HOW MANY ";P\$; " DOES THE MONSTER HAVE"

TEXAS INSTRUMENTS 99/4A

Add: 142 PRINT "REMEMBER EXACTLY"
402 PRINT "THE MONSTER HAVE";

Change: 140 PRINT "YOU'D BETTER"
210 INPUT NN

(continued)

```
220 IF (NN<1)+(NN>5) THEN 180
230 FOR I=1 TO NN+5
240 N(I)=2+INT(RND*50)
260 FOR I=1 TO NN+5
320 PRINT "PRESS THE ENTER KEY WHEN READY ";
400 PRINT "HOW MANY ";P$; " DOES "
500 SOUND (100,440,2)
```

TRS-80 COLOR COMPUTER

Change: 240 N(I)=2+RND(50)
500 SOUND 173,3

Crafty Creature

The world's smartest monster challenges you to a simple game.
The creature can be beaten — but only if you figure out its secret!

Sample Run

I AM THE WORLD'S SMARTEST MONSTER.
I CAN BEAT YOU AT THIS GAME.
THE RULES ARE SIMPLE.
WE START AT ZERO.
THEN WE TAKE TURNS
ADDING A NUMBER FROM 1 TO 8.
FIRST ONE TO REACH 94 OR OVER WINS.

DO YOU WANT TO GO FIRST? YES

YOUR TURN:
HOW MUCH DO YOU WANT TO ADD? 8
THAT MAKES THE TOTAL 8

I'LL ADD 5
THAT MAKES THE TOTAL 13

YOUR TURN:
HOW MUCH DO YOU WANT TO ADD? 4
THAT MAKES THE TOTAL 17

I'LL ADD 5
THAT MAKES THE TOTAL 22

YOUR TURN:
HOW MUCH DO YOU WANT TO ADD? 9
THAT'S NOT A NUMBER FROM 1 TO 8!
HOW MUCH DO YOU WANT TO ADD? 3
THAT MAKES THE TOTAL 25

I'LL ADD 6
THAT MAKES THE TOTAL 31

YOUR TURN:
HOW MUCH DO YOU WANT TO ADD? 1
THAT MAKES THE TOTAL 32

[There's more—when you get to the end of the program, you'll see something like this:]

I'LL ADD 6
THAT MAKES THE TOTAL 94

(continued)

I WIN!
I TOLD YOU I COULD BEAT YOU!

PLAY AGAIN? NO!

□ Program Listing

```
10 REM CRAFTY CREATURE
100 PRINT
110 PRINT "I AM THE WORLD'S SMARTEST MONSTER."
120 PRINT "I CAN BEAT YOU AT THIS GAME."
130 PRINT "THE RULES ARE SIMPLE."
140 ALL=0
150 T=50+INT(RND(1)*51)
160 LOW=1+INT(RND(1)*3)
170 HIGH=5+INT(RND(1)*6)
180 PRINT "WE START AT ZERO."
190 PRINT "THEN WE TAKE TURNS"
200 PRINT "ADDING A NUMBER FROM ";LOW;" TO ";HIGH;"."
210 PRINT "FIRST ONE TO REACH ";T;" OR OVER WINS."
220 PRINT
230 PRINT "DO YOU WANT TO GO FIRST";
240 INPUT ANS$
250 IF ANS$="" THEN 230
260 IF MID$(ANS$,1,1)="Y" THEN 320
270 GOSUB 420
280 PRINT
290 PRINT "I'LL ADD ";MOVE
300 GOSUB 540
310 IF ALL=T THEN 570
320 PRINT
330 PRINT "YOUR TURN:"
340 PRINT "HOW MUCH DO YOU WANT TO ADD";
350 INPUT MOVE
360 IF MOVE>LOW-1 AND MOVE<HIGH+1 THEN 390
370 PRINT "THAT'S NOT A NUMBER FROM ";LOW;" TO ";HIGH;"!"
380 GOTO 340
390 GOSUB 540
400 IF ALL=T THEN 600
410 GOTO 270
420 IF T-ALL>HIGH THEN 450
430 MOVE=T-ALL
440 GOTO 530
450 MY=T
460 FOR I=T TO ALL+LOW+HIGH STEP -(LOW+HIGH)
470 MY=MY-LOW-HIGH
480 NEXT I
490 MOVE=ALL-MY
500 IF MOVE<0 THEN MOVE=ABS(MOVE)
510 IF MOVE>HIGH THEN MOVE=HIGH
520 IF MOVE<LOW THEN MOVE=LOW
```

```

530 RETURN
540 ALL=ALL+MOVE
550 PRINT "THAT MAKES THE TOTAL ";ALL
560 RETURN
570 PRINT "I WIN!"
580 PRINT "I TOLD YOU I COULD BEAT YOU!"
590 GOTO 620
600 PRINT "YOU WIN!"
610 PRINT "PHOOEY!"
620 PRINT
630 PRINT "PLAY AGAIN";
640 INPUT Y$
650 PRINT
660 IF MID$(Y$,1,1)<>"N" THEN 110
670 END

```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM ANS\$(10),Y\$(1)

Change: 150 T=50+INT(RND(0)*51)
160 LOW=1+INT(RND(0)*3)
170 HIGH=5+INT(RND(0)*6)
260 IF ANS\$(1,1)="Y" THEN 320
660 IF Y\$(1,1)<>"N" THEN 110

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 230 PRINT "DO YOU WANT TO GO FIRST"
340 PRINT "HOW MUCH DO YOU WANT TO ADD:"
630 PRINT "PLAY AGAIN?"

TEXAS INSTRUMENTS 99/4A

Add: 512 MOVE=HIGH
522 MOVE=LOW

Change: 110 PRINT "I AM THE WORLD'S SMARTEST MONSTER."
150 T=50+INT(RND*51)
160 LOW=1+INT(RND*3)

(continued)

```
170 HIGH=5+INT(RND*6)
260 IF SEG$(ANS$,1,1)="Y" THEN 320
360 IF (MOVE>LOW-1)*(MOVE<HIGH+1) THEN 390
500 MOVE=ABS(MOVE)
510 IF MOVE<=HIGH THEN 520
520 IF MOVE>=LOW THEN 530
660 IF SEG$(Y$,1,1)<>"N" THEN 110
```

TRS-80 COLOR COMPUTER

```
Change: 150 T=50+RND(51)
160 LOW=1+RND(3)
170 HIGH=5+RND(6)
```

Creature Features

The dank, dark swamp is full of horrible creatures, and every one is different. They only come out when you type in this program — so beware!

Sample Run

WE'RE TRAPPED IN THIS AWFUL SWAMPY
PLACE. THEY SAY THERE ARE SCARY
MONSTERS HERE . . .

WHAT'S THAT SOUND?

LOOK OUT! THE MONSTER'S COMING!
I CAN SEE IT NOW!
IT'S HORRIBLE! IT HAS

A SICKENING DECAYING NOSE,
A HIDEOUS HUGE NECK,
A GRUESOME DEFORMED CHEEK,
A REPULSIVE SCUMMY BELLY,
A GHASTLY BANDAGED PAIR OF FEET,
A ROTTEN SCALY SET OF CLAWS,
A PUTRID GNARLED RIGHT ARM,
A GREEN GRIMY MOUTH,
A STINKY OOZING STUMP,
A FOUL WORMY WART,
A DISGUSTING TWISTED TORSO,
A SLIMY LONG FOREHEAD,
A NAUSEATING HAIRY CHIN,
A REVOLTING BROKEN LEFT ARM
AND
A VILE SCABBY SCAR.

OH NO! HERE COMES ANOTHER MONSTER...
WANT TO KNOW WHAT IT LOOKS LIKE? OKAY

LOOK OUT! THE MONSTER'S COMING!
I CAN SEE IT NOW!
IT'S HORRIBLE! IT HAS

A REPULSIVE DEFORMED NECK,
A REVOLTING GRIMY SCAR,
A GHASTLY GNARLED STUMP,
A ROTTEN DECAYING WART,
A NAUSEATING TWISTED SET OF CLAWS,
A SICKENING HUGE CHEEK,
A VILE BANDAGED RIGHT ARM,
A PUTRID SCUMMY TORSO,
A HIDEOUS OOZING PAIR OF FEET,

(continued)

A GRUESOME SCABBY MOUTH,
A STINKY SCALY CHIN,
A GREEN LONG LEFT ARM,
A FOUL HAIRY BELLY,
A SLIMY BROKEN FOREHEAD
AND
A DISGUSTING WORMY NOSE.

OH NO! HERE COMES ANOTHER MONSTER...
WANT TO KNOW WHAT IT LOOKS LIKE? NO!

Program Listing

```
10 REM CREATURE FEATURES
100 DIM Q(15),R(15),S(15)
110 PRINT "WE'RE TRAPPED IN THIS AWFUL SWAMPY"
120 PRINT "PLACE. THEY SAY THERE ARE SCARY"
130 PRINT "MONSTERS HERE . . ."
140 PRINT
150 PRINT CHR$(7)
160 PRINT "WHAT'S THAT SOUND?"
170 PRINT
180 PRINT "LOOK OUT! THE MONSTER'S COMING!"
190 PRINT "I CAN SEE IT NOW!"
200 PRINT "IT'S HORRIBLE! IT HAS"
210 PRINT
220 FOR A=1 TO 15
230 B=INT(RND(1)*15)+1
240 Q(A)=B
250 FOR J=1 TO A-1
260 IF B=Q(J) THEN 230
270 NEXT J
280 C=INT(RND(1)*15)+1
290 R(A)=C
300 FOR J=1 TO A-1
310 IF C=R(J) THEN 280
320 NEXT J
330 D=INT(RND(1)*15)+1
340 S(A)=D
350 FOR J=1 TO A-1
360 IF D=S(J) THEN 330
370 NEXT J
380 PRINT TAB(3);"A ";
390 FOR E=1 TO B
400 READ X$
410 NEXT E
420 PRINT X$;CHR$(32);
430 RESTORE
440 FOR F=1 TO 15+C
450 READ Y$
460 NEXT F
```

```

470 PRINT Y$;CHR$(32);
480 RESTORE
490 FOR G=1 TO 30+D
500 READ Z$
510 NEXT G
520 PRINT Z$;
530 IF A<14 THEN PRINT ", "
540 RESTORE
550 IF A=14 THEN PRINT TAB(12);"AND"
560 IF A=15 THEN PRINT "."
570 NEXT A
580 PRINT
590 PRINT "OH NO!  HERE COMES ANOTHER MONSTER..."
600 PRINT "WANT TO KNOW WHAT IT LOOKS LIKE";
610 INPUT N$
620 IF MID$(N$,1,1)<>"N" THEN 170
630 DATA HIDEOUS,SLIMY,DISGUSTING,STINKY,PUTRID
640 DATA REVOLTING,REPULSIVE,NAUSEATING,ROTTEN
650 DATA GHASTLY,GRUESOME,VILE,FOUL,SICKENING
660 DATA GREEN,HUGE,LONG,TWISTED,OOZING,GNARLED
670 DATA BROKEN,SCUMMY,HAIRY,SCALY,BANDAGED
680 DATA DEFORMED,SCABBY,WORMY,DECAYING,GRIMY
690 DATA FOREHEAD,NOSE,MOUTH,TORSO,NECK,SET OF CLAWS
700 DATA LEFT ARM,RIGHT ARM,PAIR OF FEET,BELLY
710 DATA WART,CHEEK,CHIN,SCAR,STUMP

```

If You Have . . .

APPLE II

```

Add: 245 IF A<2 THEN 280
      295 IF A<2 THEN 330
      345 IF A<2 THEN 380

```

ATARI

```

Add: 101 DIM X$(11), Y$(11), Z$(11), N$(11)
      245 IF A<2 THEN 280
      295 IF A<2 THEN 330
      345 IF A<2 THEN 380

```

```

Change: 150 PRINT CHR$(253)
        230 B=INT(RND(0)*15)+1
        280 C=INT(RND(0)*15)+1
        330 D=INT(RND(0)*15)+1
        380 PRINT "  A ";
        550 IF A=14 THEN PRINT "          AND"
        620 IF N$(1,1)<>"N" THEN 170

```

(continued)

COMMODORE 64

```
Add: 245 IF A<2 THEN 280
      295 IF A<2 THEN 330
      345 IF A<2 THEN 380
```

```
Change: 230 B=INT(RND(0)*15)+1
        280 C=INT(RND(0)*15)+1
        330 D=INT(RND(0)*15)+1
        550 IF A=14 THEN PRINT TAB(12); " AND"
```

COMMODORE VIC-20

```
Add: 245 IF A<2 THEN 280
      295 IF A<2 THEN 330
      345 IF A<2 THEN 380
```

```
Change: 230 B=INT(RND(0)*15)+1
        280 C=INT(RND(0)*15)+1
        330 D=INT(RND(0)*15)+1
        550 IF A=14 THEN PRINT TAB(12); " AND"
        600 PRINT "WANT TO KNOW WHAT IT LOOKS LIKE"
```

TEXAS INSTRUMENTS 99/4A

```
Add: 535 PRINT TAB(12); "AND"
      565 PRINT "."
```

```
Change: 150 SOUND(100,174,2)
        230 B=INT(RND*15)+1
        280 C=INT(RND*15)+1
        330 D=INT(RND*15)+1
        530 IF A>=14 THEN 540
        550 IF A<>14 THEN 560
        560 IF A<>15 THEN 570
        620 IF SEG$(N$,1,1)<>"N" THEN 170
```

TRS-80 COLOR COMPUTER

```
Add: 245 IF A<2 THEN 280
      295 IF A<2 THEN 330
      345 IF A<2 THEN 380
```

```
Change: 150 SOUND 153,15
        220 FOR A=2 TO 15
        230 B=RND(15)
        280 C=RND(15)
        330 D=RND(15)
```

Meet the Mummy

The mummy has been dead a very, very long time. With this program, you'll be able to find out exactly how long it's been since the mummy died. And if you're very clever, you may be able to change this program to discover exactly how old you are. Try it!

Sample Run

HOW OLD IS YOUR MUMMY?
I'LL BE HAPPY TO TELL YOU!

WHAT YEAR DID THE MUMMY DIE? 1492
A.D. OR B.C.? A.D.
WHAT MONTH DID THE MUMMY DIE (1-12)? 10
AND WHAT DAY OF THE MONTH
DID THE MUMMY DIE (1- 31)? 15

NOW LET'S MOVE ON TO THE PRESENT DAY!
WHAT YEAR IS IT NOW? 1984
WHAT MONTH IS IT (1-12)? 9
WHAT DATE IS IT (1- 30)? 3

THE MUMMY DIED 491 YEARS AGO.
THAT WAS A MERE 179646 DAYS AGO.

Program Listing

```
10 REM MEET THE MUMMY
100 DIM L(12)
110 FOR I=1 TO 12
120 READ L(I)
130 NEXT I
140 PRINT "HOW OLD IS YOUR MUMMY?"
150 PRINT "I'LL BE HAPPY TO TELL YOU!"
160 PRINT
170 PRINT "WHAT YEAR DID THE MUMMY DIE";
180 INPUT MY
190 IF MY=0 THEN 170
200 PRINT "A.D. OR B.C.";
210 INPUT AD$
220 IF MID$(AD$,1,1)="A" THEN 260
230 IF MID$(AD$,1,1)="B" THEN 260
240 PRINT "TRY AGAIN:"
250 GOTO 200
260 IF MID$(AD$,1,1)="B" THEN Z=1
270 PRINT "WHAT MONTH DID THE MUMMY DIE";
```

(continued)


```

280 PRINT " (1-12)";
290 INPUT MM
300 IF MM>0 AND MM<13 THEN 340
310 PRINT "TRY AGAIN:"
320 GOTO 270
330 PRINT
340 PRINT "AND WHAT DAY OF THE MONTH"
350 PRINT "DID THE MUMMY DIE (1-";L(MM);")";
360 INPUT MD
370 IF MD<L(MM)+1 AND MD>0 THEN 400
380 PRINT "TRY AGAIN:"
390 GOTO 340
400 PRINT
410 PRINT "NOW LET'S MOVE ON TO THE PRESENT DAY!"
420 PRINT "WHAT YEAR IS IT NOW";
430 INPUT YN
440 IF YN<1984 THEN 420
450 PRINT "WHAT MONTH IS IT (1-12)";
460 INPUT MN
470 IF MN>0 AND MN<13 THEN 500
480 PRINT "TRY AGAIN:"
490 GOTO 450
500 PRINT "WHAT DATE IS IT (1-";L(MN);")";
510 INPUT DN
520 IF DN<L(MN)+1 AND DN>0 THEN 550
530 PRINT "TRY AGAIN:"
540 GOTO 500
550 LN=INT(YN/4)-3*(INT(YN/100))
560 TN=(366*LN)+(365*(YN-LN))
570 FOR I=1 TO MN-1
580 TD=TD+L(I)
590 NEXT I
600 AN=TN+TD+DN
610 ML=INT(MY/4)-3*(INT(MY/100))
620 TM=(366*ML)+(365*(MY-ML))
630 IF Z=1 THEN 690
640 FOR I=1 TO MM-1
650 FM=FM+L(I)
660 NEXT I
670 AM=TM+FM+MD
680 GOTO 730
690 FOR I=MM+1 TO 12
700 FM=FM+L(I)
710 NEXT I
720 AM=TM+FM+L(MM)-MD
730 IF FM+MD>TD+DN THEN 770
740 IF Z=0 THEN FY=YN-MY
750 IF Z=1 THEN FY=YN+MY-1
760 GOTO 790
770 IF Z=0 THEN FY=YN-MY-1
780 IF Z=1 THEN FY=YN+MY
790 IF Z=0 THEN FD=AN-AM
800 IF Z=1 THEN FD=AN+AM
810 PRINT
820 PRINT "THE MUMMY DIED ";FY; " YEARS AGO."

```

```
830 PRINT "THAT WAS A MERE ";FD; " DAYS AGO."  
840 DATA 31,28,31,30,31,30,31,31,30,31,30,31
```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 121 L(I)=TP

Change: 100 DIM L(12), AD\$(2)
120 READ TP
220 IF AD\$(1,1)="A" THEN 260
230 IF AD\$(1,1)="B" THEN 260
260 IF AD\$(1,1)="B" THEN Z=1

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 170 PRINT "WHAT YEAR DID THE MUMMY DIE"
200 PRINT "A.D. OR B.C."
280 PRINT " (1-12) "
350 PRINT "DID THE MUMMY DIE (1-";L(MM);") "
420 PRINT "WHAT YEAR IS IT NOW"
450 PRINT "WHAT MONTH IS IT (1-12) "
500 PRINT "WHAT DATE IS IT (1-";L(MN);") "

TEXAS INSTRUMENTS 99/4A

Add: 265 Z=1
742 FY=YN-MY
752 FY=YN+MY-1
772 FY=YN-MY-1
782 FD=YN+MY
792 FD=AN-AM
802 FD=AN+AM

Change: 220 IF SEG\$(AD\$,1,1)="A" THEN 260
230 IF SEG\$(AD\$,1,1)="B" THEN 260
260 IF SEG\$(AD\$,1,1)<>"B" THEN 270
300 IF (MM>0)*(MM<13) THEN 340
370 IF (MD<L(MM)+1)*(MD>0) THEN 400
410 PRINT "NOW LET'S MOVE ON TO THE PRESENT DAY!"
470 IF (MN>0)*(MN<13) THEN 500
520 IF (DN<L(MN)+1)*(DN>0) THEN 550 (continued)

```
740 IF Z<>0 THEN 750
750 IF Z<>1 THEN 760
770 IF Z<>0 THEN 780
780 IF Z<>1 THEN 790
790 IF Z<>0 THEN 800
800 IF Z<>1 THEN 810
```

TRS-80 COLOR COMPUTER

No Changes Required

Shapeshifters

We hate to spoil your fun — but something in the room with your computer is really a monster in disguise. Is it your desk? Your pencil? This book? The only way you'll be able to find out is to type in this program!

Sample Run

HOW MANY MONSTERS DO YOU KNOW? 37

SOMETHING YOU CAN SEE IN THIS ROOM
HAS TURNED INTO A
!!!!!!!!!!!!!!!!MONSTER!!!!!!!!!!!!!!!!!!!!

TELL ME SOMETHING YOU CAN SEE
FROM WHERE YOU ARE? COMPUTER

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? DESK

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? CHAIR

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? BED

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? LAMP

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? TV

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? BOOK

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? DESK

YOU NAMED THAT ALREADY!

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? PENCIL

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? CUP

TELL ME A DIFFERENT THING YOU CAN
SEE FROM WHERE YOU ARE? SHIRT

(continued)

ONE OF THESE THINGS IS
!!!!!!A MONSTER!!!!!!

YOU HAVE 3 TURNS
TO FIGURE OUT WHICH:

- 1 COMPUTER
- 2 DESK
- 3 CHAIR
- 4 BED
- 5 LAMP
- 6 TV
- 7 BOOK
- 8 PENCIL
- 9 CUP
- 10 SHIRT

WHICH ONE IS IT--
CHOOSE A NUMBER? 2

NOPE!

ONE OF THESE THINGS IS
!!!!!!A MONSTER!!!!!!

YOU HAVE 2 TURNS
TO FIGURE OUT WHICH:

WHICH ONE IS IT--
CHOOSE A NUMBER? 6

NOPE!

ONE OF THESE THINGS IS
!!!!!!A MONSTER!!!!!!

YOU HAVE 1 TURN
TO FIGURE OUT WHICH:

WHICH ONE IS IT--
CHOOSE A NUMBER? 4

NOPE!

YOU GUESSED WRONG!
THE MONSTER IS THE CHAIR!
BETTER KEEP AN EYE ON IT!

□ Program Listing

```
10 REM SHAPESHIFTERS
15 REM THIS PROGRAM WILL NOT WORK ON ATARI COMPUTERS
100 PRINT "HOW MANY MONSTERS DO YOU KNOW";
110 INPUT R
120 RANDOMIZE R
130 PRINT
140 PRINT "SOMETHING YOU CAN SEE IN THIS ROOM"
150 PRINT "      HAS TURNED INTO A"
160 PRINT "!!!!!!!!!!!!!!MONSTER!!!!!!!!!!!!!!!"
170 PRINT
180 PRINT "TELL ME SOMETHING YOU CAN SEE"
190 PRINT "FROM WHERE YOU ARE";
200 INPUT A$(1)
210 IF A$(1)=" " THEN 180
220 FOR I=2 TO 10
230 PRINT
240 PRINT "TELL ME A DIFFERENT THING YOU CAN"
250 PRINT "SEE FROM WHERE YOU ARE";
260 INPUT A$(I)
270 IF A$(I)=" " THEN 240
280 FOR J=1 TO I-1
290 IF A$(J)<>A$(I) THEN 340
300 PRINT
310 PRINT "YOU NAMED THAT ALREADY!"
320 PRINT
330 GOTO 240
340 NEXT J
350 NEXT I
360 R=1+INT(RND(1)*10)
370 T=3+INT(RND(1)*7)
380 PRINT
390 PRINT "ONE OF THESE THINGS IS"
400 PRINT "!!!!!!A MONSTER!!!!!!"
410 PRINT
420 PRINT "YOU HAVE ";T;" TURNS"
430 PRINT "TO FIGURE OUT WHICH:"
440 PRINT
450 FOR I=1 TO 10
460 IF A$(I)<>"USED" THEN PRINT I;" ";A$(I)
470 NEXT I
480 PRINT
490 PRINT "WHICH ONE IS IT--"
500 PRINT "CHOOSE A NUMBER";
510 INPUT G
520 IF G<1 OR G>10 THEN 490
530 IF A$(G)="USED" THEN PRINT "YOU TRIED THAT ALREADY!"
540 IF A$(G)="USED" THEN 490
550 IF G<1 OR G>10 THEN 490
560 IF G=R THEN 690
570 A$(G)="USED"
580 T=T-1
590 PRINT
```

(continued)

```

600 PRINT CHR$(7)
610 PRINT "NOPE!"
620 PRINT
630 IF T<>0 THEN 390
640 PRINT
650 PRINT "YOU GUESSED WRONG!"
660 PRINT "THE MONSTER IS THE ";A$(R);"! "
670 PRINT "BETTER KEEP AN EYE ON IT!"
680 END
690 PRINT "YOU GUESSED IT!"
700 PRINT "THE MONSTER TURNED BACK INTO A PLAIN OLD"
710 PRINT A$(R);"! "

```

If You Have . . .

APPLE II

Delete: 120

Change: 700 PRINT "THE MONSTER TURNED BACK INTO A PLAIN OLD";

ATARI This program will not run in Atari BASIC

COMMODORE 64

Delete: 120

Change: 360 R=1+INT(RND(0)*10)
370 T=3+INT(RND(0)*7)

COMMODORE VIC-20

Delete: 120

Change: 100 PRINT "HOW MANY MONSTERS DO YOU KNOW"
190 PRINT "FROM WHERE YOU ARE"
250 PRINT "SEE FROM WHERE YOU ARE"
360 R=1+INT(RND(0)*10)
370 T=3+INT(RND(0)*7)
500 PRINT "CHOOSE A NUMBER"

TEXAS INSTRUMENTS 99/4A

Add: 462 PRINT I;" ";A\$(I)
542 GOTO 490

```

Change: 100 PRINT "HOW MANY MONSTERS DO YOU      KNOW";
        120 RANDOMIZE
        140 PRINT "SOMETHING YOU CAN SEE IN      THIS ROOM"
        160 PRINT "!!!!!!!!!!!!!!MONSTER!!!!!!!!!!!"
        180 PRINT "TELL ME SOMETHING YOU CAN      SEE"
        240 PRINT "TELL ME A DIFFERENT THING     YOU CAN"
        360 R=1+INT(RND*10)
        370 T=3+INT(RND*7)
        460 IF A$(I)="USED" THEN 470
        520 IF (G<1)+(G>10) THEN 490
        530 IF A$(G)<>"USED" THEN 550
        540 PRINT "YOU TRIED THAT ALREADY!"
        550 IF (G<1)+(G>10) THEN 490
        600 CALL SOUND (100,440,2)

```

TRS-80 COLOR COMPUTER

Delete: 120, 410, 440, 480, 590

```

Change: 360 R=1+RND(10)
        370 T=3+RND(7)
        600 SOUND 173,3

```


Program Listing

```
10 REM BOTTOMLESS PIT
100 PRINT
110 PRINT "WATCH OUT! YOU'RE TEETERING"
120 PRINT "ON THE VERY EDGE OF THE MONSTER'S"
130 PRINT "BOTTOMLESS PIT! WHATEVER YOU DO,"
140 PRINT "DON'T TOUCH THE ENTER KEY, OR"
150 PRINT "YOU'LL FALL RIGHT IN";
160 INPUT X$
170 PRINT
180 FOR I=1 TO 30
190 PRINT TAB(17);"OOOPS!"
200 NEXT I
210 R=6
220 A$="!---!"
230 PRINT TAB(R);A$;TAB(35-R);A$
240 ON INT(RND(1)*3)+1 GOTO 250,280,270
250 R=R-1
260 GOTO 280
270 R=R+1
280 IF R>14 THEN R=12
290 IF R<1 THEN R=2
300 GOTO 230
```

If You Have . . .

APPLE II

Change: 140 PRINT "DON'T TOUCH THE RETURN KEY, OR"

ATARI

```
Add: 91 DIM X$(1),A$(5)
      92 POKE 752,1
      315 FOR I=1 TO T
      325 PRINT " ";:NEXT I: RETURN
```

Change: 140 PRINT "DON'T TOUCH THE RETURN KEY, OR"
190 T=16:GOSUB 315:PRINT "OOPS!"
210 R=6:T=6
230 GOSUB 315:PRINT A\$;:T=35-R:GOSUB 315:PRINT A\$
240 ON INT(RND(0)*3)+1 GOTO 250,280,270

COMMODORE 64

Change: 140 PRINT "DON'T TOUCH THE RETURN KEY, OR"

(continued)

COMMODORE VIC-20

Change: 140 PRINT "DON'T TOUCH THE RETURN KEY, OR"
150 PRINT "YOU'LL FALL RIGHT IN"
190 PRINT TAB(8);"OOOPS!"
230 PRINT TAB(R);A\$;TAB(16-R);A\$

TEXAS INSTRUMENTS 99/4A

Add: 132 PRINT "WHATEVER YOU DO,"
282 R=12
292 R=2

Change: 120 PRINT "ON THE VERY EDGE OF THE "
130 PRINT "MONSTER'S BOTTOMLESS PIT!"
230 PRINT TAB(R);A\$;TAB(25-R);A\$
240 ON INT(RND*3)+1 GOTO 250,280,270
280 IF R<=14 THEN 290
290 IF R>=1 THEN 300

TRS-80 COLOR COMPUTER

Change: 230 PRINT TAB(R);A\$;TAB(31-R);A\$
240 ON RND(3) GOTO 250,280,270
280 IF R>12 THEN R=10
290 IF R<6 THEN R=7

Sweet Revenge

Is there somebody in this world that you really can't stand? Here's your own personal monster — ready to torment every enemy you have in the world.

Sample Run

MONSTER AT YOUR SERVICE, MASTER.

WHO IS YOUR WORST ENEMY? SHERMAN

I'M TICKLING SHERMAN SO HARD THAT
SHERMAN ISN'T LAUGHING ANY MORE.

SHOULD I GO ON? YES

I'M BITING SHERMAN ON THE NECK.
YUCK! SHERMAN TASTES HORRIBLE!

SHOULD I GO ON? YES

I'M ATTACKING SHERMAN
WITH MY TERRIBLE CLAWS.
HEY! NO FAIR! SHERMAN
IS FIGHTING BACK!

SHOULD I GO ON? YES

I'M PUSHING SHERMAN OFF
THE WORLD'S TALLEST BUILDING.

OOPS--I NEARLY SLIPPED!

SHOULD I GO ON? YES

I'M JUMPING UP AND DOWN ON SHERMAN.
WHEE! SHERMAN IS VERY BOUNCY--
EVEN MORE FUN THAN A TRAMPOLINE!

SHOULD I GO ON? YES

I BORROWED A DRAGON
FROM FRANK'S RENT-A-MONSTER.
IT'S MAKING THINGS HOT FOR SHERMAN.

SHOULD I GO ON? NO

AND JUST WHEN I WAS STARTING TO HAVE FUN!
POOR SHERMAN!

□ Program Listing

```
10 REM SWEET REVENGE
100 PRINT
110 PRINT "MONSTER AT YOUR SERVICE, MASTER."
120 PRINT
130 PRINT "WHO IS YOUR WORST ENEMY";
140 INPUT E$
150 IF E$="" THEN 130
160 PRINT
170 PRINT "I'M TICKLING ";E$;" SO HARD THAT"
180 PRINT E$;" ISN'T LAUGHING ANY MORE."
190 GOSUB 490
200 PRINT "I'M BITING ";E$;" ON THE NECK."
210 PRINT "YUCK! ";E$;" TASTES HORRIBLE!"
220 GOSUB 490
230 PRINT "I'M ATTACKING ";E$
240 PRINT "WITH MY TERRIBLE CLAWS."
250 PRINT "HEY! NO FAIR! ";E$
260 PRINT "IS FIGHTING BACK!"
270 GOSUB 490
280 PRINT "I'M PUSHING ";E$;" OFF"
290 PRINT "THE WORLD'S TALLEST BUILDING."
300 FOR R=1 TO 500
310 NEXT R
320 FOR I=1 TO 16
330 PRINT
340 NEXT I
350 PRINT "OOPS--I NEARLY SLIPPED!"
360 GOSUB 490
370 PRINT "I'M JUMPING UP AND DOWN ON ";E$;". "
380 PRINT "WHEE! ";E$;" IS VERY BOUNCY--"
390 PRINT "EVEN MORE FUN THAN A TRAMPOLINE!"
400 GOSUB 490
410 PRINT "I BORROWED A DRAGON"
420 PRINT "FROM FRANK'S RENT-A-MONSTER."
430 PRINT "IT'S MAKING THINGS HOT FOR ";E$;". "
440 GOSUB 490
450 PRINT "SORRY--IT'S TIME FOR MY COFFEE BREAK."
460 PRINT "I'LL LEAVE ";E$;" IN THE BOTTOMLESS PIT."
470 PRINT "DROP IN IF YOU WANT TO SAY HELLO."
480 END
490 PRINT
500 PRINT "SHOULD I GO ON";
510 INPUT A$
520 PRINT
530 IF MID$(A$,1,1)<>"N" THEN RETURN
540 PRINT "AND JUST WHEN I WAS STARTING TO HAVE FUN!"
550 PRINT "POOR " E$;"!"
560 END
```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM E\$(11), A\$(3)

Change: 530 IF A\$(1,1) <> "N" THEN RETURN

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 130 PRINT "WHO IS YOUR WORST ENEMY"
500 PRINT "SHOULD I GO ON"

TEXAS INSTRUMENTS 99/4A

Add: 532 GOTO 540
534 RETURN

Change: 110 PRINT "AT YOUR SERVICE, MASTER."
290 PRINT "THE WORLD'S TALLEST BUILDING"
390 PRINT "EVEN MORE FUN THAN A TRAMPOLINE!"
450 PRINT "SORRY--IT'S TIME FOR MY COFFEE BREAK."
470 PRINT "DROP IN IF YOU WANT TO SAY HELLO."
530 IF SEG\$(A\$,1,1) <> "N" THEN 534
550 PRINT "POOR "; E\$; "!"

TRS-80 COLOR COMPUTER

Change: 320 FOR I=1 TO 10

Monster Hunt

When you're a famous movie star (which is exactly what you are), you can do exciting things like hunting the Abominable Snowman. Catch it and it may star in your next movie. But don't take too long — it's cold in the Himalaya Mountains!

□ Sample Run

THE ABOMINABLE SNOWMAN
LEAVES TRACKS IN THE SNOW: !!

YOU'RE THE FAMOUS MOVIE STAR: *
HUNTING THE SNOWMAN.
YOU CAN MOVE LEFT (L) OR RIGHT (R)
BUT YOU'RE NOT TOO GOOD AT STOPPING
ON THE ICE.
WHEN YOU SEE THE ?, ENTER L OR R.

BUT FIRST,
HOW MANY DOLLARS WILL YOU SPEND
ON THE HUNT? 1999

YOU HAVE 4 CHANCES TO
CATCH THE SNOWMAN.
PRESS THE ENTER KEY TO BEGIN?

```
      !!      *
*      !!
      !!      *
      !! *
*      !!
      ? L
      ? R
      ? L
      ? L
```

YOU TOOK TOO LONG!
YOU FROZE TO DEATH!

RUN

YOU HAVE 9 CHANCES TO
CATCH THE SNOWMAN.
PRESS THE ENTER KEY TO BEGIN?

```
      *      !!
          !!      *
          * !!
          !!
          !!
*      !!
          !!      *
          ? R
          ? L
          ? R
          ? L
          ? R
          ? L
```

FOUND HIM!
YOU'RE THE WORLD'S GREATEST
MONSTER HUNTER!

□ Program Listing

```
10 REM MONSTER HUNT
100 PRINT "THE ABOMINABLE SNOWMAN"
110 PRINT "LEAVES TRACKS IN THE SNOW: !!"
120 PRINT
130 PRINT "YOU'RE THE FAMOUS MOVIE STAR: *"
140 PRINT "HUNTING THE SNOWMAN."
150 PRINT "YOU CAN MOVE LEFT (L) OR RIGHT (R)"
160 PRINT "BUT YOU'RE NOT TOO GOOD AT STOPPING"
170 PRINT "ON THE ICE."
180 PRINT "WHEN YOU SEE THE ?, ENTER L OR R."
190 PRINT
200 PRINT "BUT FIRST,"
210 PRINT "HOW MANY DOLLARS WILL YOU SPEND"
220 PRINT "ON THE HUNT";
230 INPUT X
240 IF X<32767 THEN 270
250 PRINT "TOO MUCH!"
260 GOTO 230
270 RANDOMIZE X
280 CH=3+INT(RND(1)*8)
290 PRINT
300 PRINT "YOU HAVE ";CH-1;" CHANCES TO"
310 PRINT "CATCH THE SNOWMAN."
320 PRINT "PRESS THE ENTER KEY TO BEGIN";
330 INPUT X$
340 FOR I=1 TO 25
350 PRINT
360 NEXT I
370 MP=1+INT(RND(1)*34)
380 YP=1+INT(RND(1)*35)
390 GOTO 560
400 IF MP=YP OR ABS(MP-YP)=1 THEN 370
410 INPUT DIR$
420 IF DIR$="L" THEN YP=YP-INT(RND(1)*YP)-1
430 IF DIR$="R" THEN YP=YP+INT(RND(1)*(35-YP))+1
440 MP=MP-1+INT(RND(1)*3)
450 IF MP>34 THEN MP=33
460 IF YP>35 THEN YP=35
470 IF YP<1 THEN YP=1
480 IF MP<1 THEN MP=1
490 IF MP=YP OR MP+1=YP THEN 510
500 GOTO 560
510 PRINT
520 PRINT "FOUND HIM!"
530 PRINT "YOU'RE THE WORLD'S GREATEST"
540 PRINT "MONSTER HUNTER!"
```

(continued)


```

550 END
560 IF MP>YP THEN 640
570 A=MP
580 A$="!!"
590 B=YP
600 B$="*"
610 FW=2
620 SW=1
630 GOTO 710
640 A=YP
650 A$="*"
660 B=MP
670 B$="!!"
680 FW=1
690 SW=2
700 IF YP=1 OR MP=1 THEN 740
710 FOR I=1 TO A-1
720 PRINT " ";
730 NEXT I
740 PRINT A$;
750 FOR I=A+FW TO B-1
760 PRINT " ";
770 NEXT I
780 PRINT B$;
790 FOR I=B+SW TO 35
800 PRINT " ";
810 NEXT I
820 CH=CH-1
830 IF CH=0 THEN 850
840 GOTO 410
850 PRINT
860 PRINT "YOU TOOK TOO LONG!"
870 PRINT "YOU FROZE TO DEATH!"

```

If You Have . . .

APPLE II

Change: 270 REM
320 PRINT "PRESS THE RETURN KEY TO BEGIN";

ATARI

Add: 91 DIM X\$(1),A\$(2),B\$(2),DIR\$(5)

Change: 240 IF X<32767 THEN 280
270 REM
320 PRINT "PRESS THE RETURN KEY TO BEGIN";
370 MP=1+INT(RND(0)*32)
380 YP=1+INT(RND(0)*33)
420 IF DIR\$="L" THEN YP=YP-INT(RND(0)*YP)-1

```

430 IF DIR$="R" THEN YP=YP+INT(RND(0)*(33-YP))+1
440 MP=MP-1+INT(RND(0)*3)
450 IF MP>32 THEN MP=31
460 IF YP>33 THEN YP=33
790 FOR I=B+SW TO 33

```

COMMODORE 64

Delete: 270

```

Change: 240 IF X<32767 THEN 280
280 CH=3+INT(RND(0)*8)
320 PRINT "PRESS THE RETURN KEY TO BEGIN";
370 MP=1+INT(RND(0)*34)
380 YP=1+INT(RND(0)*35)
420 IF DIR$="L" THEN YP=YP-INT(RND(0)*YP)-1
430 IF DIR$="R" THEN YP=YP+INT(RND(0)*(35-YP))+1
440 MP=MP-1+INT(RND(0)*3)

```

COMMODORE VIC-20

Delete: 270

```

Change: 220 PRINT "ON THE HUNT"
240 IF X<32767 THEN 280
280 CH=3+INT(RND(0)*8)
320 PRINT "PRESS THE RETURN KEY TO BEGIN"
370 MP=1+INT(RND(0)*15)
380 YP=1+INT(RND(0)*16)
420 IF DIR$="L" THEN YP=YP-INT(RND(0)*YP)-1
430 IF DIR$="R" THEN YP=YP+INT(RND(0)*(16-YP))+1
440 MP=MP-1+INT(RND(0)*3)
450 IF MP>15 THEN MP=14
460 IF YP>16 THEN YP=16
790 FOR I=B+SW TO 16

```

TEXAS INSTRUMENTS 99/4A

```

Add: 422 YP=YP-INT(RND*YP)-1
432 YP=YP+INT(RND*(30-YP))+1
452 MP=28
462 YP=30
472 YP=1
482 MP=1

```

```

Change: 110 PRINT "LEAVES TRACKS IN THE SNOW:!!"
130 PRINT "YOU'RE THE FAMOUS MOVIE STAR: *"
150 PRINT "YOU CAN MOVE LEFT (L) OR RIGHT (R) "
160 PRINT "BUT YOU'RE NOT TOO GOOD AT"
170 PRINT "STOPPING ON THE ICE."
180 PRINT "WHEN YOU SEE THE ?, ENTER L OR R."
210 PRINT "HOW MANY DOLLARS WILL YOU" (continued)

```

```

220 PRINT "SPEND ON THE HUNT";
270 RANDOMIZE
280 CH=3+INT(RND*8)
370 MP=1+INT(RND*29)
380 YP=1+INT(RND*30)
400 IF (MP=YP)+(ABS(MP-YP)=1) THEN 370
420 IF DIR$<>"L" THEN 430
430 IF DIR$<>"R" THEN 440
440 MP=MP-1+INT(RND*3)
450 IF MP<=29 THEN 460
460 IF YP<=30 THEN 470
470 IF YP>1 THEN 480
480 IF MP>=1 THEN 490
490 IF (MP=YP)+(MP+1=YP) THEN 510
530 PRINT "YOU'RE THE WORLD'S GREATEST"
700 IF (YP=1)+(MP=1) THEN 740
790 FOR I=B+SW TO 30

```

TRS-80 COLOR COMPUTER

```

Change: 240 IF X<32767 THEN 280
270 REM
280 CH=3+RND(8)
340 FOR I=1 TO 15
370 MP=RND(28)
380 YP=RND(29)
420 IF DIR$="L" THEN YP=YP-RND(YP)-1
430 IF DIR$="R" THEN YP=YP+RND(29-YP)+1
440 MP=MP-1+RND(3)
450 IF MP>28 THEN MP=27
460 IF YP>29 THEN YP=29
790 FOR I=B+SW TO 28

```

Troll Bridge

There's a prince or princess to be rescued, and you dare not refuse the challenge. The problem is that you have to cross three bridges guarded by greedy trolls. The trolls want your money, and grobleks are the coin of the realm. If you're clever, you can bargain with the trolls — but don't insult them by being too cheap!

Sample Run

ARE YOU A BOY OR A GIRL? BOY

THE PRINCESS IS TRAPPED IN A TOWER.
YOU ARE ASSIGNED TO SAVE THE PRINCESS.

YOU MUST CROSS THREE BRIDGES.
EACH ONE IS GUARDED BY A TROLL.
YOU MUST BARGAIN WITH THE TROLLS
TO LET YOU CROSS.

YOU MAY OFFER AS MUCH AS YOU HAVE.
IF THE TROLL ACCEPTS, YOU MAY
MAKE A LOWER OFFER.

BUT WATCH OUT!
IF YOU OFFER TOO LITTLE, THE TROLL
WILL BE OFFENDED AND EAT YOU UP.

HOW MANY GROBLEKS WOULD YOU LIKE? 1000

SORRY: YOU'VE ONLY GOT 144 GROBLEKS.

I AM TROLL NUMBER 1.
HOW MANY GROBLEKS WILL YOU PAY
TO CROSS THIS BRIDGE? 70

I ACCEPT.

TO PAY, ENTER P--
TO MAKE ANOTHER OFFER,
JUST HIT THE ENTER KEY?

HOW MANY GROBLEKS WILL YOU PAY
TO CROSS THIS BRIDGE? 40

I ACCEPT.

(continued)

[There's more—when you get to the end of the program, you'll see something like this:]

I AM TROLL NUMBER 3.
HOW MANY GROBLEKS WILL YOU PAY
TO CROSS THIS BRIDGE? 62

I ACCEPT.

TO PAY, ENTER P--
TO MAKE ANOTHER OFFER,
JUST HIT THE ENTER KEY? P

YOU CROSSED ALL THREE BRIDGES.
YOU RESCUE THE PRINCESS
AND START LIFE TOGETHER WITH THE
2 GROBLEKS YOU HAVE LEFT.

Program Listing

```
10 REM TROLL BRIDGE
100 PRINT
110 PRINT "ARE YOU A BOY OR A GIRL";
120 INPUT S$
130 IF MID$(S$,1,1)="B" THEN 160
140 IF MID$(S$,1,1)="G" THEN 180
150 GOTO 110
160 R$="PRINCESS"
170 GOTO 190
180 R$="PRINCE"
190 PRINT
200 PRINT "THE ";R$; " IS TRAPPED IN A TOWER."
210 PRINT "YOU ARE ASSIGNED TO SAVE THE "; R$; "."
220 PRINT
230 PRINT "YOU MUST CROSS THREE BRIDGES."
240 PRINT "EACH ONE IS GUARDED BY A TROLL."
250 PRINT "YOU MUST BARGAIN WITH THE TROLLS"
260 PRINT "TO LET YOU CROSS."
270 PRINT
280 PRINT "YOU MAY OFFER AS MUCH AS YOU HAVE."
290 PRINT "IF THE TROLL ACCEPTS, YOU MAY"
300 PRINT "MAKE A LOWER OFFER."
310 PRINT
320 PRINT "BUT WATCH OUT!"
330 PRINT "IF YOU OFFER TOO LITTLE, THE TROLL"
340 PRINT "WILL BE OFFENDED AND EAT YOU UP."
350 PRINT
360 PRINT "HOW MANY GROBLEKS WOULD YOU LIKE";
```

```

370 INPUT N
380 IF N>100 THEN 410
390 PRINT "THAT'S NOT ENOUGH GROBLEKS!"
400 GOTO 360
410 PRINT
420 IF N>32767 THEN N=1000
430 RANDOMIZE N
440 C=N-INT(RND(1)*N)
450 PRINT "SORRY: YOU'VE ONLY GOT ";C;" GROBLEKS."
460 PRINT
470 FOR I=1 TO 3
480 TR(I)=1+(RND(1)*C/2)
490 NEXT I
500 FOR I=1 TO 3
510 PRINT "I AM TROLL NUMBER ";I; "."
520 PRINT "HOW MANY GROBLEKS WILL YOU PAY"
530 PRINT "TO CROSS THIS BRIDGE";
540 INPUT T
550 IF T<TR(I) THEN 820
560 IF T<C+1 THEN 590
570 PRINT "YOU DON'T HAVE THAT MUCH!"
580 GOTO 520
590 PRINT
600 PRINT "I ACCEPT."
610 PRINT
620 PRINT "TO PAY, ENTER P--"
630 PRINT "TO MAKE ANOTHER OFFER,"
640 PRINT "JUST HIT THE ENTER KEY";
650 INPUT A$
660 PRINT
670 IF A$="P" THEN 700
680 IF A$="" THEN 520
690 GOTO 620
700 C=C-T
710 IF I<3 THEN PRINT "YOU'VE NOW GOT ";C; " GROBLEKS."
720 PRINT
730 NEXT I
740 PRINT "YOU CROSSED ALL THREE BRIDGES."
750 PRINT "YOU RESCUE THE "; R$
760 IF C>0 THEN 790
770 PRINT "BUT YOU'RE BROKE!"
780 END
790 PRINT "AND START LIFE TOGETHER WITH THE"
800 PRINT C;" GROBLEKS YOU HAVE LEFT. "
810 END
820 PRINT "THAT OFFER IS INSULTING!"
830 PRINT "I'LL EAT YOU INSTEAD!"
840 END

```

If You Have . . .

APPLE II

Delete: 430

Change: 640 PRINT "JUST HIT THE RETURN KEY;"

ATARI

Add: 91 DIM S\$(4),R\$(8),A\$(1),TR(3)

Change: 130 IF S\$(1,1)="B" THEN 160
140 IF S\$(1,1)="G" THEN 180
430 REM
440 C=N-INT(RND(0)*N)
480 TR(I)=1+(RND(0)*C/2)
560 IF T<C+1 THEN 590
640 PRINT "JUST HIT THE RETURN KEY";

COMMODORE 64

Delete: 430

Change: 440 C=N-INT(RND(0)*N)
480 TR(I)=1+(RND(0)*C/2)
640 PRINT "JUST HIT THE RETURN KEY";

COMMODORE VIC-20

Delete: 430

Add: 315 PRINT "HIT RETURN TO CONTINUE"
316 INPUT X\$

Change: 110 PRINT "ARE YOU A BOY OR A GIRL"
360 PRINT "HOW MANY GROBLEKS WOULD YOU LIKE"
440 C=N-INT(RND(0)*N)
480 TR(I)=1+(RND(0)*C/2)
530 PRINT "TO CROSS THIS BRIDGE"
640 PRINT "JUST HIT THE RETURN KEY"

TEXAS INSTRUMENTS 99/4A

Add: 212 PRINT "TO SAVE THE ";R\$;"."
422 N=1000
712 PRINT "YOU HAVE NOW GOT ";C
714 PRINT "GROBLEKS"

Change: 130 IF SEG\$(S\$,1,1)="B" THEN 160
140 IF SEG\$(S\$,1,1)="G" THEN 180

```

200 PRINT "THE ";R$; " IS TRAPPED"
210 PRINT "IN A TOWER. YOU ARE ASSIGNED"
230 PRINT "YOU MUST CROSS THREE BRIDGES"
240 PRINT "EACH ONE IS GUARDED BY A "
250 PRINT "TROLL. YOU MUST BARGAIN WITH"
260 PRINT "THE TROLLS TO LET YOU CROSS."
290 PRINT "IF THE TROLL ACCEPTS, YOU MAY"
380 IF N>100 THEN 410
420 IF N<32767 THEN 430
430 RANDOMIZE
440 C=N-INT(RND*N)
480 TR(I)=1+(RND*C/2)
520 PRINT "HOW MANY GROBLEKS WILL YOU PAY"
560 IF T<C+1 THEN 590
710 IF I>=3 THEN 720
740 PRINT "YOU CROSSED ALL THREE BRIDGES."
760 IF C>0 THEN 790

```

TRS-80 COLOR COMPUTER

Delete: 190, 220, 270, 310, 350, 430

Change:

```

170 GOTO 200
250 PRINT "YOU MUST BARGAIN WITH THE TROLLS";
260 PRINT "TO LET YOU CROSS ";
280 PRINT "YOU MAY OFFER AS MUCH AS YOU HAVE ";
290 PRINT "IF THE TROLL ACCEPTS, YOU MAY";
300 PRINT "MAKE A LOWER OFFER.";
330 PRINT "IF YOU OFFER TOO LITTLE, THE TROLL ";
340 PRINT "WILL BE OFFENDED AND EAT YOU UP.";
440 C=N-RND(N)
480 TR(I)=1+RND(C/2)

```


Transylvanian Trip

The world's most famous vampire likes to visit home now and then. Unfortunately, he has lots of trouble with Transylvania's metric system. You can help him solve his monstrous problems.

Sample Run

DRACULA LIKES TO GO HOME TO
TRANSYLVANIA ONCE IN AWHILE.
THERE'S JUST ONE PROBLEM.
TRANSYLVANIA'S ON THE METRIC
SYSTEM. DRACULA NEEDS TO CONVERT
FROM OUR SYSTEM.

HELP HIM CONVERT LITERS OF BLOOD
TO PINTS, QUARTS AND GALLONS
--AND VICE VERSA.

DO YOU WANT TO CONVERT
QUARTS, PINTS, GALLONS, OR LITERS? LITERS
HOW MANY DO YOU WANT TO CONVERT? 1500

1500 LITERS EQUALS:

3169.5 PINTS
1584.75 QUARTS
396.1875 GALLONS

DO YOU WANT TO TRY AGAIN? YES

DO YOU WANT TO CONVERT
QUARTS, PINTS, GALLONS, OR LITERS? PINTS
HOW MANY DO YOU WANT TO CONVERT? 1273

1273 PINTS EQUALS 602.129 LITERS

DO YOU WANT TO TRY AGAIN? NO

□ Program Listing

```
10 REM TRANSYLVANIAN TRIP
100 PRINT "DRACULA LIKES TO GO HOME TO"
110 PRINT "TRANSYLVANIA ONCE IN AWHILE."
120 PRINT "THERE'S JUST ONE PROBLEM."
130 PRINT "TRANSYLVANIA'S ON THE METRIC"
140 PRINT "SYSTEM. DRACULA NEEDS TO CONVERT"
150 PRINT "FROM OUR SYSTEM."
160 PRINT
170 PRINT "HELP HIM CONVERT LITERS OF BLOOD"
180 PRINT "TO PINTS, QUARTS AND GALLONS"
190 PRINT "--AND VICE VERSA."
200 PRINT
210 PRINT "DO YOU WANT TO CONVERT"
220 PRINT "QUARTS, PINTS, GALLONS, OR LITERS";
230 INPUT M$
240 PRINT "HOW MANY DO YOU WANT TO CONVERT";
250 INPUT A
260 PRINT
270 IF MID$(M$,1,1)="P" THEN C=.473
280 IF MID$(M$,1,1)="Q" THEN C=.946
290 IF MID$(M$,1,1)="G" THEN C=3.785
300 IF MID$(M$,1,1)="L" THEN C=2.113
310 IF MID$(M$,1,1)<>"L" THEN 380
320 PRINT A; M$; " EQUALS:"
330 PRINT
340 PRINT C*A; "PINTS"
350 PRINT C*A/2; "QUARTS"
360 PRINT C*A/8; "GALLONS"
370 GOTO 400
380 PRINT A; " ";M$;" EQUALS ";
390 PRINT C*A; "LITERS"
400 PRINT
410 PRINT "DO YOU WANT TO TRY AGAIN";
420 INPUT Y$
430 IF MID$(Y$,1,1)="N" THEN END
440 PRINT
450 GOTO 210
```

□ If You Have . . .

APPLE II

```
Change: 320 PRINT A; " ";M$;" EQUALS:"
        340 PRINT C*A; " PINTS"
        350 PRINT C*A/2; " QUARTS"
        360 PRINT C*A/8; " GALLONS"
        390 PRINT C*A; " LITERS"
```

(continued)

ATARI

Add: 91 DIM M\$(11),Y\$(11)

Change: 270 IF M\$(1,1) = "P" THEN C=.473
280 IF M\$(1,1) = "Q" THEN C=.946
290 IF M\$(1,1) = "G" THEN C=3.785
300 IF M\$(1,1) = "L" THEN C=2.113
310 IF M\$(1,1) <>"L" THEN 380
430 IF Y\$(1,1) = "N" THEN END

COMMODORE 64 No Changes Required

COMMODORE VIC-20

Change: 220 PRINT "QUARTS, PINTS, GALLONS, LITERS"
240 PRINT "HOW MANY DO YOU WANT TO CONVERT"
410 PRINT "DO YOU WANT TO TRY AGAIN"

TEXAS INSTRUMENTS 99/4A

Add: 272 C=0.473
282 C=0.946
292 C=3.785
302 C=2.113
460 END

Change: 170 PRINT "HELP HIM CONVERT LITERS OF"
180 PRINT "BLOOD TO PINTS, QUARTS, AND"
190 PRINT "GALLONS--AND VICE VERSA."
220 PRINT "QUARTS, PINTS, GALLONS, OR LITERS";
240 PRINT "HOW MANY DO YOU WANT TO CONVERT";
270 IF SEG\$(M\$,1,1) <> "P" THEN 280
280 IF SEG\$(M\$,1,1) <> "Q" THEN 290
290 IF SEG\$(M\$,1,1) <> "G" THEN 300
300 IF SEG\$(M\$,1,1) <> "L" THEN 310
310 IF SEG\$(M\$,1,1) <> "L" THEN 380
430 IF SEG\$(Y\$,1,1) = "N" THEN 460

TRS-80 COLOR COMPUTER No Changes Required

Save the Monsters

You — and a telephone — are all that stand between your favorite monsters and a horde of angry humans. You're running the monster emergency hot line. Warn your friends, and they'll escape. Fail — and run for your life!

Sample Run

WHAT IS YOUR NAME, MONSTER? STEVE

WELCOME TO THE MONSTER PHONE BOOTH.

YOU, STEVE, MUST PHONE THREE MONSTERS
AND WARN THEM THAT HUMANS ARE COMING.
IF YOU DON'T, THE MONSTERS WILL GET
EXTREMELY UPSET.

EASY (1), MEDIUM (2), OR HARD (3)? 1

MONSTER PHONE DIRECTORY

CAT WOMAN	884
WEREWOLF	166
BLOB	558

PRESS THE ENTER KEY WHEN READY?

QUICK! HUMANS COMING!
DIAL THE CAT WOMAN? 884

THIS IS THE CAT WOMAN.
THANKS FOR WARNING ME!
I'M GETTING OUT OF HERE!

ONLY 2 MORE TO GO!

QUICK! HUMANS COMING!
DIAL THE WEREWOLF? 123

NOPE! HURRY!
DIAL THE WEREWOLF? 166

THIS IS THE WEREWOLF.
THANKS FOR WARNING ME!
I'M GETTING OUT OF HERE!

ONLY 1 MORE TO GO!

QUICK! HUMANS COMING!
DIAL THE BLOB? 901

(continued)

ONE LAST CHANCE--QUICK!
DIAL THE BLOB? 558

THIS IS THE BLOB.
THANKS FOR WARNING ME!
I'M GETTING OUT OF HERE!

YOU DID IT!
THE MONSTERS THANK YOU,
BUY YOU DINNER,
AND PAY YOU A MILLION DOLLARS!

Program Listing

```
10 REM SAVE THE MONSTERS!
100 PRINT
110 PRINT "WHAT IS YOUR NAME, MONSTER";
120 INPUT N$
130 IF N$="" THEN 110
140 PRINT
150 PRINT "WELCOME TO THE MONSTER PHONE BOOTH."
160 PRINT
170 PRINT "YOU, ";N$;", MUST PHONE THREE MONSTERS"
180 PRINT "AND WARN THEM THAT HUMANS ARE COMING."
190 PRINT "IF YOU DON'T, THE MONSTERS WILL GET"
200 PRINT "EXTREMELY UPSET."
210 PRINT
220 PRINT "EASY (1), MEDIUM (2), OR HARD (3)";
230 INPUT L
240 IF L>0 AND L<4 THEN 270
250 PRINT "1, 2 OR 3, PLEASE!"
260 GOTO 220
270 L=L*3
280 PRINT
290 PRINT TAB(8);"MONSTER PHONE DIRECTORY"
300 PRINT
310 C=3
320 FOR I=1 TO L
330 READ M$
340 PRINT M$;
350 PRINT TAB(20);
360 N(I)=100+INT(RND(1)*900)
370 PRINT N(I)
380 NEXT I
390 RESTORE
400 PRINT
410 PRINT "PRESS THE ENTER KEY WHEN READY";
420 INPUT X$
430 FOR I=1 TO 25
440 PRINT
450 NEXT I
```

```

460 Z=1+INT(RND(1)*L)
470 FOR I=1 TO Z
480 READ M$
490 NEXT I
500 RESTORE
510 IF N(Z)=1000 THEN 460
520 PRINT "QUICK! HUMANS COMING!"
530 PRINT "DIAL THE ";M$;
540 INPUT A
550 IF A=0 THEN 520
560 PRINT
570 IF A<>N(Z) THEN 740
580 PRINT "THIS IS THE ";M$;"."
590 PRINT "THANKS FOR WARNING ME!"
600 PRINT "I'M GETTING OUT OF HERE!"
610 PRINT
620 N(Z)=1000
630 C=C-1
640 IF C=0 THEN 680
650 PRINT "ONLY ";C;" MORE TO GO!"
660 PRINT
670 GOTO 460
680 PRINT
690 PRINT "YOU DID IT!"
700 PRINT "THE MONSTERS THANK YOU,"
710 PRINT "BUY YOU DINNER,"
720 PRINT "AND PAY YOU A MILLION DOLLARS!"
730 END
740 F=F+1
750 IF F=1 THEN PRINT "NOPE! HURRY!"
760 IF F=2 THEN PRINT "ONE LAST CHANCE--QUICK!"
770 IF F=3 THEN 790
780 GOTO 530
790 PRINT "TOO LATE!"
800 PRINT "THE NUMBER WAS ";N(Z);"!"
810 PRINT "THE HUMANS ARE CHASING THE ";M$
820 PRINT "AND THE ";M$;" IS CHASING YOU!"
830 DATA CAT,WOMAN,WEREWOLF,BLOB,TROLL
840 DATA ZOMBIE,VAMPIRE,MUMMY,OGRE,DRAGON

```

If You Have . . .

APPLE II

Change: 410 PRINT "PRESS THE RETURN KEY WHEN READY";

ATARI

Add: 91 DIM N\$(40),M\$(9),N(9),X\$(1)
291 PRINT "MONSTER PHONE DIRECTORY"
849 END

(continued)

```
855 FOR B=1 TO R
856 PRINT " ";
857 NEXT B: RETURN
```

```
Change: 290 R=7: GOSUB 855
350 R=20-(LEN(M$)): GOSUB 855
360 N(I)=100+INT(RND(0)*900)
410 PRINT "PRESS THE RETURN KEY WHEN READY";
460 Z=1+INT(RND(0)*L)
```

COMMODORE 64

```
Change: 410 PRINT "PRESS THE RETURN KEY WHEN READY";
```

COMMODORE VIC-20

```
Change: 110 PRINT "WHAT IS YOUR NAME, MONSTER"
220 PRINT "EASY (1), MEDIUM (2), OR HARD (3)"
350 PRINT TAB(11);
410 PRINT "PRESS THE RETURN KEY WHEN READY"
530 PRINT "DIAL THE ";M$
```

TEXAS INSTRUMENTS 99/4A

```
Add: 752 PRINT "NOPE! HURRY!"
762 PRINT "ONE LAST CHANCE--QUICK!"
```

```
Change: 170 PRINT "YOU, ";N$;",";" MUST PHONE THREE MONSTERS"
180 PRINT "AND WARN THEM THAT HUMANS ARE COMING."
220 PRINT "EASY (1), MEDIUM (2), OR HARD (3)"
240 IF (L>0) * (L<4) THEN 270
350 PRINT TAB(15);
360 N(I)=100+INT(RND*900)
460 Z=1+INT(RND*L)
750 IF F <> 1 THEN 760
760 IF F <> 2 THEN 770
```

TRS-80 COLOR COMPUTER

```
Change: 360 N(I)=100+RND(900)
430 FOR I=1 TO 15
460 Z=RND(L)
```

Sneaky Sea Serpent

A sneaky, snaky sea serpent slyly slithered into your screen. Run this program, and sea for yourself.

Sample Run

UH-OH! YOUR COMPUTER HAS BEEN
TAKEN OVER BY A SNEAKY SEA
SERPENT. AND WORSE YET, THE
CREATURE HAS THE HICCUPS!

HIT THE ENTER KEY TO SEE?

```
QQQQQQQQ:>
Q
Q
Q
QQQQQQQQQQQQQQ
          Q
          Q
          Q
QQQQQQQQQQQQQQ
Q
Q
Q
QQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQ
          Q
          Q
          Q
QQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQ
Q
Q
Q
          HIC
QQQQQQQQQQQQQQQQQQQQQQ
          Q
QQQQQQQQQQQQQQQQQQQQ
Q
Q
Q
QQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQ
          Q
          Q
          Q
QQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQQ
Q
Q
Q
QQQQQQQQQQ
          Q
QQQQQQQQQQ
```

Break

Program Listing

```
10 REM SNEAKY SEA SERPENT
100 PRINT "UH-OH! YOUR COMPUTER HAS BEEN"
110 PRINT "TAKEN OVER BY A SNEAKY SEA"
120 PRINT "SERPENT. AND WORSE YET, THE"
130 PRINT "CREATURE HAS THE HICCUPS!"
140 PRINT
150 PRINT "HIT THE ENTER KEY TO SEE";
160 INPUT D$
170 PRINT
180 PRINT "QQQQQQQQQ:>"
190 K=INT(RND(1)*10)+2
200 FOR A=1 TO K
210 PRINT "Q"
220 NEXT A
230 L=INT(RND(1)*39)
240 FOR B=1 TO L
250 PRINT "Q";
260 NEXT B
270 M=INT(RND(1)*10)
280 FOR C=1 TO M+2
290 PRINT TAB(B);"Q"
300 NEXT C
310 FOR D=B TO 2 STEP -1
320 PRINT "Q";
330 NEXT D
340 PRINT "Q"
350 Z=Z+1
360 IF L<>10+K THEN 200
370 PRINT TAB(18);"HIC"
380 PRINT CHR$(7)
390 L=0
400 GOTO 200
```

If You Have . . .

APPLE II

Change: 150 PRINT "HIT THE RETURN KEY TO SEE";

ATARI

```
Add: 91 DIM D$(1)
      261 PRINT "QQ"
      415 FOR X=1 TO R
      425 PRINT " "
      435 NEXT X:RETURN
```

Change: 150 PRINT "HIT THE RETURN KEY TO SEE";
190 K=INT(RND(0)*10)+2
230 L=INT(RND(0)*37)
270 M=INT(RND(0)*10)
290 R=B:GOSUB 415:PRINT "Q"
340 PRINT "QQ"
370 R=17:GOSUB 415:PRINT "HIC"
380 PRINT CHR\$(253)

COMMODORE 64

Change: 150 PRINT "HIT THE RETURN KEY TO SEE";

COMMODORE VIC-20

Change: 150 PRINT "HIT THE RETURN KEY TO SEE"
230 L=INT(RND(1)*21)

TEXAS INSTRUMENTS 99/4A

Change: 100 PRINT "UH-OH! YOUR COMPUTER HAS"
110 PRINT "BEEN TAKEN OVER BY A SNEAKY"
120 PRINT "SEA SERPENT. AND WORSE YET,"
130 PRINT "THE CREATURE HAS THE HICCUPS!"
190 K=INT(RND*10)+2
230 L=INT(RND*28)
270 M=INT(RND*10)
380 CALL SOUND(100,440,2)

TRS-80 COLOR COMPUTER

Change: 190 K=RND(10)+2
230 L=RND(30)
270 M=RND(10)
290 PRINT TAB(B-1);"Q"
380 SOUND 173,2

My Hero!

It's front-page news! It's a tale of heroism! You show the monsters your superhuman genius. Type in this program and read all about it!

Sample Run

WHAT'S YOUR FIRST NAME? CHERYL
WHAT'S YOUR LAST NAME? BINKLEY
ENTER YOUR CITY? PHILADELPHIA
ENTER YOUR STATE? INDIANA
WHAT KIND OF COMPUTER DO YOU HAVE? APPLARI
WHAT'S THE FIRST NAME OF
ONE OF YOUR PARENTS? WHITNEY
ENTER A FRIEND'S NAME? SALLY
ENTER AN ENEMY'S NAME? DARRYL
WHAT SCHOOL DO YOU ATTEND? WARBURTON
WHO IS YOUR FAVORITE TEACHER? MR. SHUSTER

EXTRA! EXTRA!

THE INDIANA GAZETTE

BINKLEY SAVES PHILADELPHIA

PHILADELPHIA--FLASH! CHERYL BINKLEY HAS
SAVED THE CITY FROM A HORRIBLE FATE.
THIS MORNING, SCALY MONSTERS FROM FAR
BENEATH THE EARTH INVADED INDIANA.
CHERYL WAS LEAVING FOR WARBURTON
SCHOOL WHEN ONE OF THE HORRIBLE
CREATURES DEMANDED TO BE TAKEN TO THE
BINKLEY HOME.

THE ANGRY INVADERS HAD BOUGHT A COMPUTER
FROM APPLARI, BUT WERE UNABLE TO
MAKE IT WORK. THEY THREATENED TO
DESTROY THE ENTIRE CITY OF PHILADELPHIA
UNLESS CHERYL COULD SHOW THEM HOW TO
MAKE THEIR APPLARI PRINT THE WORDS
"WE HATE HUMANS!" ON THE SCREEN.

(HIT THE ENTER KEY FOR PAGE 2)?

JUST AS TWO OF THESE EVIL BEINGS WERE ABOUT TO EAT WHITNEY BINKLEY FOR LUNCH, CHERYL FINISHED DEBUGGING AND RAN THE PROGRAM. THE MONSTERS WERE DELIGHTED AND INVITED CHERYL TO BE THEIR OWN PERSONAL PROGRAMMER. BINKLEY POLITELY REFUSED, BUT SUGGESTED THEY MIGHT ASK SALLY, ONE OF BINKLEY'S CLOSEST FRIENDS.

CHERYL'S TEACHER, MR. SHUSTER, PRAISED BINKLEY'S ACTIONS, AND NOTED THAT THE ONLY PERSON THE MONSTERS HAD MANAGED TO EAT WAS DARRYL, WHO GAVE THEM A NASTY CASE OF INDIGESTION.

□ Program Listing

```
10 REM MY HERO!
100 PRINT "WHAT'S YOUR FIRST NAME";
110 INPUT F$
120 IF F$="" THEN 100
130 PRINT "WHAT'S YOUR LAST NAME";
140 INPUT L$
150 IF L$="" THEN 130
160 PRINT "ENTER YOUR CITY";
170 INPUT C$
180 IF C$="" THEN 160
190 PRINT "ENTER YOUR STATE";
200 INPUT S$
210 IF S$="" THEN 190
220 PRINT "WHAT KIND OF COMPUTER DO YOU HAVE";
230 INPUT M$
240 IF M$="" THEN 220
250 PRINT "WHAT'S THE FIRST NAME OF"
260 PRINT "ONE OF YOUR PARENTS";
270 INPUT P$
280 IF P$="" THEN 250
290 PRINT "ENTER A FRIEND'S NAME";
300 INPUT B$
310 IF B$="" THEN 290
320 PRINT "ENTER AN ENEMY'S NAME";
330 INPUT E$
340 IF E$="" THEN 320
350 PRINT "WHAT SCHOOL DO YOU ATTEND";
360 INPUT Z$
370 IF Z$="" THEN 350
380 PRINT "WHO IS YOUR FAVORITE TEACHER";
390 INPUT T$
400 IF T$="" THEN 380
410 FOR A=1 TO 24
420 PRINT
430 NEXT A
```

(continued)

```

440 A$="EXTRA! EXTRA!"
450 GOSUB 990
460 GOSUB 920
470 A$="THE "+S$+" GAZETTE"
480 GOSUB 990
490 GOSUB 920
500 A$=L$+" SAVES "+C$
510 GOSUB 990
520 PRINT
530 PRINT C$; "--FLASH! ";F$;" ";L$;" HAS"
540 PRINT "SAVED THE CITY FROM A HORRIBLE FATE."
550 PRINT "THIS MORNING, SCALY MONSTERS FROM FAR"
560 PRINT "BENEATH THE EARTH INVADED ";S$;". "
570 PRINT F$;" WAS LEAVING FOR ";Z$
580 PRINT "SCHOOL WHEN ONE OF THE HORRIBLE"
590 PRINT "CREATURES DEMANDED TO BE TAKEN TO THE"
600 PRINT L$;" HOME."
610 PRINT
620 PRINT "THE ANGRY INVADERS HAD BOUGHT A COMPUTER"
630 PRINT "FROM ";M$;" , BUT WERE UNABLE TO"
640 PRINT "MAKE IT WORK. THEY THREATENED TO"
650 PRINT "DESTROY THE ENTIRE CITY OF ";C$
660 PRINT "UNLESS ";F$;" COULD SHOW THEM HOW TO"
670 PRINT "MAKE THEIR ";M$;" PRINT THE WORDS"
680 PRINT CHR$(34);"WE HATE HUMANS!";CHR$(34);" ON THE SCREEN."
690 PRINT
700 PRINT "(HIT THE ENTER KEY FOR PAGE 2)";
710 INPUT X$
720 PRINT
730 PRINT
740 PRINT
750 PRINT "JUST AS TWO OF THESE EVIL BEINGS WERE"
760 PRINT "ABOUT TO EAT ";P$;" ";L$;" FOR LUNCH,"
770 PRINT F$;" FINISHED DEBUGGING AND RAN THE"
780 PRINT "PROGRAM. THE MONSTERS WERE DELIGHTED"
790 PRINT "AND INVITED ";F$;" TO BE THEIR OWN PERSONAL"
800 PRINT "PROGRAMMER. ";L$;" POLITELY REFUSED, BUT"
810 PRINT "SUGGESTED THEY MIGHT ASK ";B$;" , ONE OF"
820 PRINT L$;" 'S CLOSEST FRIENDS."
830 PRINT
840 PRINT F$;" 'S TEACHER, ";T$;" , PRAISED"
850 PRINT L$;" 'S ACTIONS, AND NOTED THAT THE"
860 PRINT "ONLY PERSON THE MONSTERS HAD MANAGED"
870 PRINT "TO EAT WAS ";E$;" , WHO GAVE THEM A"
880 PRINT "NASTY CASE OF INDIGESTION."
890 PRINT
900 END
910 REM --- LINE DRAWING SUBROUTINE ---
920 FOR Z=1 TO 38
930 PRINT "-";
940 NEXT Z
950 PRINT
960 RETURN
970 RETURN
980 REM --- CENTERING SUBROUTINE ---

```

```
990 K=19-LEN(A$)/2
1000 PRINT TAB(K);A$
1010 RETURN
```

If You Have . . .

APPLE II

```
Add: 725 PRINT
      745 PRINT
```

```
Change: 620 PRINT "THE ANGRY INVADERS HAD BOUGHT A COMPUTER";
      700 PRINT "(HIT THE RETURN KEY FOR PAGE 2)";
```

ATARI

```
Add: 91 DIM F$(17),L$(17),C$(25),S$(17),M$(17),P$(17),
      E$(25),B$(25)
      92 DIM Z$(25),T$(25),A$(27),X$(1),AA$(11),AB$(11)
      441 AA$="":AB$=""
      1035 FOR R=1 TO K
      1045 PRINT " ";:NEXT R:RETURN
      1055 PRINT
```

```
Change: 470 A$="THE ":AA$=S$:AB$=" GAZETTE"
      500 A$=L$:AA$=" SAVES":AB$=C$
      700 PRINT "(HIT THE RETURN KEY FOR PAGE 2)";
      990 K=19-(LEN(A$)+LEN(AA$)+LEN(AB$))/2
      1000 GOSUB 1035:PRINT A$;AA$;AB$
```

COMMODORE 64

```
Change: 700 PRINT "(HIT THE RETURN KEY FOR PAGE 2)";
      790 PRINT "AND INVITED ";F$;" TO BE THEIR PERSONAL"
```

COMMODORE VIC-20

```
Add: 604 PRINT "(HIT RETURN FOR PAGE 2) "
      605 INPUT X$
      611 PRINT
      612 PRINT
      613 PRINT
      614 PRINT
      705 PRINT
      835 PRINT "(HIT RETURN FOR PAGE 4) "
      836 INPUT X$
      837 FOR I=1 TO 5
      838 PRINT
```

(continued)

```
839 NEXT I
891 PRINT
892 PRINT
```

```
Change: 100 PRINT "WHAT'S YOUR FIRST NAME"
130 PRINT "WHAT'S YOUR LAST NAME"
160 PRINT "ENTER YOUR CITY"
190 PRINT "ENTER YOUR STATE"
220 PRINT "WHAT KIND OF COMPUTER DO YOU HAVE"
260 PRINT "ONE OF YOUR PARENTS"
290 PRINT "ENTER A FRIEND'S NAME"
320 PRINT "ENTER AN ENEMY'S NAME"
350 PRINT "WHAT SCHOOL DO YOU ATTEND"
380 PRINT "WHO IS YOUR FAVORITE TEACHER"
700 PRINT "(HIT RETURN FOR PAGE 3)"
720 FOR I=1 TO 5
740 NEXT I
790 PRINT "AND INVITED ";F$;" TO BE THEIR PERSONAL"
920 FOR Z=1 TO 20
990 K=10-LEN(A$)/2
```

TEXAS INSTRUMENTS 99/4A

```
Add: 615 PRINT "(HIT THE ENTER KEY FOR PAGE 2)";
617 INPUT X$
835 PRINT "(HIT THE ENTER KEY FOR PAGE 4)";
837 INPUT X$
```

```
Change: 470 A$="THE "&S$&" GAZETTE"
500 A$=L$&" SAVES "&C$
700 PRINT "(HIT THE ENTER KEY FOR PAGE 3)";
920 FOR Z=1 TO 28
990 K=14-LEN(A$)/2
```

TRS-80 COLOR COMPUTER

```
Delete: 420,520
```

```
Add: 615 PRINT "(HIT ENTER KEY FOR PAGE 2)";
616 INPUT X$
617 PRINT
618 PRINT
619 PRINT
831 PRINT "(HIT ENTER KEY FOR PAGE 4)";
832 INPUT X$
833 PRINT
834 PRINT
```

```
Change: 410 FOR A=1 TO 15
530 PRINT C$;"--FLASH! ";F$;" ";L$;" HAS ";
540 PRINT "SAVED THE CITY FROM A HORRIBLE FATE. ";
550 PRINT "THIS MORNING, SCALY MONSTERS FROM FAR ";
560 PRINT "BENEATH THE EARTH INVADED ";S$;" . ";
```

```
580 PRINT "SCHOOL WHEN ONE OF THE HORRIBLE ";
590 PRINT "CREATURES DEMANDED TO BE TAKEN TO THE ";
620 PRINT "THE ANGRY INVADERS HAD BOUGHT A
    COMPUTER ";
630 PRINT "FROM ";M$;" , BUT WERE UNABLE TO ";
640 PRINT "MAKE IT WORK. THEY THREATENED TO ";
650 PRINT "DESTROY THE ENTIRE CITY OF ";C$;" ";
660 PRINT "UNLESS ";F$;" COULD SHOW THEM HOW TO ";
670 PRINT "MAKE THEIR ";M$;" PRINT THE WORDS ";
700 PRINT "(HIT ENTER KEY FOR PAGE 3)";
750 PRINT "JUST AS TWO OF THESE EVIL BEINGS WERE ";
760 PRINT "ABOUT TO EAT ";P$;" ";L$;" FOR LUNCH, ";
770 PRINT F$;" FINISHED DEBUGGING AND RAN THE ";
780 PRINT "PROGRAM. THE MONSTERS WERE DELIGHTED ";
790 PRINT "AND INVITED ";F$;" TO BE THEIR OWN
    PERSONAL ";
800 PRINT "PROGRAMMER. ";L$;" POLITELY REFUSED,
    BUT ";
810 PRINT "SUGGESTED THEY MIGHT ASK ";B$;" ,
    ONE OF ";
840 PRINT F$;" 'S TEACHER, ";T$;" , PRAISED ";
850 PRINT L$;" 'S ACTIONS, AND NOTED THAT THE ";
860 PRINT "ONLY PERSON THE MONSTERS HAD MANAGED ";
870 PRINT "TO EAT WAS ";E$;" , WHO GAVE THEM A ";
880 PRINT "NASTY CASE OF INDIGESTION. ";
920 FOR Z=1 TO 30
990 K=15-LEN(A$)/2
```


When's the Party?

Want to know how many minutes till Halloween? The goblins and ghouls and horrible hairy creatures use their personal computers to keep track. Now you can, too!

Sample Run

```
HOW
  LONG
    TILL
      HALLOWEEN?
```

```
WHAT TIME IS IT? 4:57
AM OR PM? PM
WHAT'S THE NUMBER OF THE MONTH (1-12)? 5
WHAT'S THE DATE (1- 31 )? 24
```

HALLOWEEN IS 160 DAYS AWAY.

THERE ARE JUST
229383 MINUTES TILL HALLOWEEN.

BOO!

Program Listing

```
10 REM WHEN'S THE PARTY?
100 DIM ML(12)
110 PRINT
120 PRINT "HOW"
130 PRINT "    LONG"
140 PRINT "    TILL"
150 PRINT "    HALLOWEEN?"
160 PRINT
170 PRINT "WHAT TIME IS IT";
180 INPUT CT$
190 FOR I=2 TO LEN(CT$)
200 IF MID$(CT$,I,1)=":" THEN DD=I
210 NEXT I
220 IF DD=0 THEN 310
230 H$=MID$(CT$,1,DD-1)
240 MN$=MID$(CT$,DD+1,DD+3)
250 H=VAL(H$)
```

```

260 M=VAL(MN$)
270 IF H>12 OR H<1 THEN 320
280 IF H=12 THEN H=0
290 IF M<0 OR M>59 THEN 320
300 GOTO 340
310 PRINT "TIME MUST INCLUDE A COLON (:)"
320 PRINT "TRY AGAIN: ";
330 GOTO 170
340 PRINT "AM OR PM";
350 INPUT PM$
360 A$=MID$(PM$,1,1)
370 IF A$="A" THEN 410
380 IF A$="P" THEN 410
390 PRINT "TRY AGAIN: ";
400 GOTO 340
410 IF A$="P" THEN H=H+12
420 M=60-M
430 H=23-H
440 TM=M+(60*H)
450 PRINT "WHAT'S THE NUMBER OF THE MONTH (1-12)";
460 INPUT MO
470 IF MO>0 AND MO<13 THEN 500
480 PRINT "TRY AGAIN:"
490 GOTO 450
500 FOR I=1 TO 12
510 READ ML(I)
520 NEXT I
530 PRINT "WHAT'S THE DATE (1-";ML(MO);")";
540 INPUT DAY
550 IF DAY<ML(MO)+1 AND DAY>0 THEN 580
560 PRINT "TRY AGAIN:"
570 GOTO 530
580 DT=ML(MO)-DAY
590 IF MO=10 AND DAY=31 THEN 770
600 IF MO=0 THEN 660
610 IF MO=11 THEN TD=31
620 IF MO=11 OR MO=12 THEN MO=0
630 FOR I=10 TO MO+1 STEP -1
640 TD=TD+ML(I)
650 NEXT I
660 TD=TD+DT-1
670 FD=24*60*TD
680 FT=FD+TM
690 PRINT
700 PRINT "HALLOWEEN IS ";TD+1; " DAYS AWAY."
710 PRINT
720 PRINT "THERE ARE JUST"
730 PRINT FT;" MINUTES TILL HALLOWEEN."
740 PRINT
750 PRINT "          BOO!"
760 END
770 PRINT "THAT'S HALLOWEEN, YOU MONSTER!"
780 PRINT "          BOO!"
790 END
800 DATA 31,28,31,30,31,30,31,31,30,31,30,31

```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM CT\$(6),H\$(2),MN\$(6),PM\$(2),A\$(1)
201 IF CT\$(3,3) = ":" THEN DD=2
511 ML(I)=X

Change: 190 REM
200 IF CT\$(2,2) = ":" THEN DD=1
210 REM
230 H\$=CT\$(1,DD)
240 MN\$=CT\$(DD+2,DD+3)
360 A\$=PM\$(1,1)
510 READ X

COMMODORE 64

Delete: 180, 190, 200, 210, 220, 230, 240, 250, 260, 310

Add: 175 PRINT "WHAT'S THE HOUR";
185 INPUT H
195 PRINT "WHAT'S THE NUMBER OF MINUTES"
205 PRINT "PAST THE HOUR";
215 INPUT M

Change: 170 PRINT "WHAT TIME IS IT--"

COMMODORE VIC-20

Delete: 180, 190, 200, 210, 220, 230, 240, 250, 260, 310

Add: 175 PRINT "WHAT'S THE HOUR"
185 INPUT H
195 PRINT "WHAT'S THE NUMBER OF MINUTES"
205 PRINT "PAST THE HOUR"
215 INPUT M

Change: 170 PRINT "WHAT TIME IS IT--"
340 PRINT "AM OR PM"
450 PRINT "WHAT'S THE NUMBER OF THE MONTH (1-12)"
530 PRINT "WHAT'S THE DATE (1-";ML(MO);")"

TEXAS INSTRUMENTS 99/4A

Add: 202 DD=I
282 H=0

```
412 H=H+12
612 TD=31
624 GOTO 630
626 MO=0
```

```
Change: 200 IF SEG$(CT$,I,1)<>":" THEN 210
230 H$=SEG$(CT$,1,DD-1)
240 MN$=SEG$(CT$,DD+1,DD+3)
270 IF (H>12) + (H<1) THEN 320
280 IF (H<>12) THEN 290
290 IF (M<0) + (M>59) THEN 320
360 A$=SEG$(PM$,1,1)
410 IF A$<>"P" THEN 420
450 PRINT "WHAT'S THE NUMBER OF THE      MONTH (1-12) "
470 IF (MO>0) * (MO<13) THEN 500
550 IF (DAY<ML(MO)+1) * (DAY>0) THEN 580
590 IF (MO=10) * (DAY=31) THEN 770
610 IF MO<>11 THEN 620
620 IF (MO=11) + (MO=12) THEN 626
770 PRINT "THAT'S HALLOWEEN, YOU      MONSTER!"
```

TRS-80 COLOR COMPUTER

Change: 180 LINE INPUT CT\$

Humans Keep Out!

The Monster Construction Company strikes again! This time they'll build you a wall that's guaranteed human-proof!

Sample Run

STILL GOT HUMAN TROUBLE? WHAT
YOU NEED IS OUR FINEST WALL!
AT MONSTER CONSTRUCTION COMPANY
OUR MOTTO IS "WE NEVER QUIT!"

HIT THE ENTER KEY TO BEGIN BUILDING?

```
*****  
*****  
*****
```

```
****  
****  
****
```

```
*****  
*****  
*****
```

```
****  
****  
****
```

```
*****  
*****  
*****
```

```
****  
****  
****
```

```
*****  
*****  
*****
```

```
****  
****  
****
```

```
*****  
*****  
*****
```

```
****  
****  
****
```

Break

Program Listing

```
10 REM HUMANS KEEP OUT!
100 PRINT "STILL GOT HUMAN TROUBLE? WHAT"
110 PRINT "YOU NEED IS OUR FINEST WALL!"
120 PRINT "AT MONSTER CONSTRUCTION COMPANY"
130 PRINT "OUR MOTTO IS ";CHR$(34);" WE NEVER QUIT!";
    CHR$(34)
140 PRINT
150 PRINT "HIT THE ENTER KEY TO BEGIN BUILDING";
160 INPUT X$
170 PRINT
180 B$="*****"
190 L$="****"
200 REM --- WHOLE BRICKS ---
210 FOR D=1 TO 3
220 FOR A=1 TO 4
230 IF A<4 THEN 260
240 PRINT B$
250 GOTO 270
260 PRINT B$;" ";
270 NEXT A
280 NEXT D
290 PRINT
300 REM --- HALVES AND WHOLES ---
310 FOR E=1 TO 3
320 FOR A=1 TO 5
330 IF A<>5 THEN 360
340 PRINT L$
350 GOTO 400
360 IF A>1 THEN 390
370 PRINT L$;" ";
380 GOTO 400
390 PRINT B$;" ";
400 NEXT A
410 NEXT E
420 PRINT
430 GOTO 210
```

If You Have . . .

APPLE II

Change: 150 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";

ATARI

Add: 91 DIM X\$(1),B\$(9),L\$(4)

(continued)

```
Change: 150 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";
        180 B$="*****"
        190 L$="***"
        340 PRINT L$;"*"
```

COMMODORE 64

```
Change: 150 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING";
```

COMMODORE VIC-20

```
Add: 275 PRINT " ";
      345 PRINT " ";
```

```
Change: 150 PRINT "HIT THE RETURN KEY TO BEGIN BUILDING"
        220 FOR A=1 TO 2
        320 FOR A=1 TO 3
        330 IF A<>3 THEN 360
        340 PRINT L$;
```

TEXAS INSTRUMENTS 99/4A

```
Change: 180 B$="*****"
        220 FOR A=1 TO 3
        230 IF A<3 THEN 260
        320 FOR A=1 TO 4
        330 IF A<>4 THEN 360
```

TRS-80 COLOR COMPUTER

```
Change: 220 FOR A=1 TO 3
        230 IF A<3 THEN 260
        320 FOR A=1 TO 4
        330 IF A<>4 THEN 360
```

Werewolf Computer

It's not easy being a werewolf! Every time there's a full moon, you change into a human — and back again. In a lifetime that can really add up. Run this program and see!

Sample Run

IS THE WEREWOLF MALE OR FEMALE? FEMALE

THE WEREWOLF CHANGES TO A HUMAN
AND BACK AGAIN EVERY FULL MOON.

SHE WANTS TO KNOW HOW MANY TIMES
IN HER LIFE SHE'S CHANGED BACK
AND FORTH.

HOW MANY YEARS OLD IS THE WEREWOLF? 12
THE WEREWOLF HAS CHANGED INTO
A HUMAN AND BACK AGAIN 160 TIMES.

SHE'S SICK AND TIRED OF IT, TOO!

Program Listing

```
10 REM WEREWOLF COMPUTER
100 PRINT "IS THE WEREWOLF MALE OR FEMALE";
110 INPUT M$
120 T$=MID$(M$,1,1)
130 IF T$<>"F" AND T$<>"M" THEN 100
140 PRINT
150 PRINT "THE WEREWOLF CHANGES TO A HUMAN"
160 PRINT "AND BACK AGAIN EVERY FULL MOON."
170 PRINT
180 IF T$="F" THEN PRINT "S";
190 PRINT "HE WANTS TO KNOW HOW MANY TIMES"
200 PRINT "IN ";
210 IF T$="F" THEN PRINT "HER";" LIFE S";
220 IF T$="M" THEN PRINT "HIS";" LIFE ";
230 PRINT "HE'S CHANGED BACK"
240 PRINT "AND FORTH."
250 PRINT
260 PRINT "HOW MANY YEARS OLD IS THE WEREWOLF";
270 INPUT AGE
280 IF AGE<1 THEN 260
290 TIMES=INT(AGE*365/27.3215)
```

(continued)


```

300 PRINT "THE WEREWOLF HAS CHANGED INTO"
310 PRINT "A HUMAN AND BACK AGAIN ";TIMES;" TIMES."
320 PRINT
330 IF T$="F" THEN PRINT "S";
340 PRINT "HE'S SICK AND TIRED OF IT, TOO!"

```

If You Have . . .

APPLE II No Changes Required

ATARI

Add: 91 DIM M\$(6),T\$(1)

Change: 120 T\$=M\$(1,1)

COMMODORE 64

Change: 290 TMES=INT(AGE*365/27.3215)
310 PRINT "A HUMAN AND BACK AGAIN ";TMES;" TIMES."

COMMODORE VIC-20

Change: 100 PRINT "IS THE WEREWOLF MALE OR FEMALE"
260 PRINT "HOW MANY YEARS OLD IS THE WEREWOLF"
290 TMES=INT(AGE*365/27.3215)
310 PRINT "A HUMAN AND BACK AGAIN ";TMES;" TIMES."

TEXAS INSTRUMENTS 99/4A

Add: 182 PRINT "S";
212 PRINT "HER";" LIFE S";
222 PRINT "HIS";" LIFE ";
332 PRINT "S";

Change: 100 PRINT "IS THE WEREWOLF MALE OR FEMALE";
120 T\$=SEG\$(M\$,1,1)
130 IF (T\$<>"F") * (T\$<>"M") THEN 100
150 PRINT "THE WEREWOLF CHANGES TO A"
160 PRINT "HUMAN AND BACK AGAIN EVERY FULL MOON."
180 IF T\$<>"F" THEN 190
190 PRINT "HE WANTS TO KNOW HOW MANY"
200 PRINT "TIMES IN "
210 IF T\$<>"F" THEN 220
220 IF T\$<>"M" THEN 230
230 PRINT "HE'S ";" CHANGED BACK ";
260 PRINT "HOW MANY YEARS OLD IS THE WEREWOLF";

```
300 PRINT "THE WEREWOLF HAS CHANGED"  
310 PRINT "INTO A HUMAN AND BACK AGAIN"; TIMES;  
    " TIMES."  
330 IF T$<>"F" THEN 340  
340 PRINT "HE'S SICK AND "; " TIRED OF IT, TOO!"
```

TRS-80 COLOR COMPUTER No Changes Required

Escape from Monster Caverns

Okay, explorer — pack all your wits! No one has entered Monster Caverns and come out alive. Perhaps you will be the first. You'll need nerves of steel, the memory of an elephant, and the program below — and even then you may not survive!

You may want to draw a map of the caverns as you travel through them. And here's one clue we've learned: The rooms of Monster Caverns aren't all on the same level. Twisting passageways sometimes connect the rooms. That means you might have to go south to get from, say, Room 6 to Room 5 — and south from Room 5 to Room 6!

Sample Run

YOU CAN PLAY THIS GAME IN 3 VERSIONS:

- 1--HARD
- 2--MEDIUM
- 3--EASY

WHICH ONE DO YOU WANT? 2

YOU ARE A FAMOUS EXPLORER WHO HAS SPENT YEARS PREPARING FOR A GREAT EXPEDITION TO THE ISLAND OF PIHC, AN UNCHARTED DOT OF LAND THOUSANDS OF MILES FROM CIVILIZATION. YOUR RESEARCH HAS LED TO A DUSTY OLD BOOK WITH A SECRET MAP SEWN INSIDE THE BINDING. YOU'VE FOLLOWED THE MAP AND HAVE TRAVELED THROUGH THE THICK JUNGLE OF PIHC. NOW YOU ARE DEEP INSIDE THE MAZE OF TUNNELS BENEATH ITS SURFACE.

YOUR NATIVE GUIDES REFUSED TO GO THIS FAR, BECAUSE THEY HAD HEARD THE CAVE CONTAINED TERRIBLE SCARY MONSTERS. THEY SAID THERE WAS ONLY ONE WAY TO ESCAPE FROM THOSE HORRIBLE CREATURES AND THAT WAS TO FIND THE GIANT RUBY IDOL'S EYE HIDDEN IN THE DARKEST CHAMBER OF THE CAVE.

(HIT THE ENTER KEY FOR MORE)?

THE GUIDES TOLD YOU THAT THE GIANT RUBY WAS JUST PART OF A VAST TREASURE. SEVERAL OTHERS HAD FOUND THE CAVE, AND ONE HAD EVEN LOCATED THE TREASURE. BUT THE MONSTERS DID NOT LET THEM ESCAPE. THE EXPLORERS WERE NOT ABLE TO FIND THE GIANT RUBY THAT COULD SAVE THEM.

ONE MORE THING: SOMEWHERE IN THE MAZE OF TUNNELS IS A MAGIC WORD WRITTEN BY AN OLD VICTIM. (HE DIED HORRIBLY BECAUSE HE FORGOT IT.) THE WORD CHANGES FROM TIME TO TIME, SO BE CAREFUL! THE MONSTER THAT PROWLs THE CAVE WILL SMOTHER YOU IN SLIME UNLESS YOU REMEMBER IT!! FINALLY, AT THE END, YOU MUST GET TO ROOM 1 TO LEAVE THE CAVE. GOOD LUCK!

HOW MANY GUIDES DID YOU HAVE? 7

=====

: THE ONLY DIRECTIONS YOU CAN GIVE :
: ARE: "N", "E", "S", OR "W" :
=====

\$\$\$\$\$\$\$\$\$\$\$\$ INVENTORY \$\$\$\$\$\$\$\$\$\$\$\$\$

-- NOTHING BUT YOUR WITS --

\$

-----> YOU ARE IN ROOM 1 <-----
WHICH WAY DO YOU WANT TO GO? N
SORRY, THERE IS NO DOOR HERE.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 1 <-----
WHICH WAY DO YOU WANT TO GO? S
FEROCIOUS GILA MONSTERS BAR THE WAY.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 1 <-----
WHICH WAY DO YOU WANT TO GO? E

-----> YOU ARE IN ROOM 2 <-----
WHICH WAY DO YOU WANT TO GO? S
SORRY, THERE IS NO DOOR HERE.
YOU CAN'T GO THIS WAY.

(continued)

-----> YOU ARE IN ROOM 2 <-----
WHICH WAY DO YOU WANT TO GO? E

-----> YOU ARE IN ROOM 3 <-----
WHICH WAY DO YOU WANT TO GO? S

-----> YOU ARE IN ROOM 6 <-----
WHICH WAY DO YOU WANT TO GO? S

QQ

YOU SEE A WORD WRITTEN IN BLOOD ON THE
WALL: FTIYGLB REMEMBER THIS WORD--
IT MAY JUST SAVE YOUR LIFE!!!

TT

-----> YOU ARE IN ROOM 9 <-----
* YOU'VE FOUND THE RUBY TO ESCAPE!! *
(NOW TRY TO LOCATE THE VAST TREASURE)

WHICH WAY DO YOU WANT TO GO? N

-----> YOU ARE IN ROOM 6 <-----
WHICH WAY DO YOU WANT TO GO? E
AHEAD ARE SHEETS OF UNPASSABLE FLAME.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 6 <-----
WHICH WAY DO YOU WANT TO GO? W

-----> YOU ARE IN ROOM 5 <-----
** YOU'VE FOUND ALL THE TREASURE!!! **
***** YOU DID IT!!! *****
YOU FOUND THE TREASURE--AND ALSO FOUND
THE GIANT RUBY TO LET YOU ESCAPE !!!!!
NOW JUST RETURN TO ROOM 1 TO GET BACK!

WHICH WAY DO YOU WANT TO GO? S
THE WALLS START TO CRUMBLE AROUND YOU,
BUT YOU NARROWLY ESCAPE.

-----> YOU ARE IN ROOM 5 <-----
WHICH WAY DO YOU WANT TO GO? N
YOU ALMOST WALK INTO A CLOUD OF POISON.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 5 <-----
WHICH WAY DO YOU WANT TO GO? E

-----> YOU ARE IN ROOM 6 <-----
WHICH WAY DO YOU WANT TO GO? N

-----> YOU ARE IN ROOM 3 <-----
WHICH WAY DO YOU WANT TO GO? N
FEROCIOUS GILA MONSTERS BAR THE WAY.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 3 <-----
WHICH WAY DO YOU WANT TO GO? W

-----> YOU ARE IN ROOM 2 <-----
WHICH WAY DO YOU WANT TO GO? S
SORRY, THERE IS NO DOOR HERE.
YOU CAN'T GO THIS WAY.

-----> YOU ARE IN ROOM 2 <-----
WHICH WAY DO YOU WANT TO GO? W

!!

YOU DID IT!!!! YOU FOUND THE TREASURE
AND THE GIANT RUBY--AND YOU MADE IT
BACK TO ROOM 1 TO GET OUT OF THE MAZE!

!!

Program Listing

```
10 REM ESCAPE FROM MONSTER CAVERNS
100 PRINT "YOU CAN PLAY THIS GAME IN 3 VERSIONS:"
110 PRINT
120 PRINT TAB(5);"1--HARD"
130 PRINT TAB(5);"2--MEDIUM"
140 PRINT TAB(5);"3--EASY"
150 PRINT
160 PRINT "WHICH ONE DO YOU WANT";
170 INPUT DI
180 IF DI<1 THEN 160
190 IF DI>3 THEN 160
200 PRINT CHR$(12)
210 PRINT "YOU ARE A FAMOUS EXPLORER WHO HAS"
220 PRINT "SPENT YEARS PREPARING FOR A GREAT"
230 PRINT "EXPEDITION TO THE ISLAND OF PIHC,"
240 PRINT "AN UNCHARTED DOT OF LAND THOUSANDS"
250 PRINT "OF MILES FROM CIVILIZATION. YOUR"
260 PRINT "RESEARCH HAS LED TO A DUSTY OLD"
270 PRINT "BOOK WITH A SECRET MAP SEWN INSIDE"
280 PRINT "THE BINDING. YOU'VE FOLLOWED THE"
290 PRINT "MAP AND HAVE TRAVELED THROUGH THE"
300 PRINT "THICK JUNGLE OF PIHC. NOW YOU"
310 PRINT "ARE DEEP INSIDE THE MAZE OF"
320 PRINT "TUNNELS BENEATH ITS SURFACE."
330 PRINT
```

(continued)

```

340 PRINT "YOUR NATIVE GUIDES REFUSED TO GO"
350 PRINT "THIS FAR, BECAUSE THEY HAD HEARD"
360 PRINT "THE CAVE CONTAINED TERRIBLE SCARY"
370 PRINT "MONSTERS. THEY SAID THERE WAS"
380 PRINT "ONLY ONE WAY TO ESCAPE FROM"
390 PRINT "THOSE HORRIBLE CREATURES AND"
400 PRINT "THAT WAS TO FIND THE GIANT RUBY"
410 PRINT "IDOL'S EYE HIDDEN IN THE DARKEST"
420 PRINT "CHAMBER OF THE CAVE."
430 PRINT
440 PRINT "(HIT THE ENTER KEY FOR MORE)";
450 INPUT X$
460 PRINT CHR$(12)
470 PRINT "THE GUIDES TOLD YOU THAT THE"
480 PRINT "GIANT RUBY WAS JUST PART OF A"
490 PRINT "VAST TREASURE. SEVERAL OTHERS"
500 PRINT "HAD FOUND THE CAVE, AND ONE HAD"
510 PRINT "EVEN LOCATED THE TREASURE. BUT"
520 PRINT "THE MONSTERS DID NOT LET THEM"
530 PRINT "ESCAPE. THE EXPLORERS WERE NOT"
540 PRINT "ABLE TO FIND THE GIANT RUBY"
550 PRINT "THAT COULD SAVE THEM."
560 PRINT
570 PRINT "ONE MORE THING: SOMEWHERE IN"
580 PRINT "THE MAZE OF TUNNELS IS A MAGIC"
590 PRINT "WORD WRITTEN BY AN OLD VICTIM."
600 PRINT "(HE DIED HORRIBLY BECAUSE HE"
610 PRINT "FORGOT IT.) THE WORD CHANGES"
620 PRINT "FROM TIME TO TIME, SO BE CAREFUL!"
630 PRINT "THE MONSTER THAT PROWLs THE CAVE"
640 PRINT "WILL SMOTHER YOU IN SLIME UNLESS"
650 PRINT "YOU REMEMBER IT!! FINALLY, AT"
660 PRINT "THE END, YOU MUST GET TO ROOM 1"
670 PRINT "TO LEAVE THE CAVE. GOOD LUCK!"
680 PRINT
690 Q$=CHR$(34)
700 PRINT "HOW MANY GUIDES DID YOU HAVE";
710 INPUT LOST
720 IF LOST<1 THEN 700
730 RANDOMIZE(LOST)
740 PRINT CHR$(12)
750 GOSUB 3240
760 PRINT
770 FIND=0
780 TR=INT(RND(1)*15)+3
790 MG=INT(RND(1)*15)+3
800 IF MG=TR THEN 790
810 REM -----
820 ROOM=1
830 GOSUB 3870
840 GOSUB 2840
850 IF I$="E" THEN 920
860 IF I$="W" THEN 1370
870 IF I$="N" THEN GOSUB 2360
880 IF I$="S" THEN GOSUB 2360

```

```

890 GOSUB 3120
900 GOTO 840
910 REM -----
920 ROOM=2
930 GOSUB 2840
940 IF I$="E" THEN 1010
950 IF I$="W" THEN 820
960 IF I$="N" THEN 2270
970 IF I$="S" THEN GOSUB 2360
980 GOSUB 3120
990 GOTO 930
1000 REM -----
1010 ROOM=3
1020 GOSUB 2840
1030 IF I$="E" THEN GOSUB 2360
1040 IF I$="W" THEN 920
1050 IF I$="N" THEN GOSUB 2360
1060 IF I$="S" THEN 1280
1070 GOSUB 3120
1080 GOTO 1020
1090 REM -----
1100 ROOM=4
1110 GOSUB 2840
1120 IF I$="E" THEN 1190
1130 IF I$="W" THEN GOSUB 2360
1140 IF I$="N" THEN GOSUB 2360
1150 IF I$="S" THEN GOSUB 2360
1160 GOSUB 3120
1170 GOTO 1110
1180 REM -----
1190 ROOM=5
1200 GOSUB 2840
1210 IF I$="E" THEN 1280
1220 IF I$="W" THEN 1100
1230 IF I$="N" THEN GOSUB 2360
1240 IF I$="S" THEN GOSUB 2360
1250 GOSUB 3120
1260 GOTO 1200
1270 REM -----
1280 ROOM=6
1290 GOSUB 2840
1300 IF I$="E" THEN GOSUB 2360
1310 IF I$="W" THEN 1190
1320 IF I$="N" THEN 1010
1330 IF I$="S" THEN 1550
1340 GOSUB 3120
1350 GOTO 1290
1360 REM -----
1370 ROOM=7
1380 GOSUB 2840
1390 IF I$="E" THEN 1460
1400 IF I$="W" THEN 820
1410 IF I$="N" THEN GOSUB 2360
1420 IF I$="S" THEN GOSUB 2360
1430 GOSUB 3120

```

(continued)


```

1440 GOTO 1380
1450 REM -----
1460 ROOM=8
1470 GOSUB 2840
1480 IF I$="E" THEN GOSUB 2360
1490 IF I$="W" THEN 1370
1500 IF I$="N" THEN GOSUB 2360
1510 IF I$="S" THEN 1730
1520 GOSUB 3120
1530 GOTO 1470
1540 REM -----
1550 ROOM=9
1560 GOSUB 2840
1570 IF I$="E" THEN GOSUB 2360
1580 IF I$="W" THEN GOSUB 2360
1590 IF I$="N" THEN 1280
1600 IF I$="S" THEN 1820
1610 GOSUB 3120
1620 GOTO 1560
1630 REM -----
1640 ROOM=10
1650 GOSUB 2840
1660 IF I$="E" THEN 1730
1670 IF I$="W" THEN GOSUB 2360
1680 IF I$="N" THEN GOSUB 2360
1690 IF I$="S" THEN 1910
1700 GOSUB 3120
1710 GOTO 1650
1720 REM -----
1730 ROOM=11
1740 GOSUB 2840
1750 IF I$="E" THEN GOSUB 2360
1760 IF I$="W" THEN 1640
1770 IF I$="N" THEN 1460
1780 IF I$="S" THEN GOSUB 2360
1790 GOSUB 3120
1800 GOTO 1740
1810 REM -----
1820 ROOM=12
1830 GOSUB 2840
1840 IF I$="E" THEN 2090
1850 IF I$="W" THEN GOSUB 2360
1860 IF I$="N" THEN 1550
1870 IF I$="S" THEN 2000
1880 GOSUB 3120
1890 GOTO 1830
1900 REM -----
1910 ROOM=13
1920 GOSUB 2840
1930 IF I$="E" THEN GOSUB 2360
1940 IF I$="W" THEN GOSUB 2360
1950 IF I$="N" THEN 1640
1960 IF I$="S" THEN 2090
1970 GOSUB 3120
1980 GOTO 1920

```

```

1990 REM -----
2000 ROOM=14
2010 GOSUB 2840
2020 IF I$="E" THEN GOSUB 2360
2030 IF I$="W" THEN GOSUB 2360
2040 IF I$="N" THEN 1820
2050 IF I$="S" THEN GOSUB 2360
2060 GOSUB 3120
2070 GOTO 2010
2080 REM -----
2090 ROOM=15
2100 GOSUB 2840
2110 IF I$="E" THEN 1910
2120 IF I$="W" THEN 1820
2130 IF I$="N" THEN 2180
2140 IF I$="S" THEN GOSUB 2360
2150 GOSUB 3120
2160 GOTO 2100
2170 REM -----
2180 ROOM=16
2190 GOSUB 2840
2200 IF I$="E" THEN 2270
2210 IF I$="W" THEN GOSUB 2360
2220 IF I$="N" THEN GOSUB 2360
2230 IF I$="S" THEN 2090
2240 GOSUB 3120
2250 GOTO 2190
2260 REM -----
2270 ROOM=17
2280 GOSUB 2840
2290 IF I$="E" THEN GOSUB 2360
2300 IF I$="W" THEN 2180
2310 IF I$="N" THEN 920
2320 IF I$="S" THEN GOSUB 2360
2330 GOSUB 3120
2340 GOTO 2280
2350 REM ----- DEAD END SUBROUTINE -----
2360 ON DI GOTO 2370,2380,2390
2370 ON INT(RND(1)*5)+1 GOTO 2400,2430,2520,2550,2580
2380 ON INT(RND(1)*6)+1 GOTO 2400,2460,2490,2520,2550,2580
2390 ON INT(RND(1)*7)+1 GOTO 2400,2430,2460,2490,2520,2550,
2580
2400 PRINT "SORRY, THERE IS NO DOOR HERE."
2410 PRINT "YOU CAN'T GO THIS WAY."
2420 RETURN
2430 PRINT "THIS WHOLE WALL IS OOZING ACID."
2440 PRINT "YOU CAN'T GO THIS WAY."
2450 RETURN
2460 PRINT "THIS DOOR HAS SUDDENLY CLOSED."
2470 PRINT "YOU CAN'T GO THIS WAY."
2480 RETURN
2490 PRINT "FEROCIOUS GILA MONSTERS BAR THE WAY."
2500 PRINT "YOU CAN'T GO THIS WAY."
2510 RETURN

```

(continued)

```

2520 PRINT "AHEAD ARE SHEETS OF UNPASSABLE FLAME."
2530 PRINT "YOU CAN'T GO THIS WAY."
2540 RETURN
2550 PRINT "YOU ALMOST WALK INTO A CLOUD OF POISON."
2560 PRINT "YOU CAN'T GO THIS WAY."
2570 RETURN
2580 ON INT(RND(1)*4)+1 GOTO 2590,2650,2710,2770
2590 PRINT "THE WALLS START TO CRUMBLE AROUND YOU,"
2600 ON INT(RND(1)*6)+1 GOTO 2610,2610,2610,2610,2610,2630
2610 PRINT "BUT YOU NARROWLY ESCAPE."
2620 RETURN
2630 PRINT "AND CRUSH YOU INTO DUST. THE END."
2640 END
2650 PRINT "YOU FALL INTO THE SCUM BUGS' PIT"
2660 ON INT(RND(1)*6)+1 GOTO 2670,2670,2670,2670,2670,2690
2670 PRINT "BUT MANAGE TO PULL YOURSELF TO SAFETY."
2680 RETURN
2690 PRINT "AND ARE SAVAGELY BITTEN. THE END."
2700 END
2710 PRINT "HISSING SLIME CREATURES BLOCK THE WAY,"
2720 ON INT(RND(1)*6)+1 GOTO 2730,2730,2730,2730,2730,2750
2730 PRINT "BUT YOU FREEZE AND THEY IGNORE YOU."
2740 RETURN
2750 PRINT "AND TURN YOU INTO JELLY. THE END."
2760 END
2770 PRINT "A TORNADO OF SAWFLIES BUZZES AT YOU,"
2780 ON INT(RND(1)*6)+1 GOTO 2790,2790,2790,2790,2790,2810
2790 PRINT "BUT YOU DUCK SAFELY BEHIND FALLEN ROCK."
2800 RETURN
2810 PRINT "AND SLASHES YOU TO SHREDS. THE END"
2820 END
2830 REM ----- RUBY SUBROUTINE -----
2840 PRINT
2850 GOSUB 3480
2860 PRINT " -----> YOU ARE IN ROOM";ROOM;"<-----"
2870 IF ROOM<>TR THEN 2950
2880 IF F1=1 THEN 3030
2890 PRINT "*** YOU'VE FOUND ALL THE TREASURE!!! **"
2900 GOSUB 3410
2910 F1=1
2920 IF F2=1 THEN 3060
2930 PRINT "(NOW TRY TO FIND THE RUBY TO ESCAPE)"
2940 PRINT
2950 IF ROOM<>MG THEN 3030
2960 IF F2=1 THEN 3030
2970 PRINT "* YOU'VE FOUND THE RUBY TO ESCAPE!! *"
2980 GOSUB 3410
2990 F2=1
3000 IF F1=1 THEN 3060
3010 PRINT "(NOW TRY TO LOCATE THE VAST TREASURE)"
3020 PRINT
3030 PRINT "WHICH WAY DO YOU WANT TO GO";
3040 INPUT I$
3050 RETURN
3060 PRINT "***** YOU DID IT!!! *****"

```

```

3070 PRINT "YOU FOUND THE TREASURE--AND ALSO FOUND"
3080 PRINT "THE GIANT RUBY TO LET YOU ESCAPE !!!!!"
3090 PRINT "NOW JUST RETURN TO ROOM 1 TO GET BACK!"
3100 GOTO 3020
3110 REM ----- DIRECTION SUBROUTINE -----
3120 IF I$="N" OR I$="E" OR I$="W" OR I$="S" THEN 3390
3130 PRINT CHR$(7)
3140 K=K+1
3150 IF K<5 THEN 3220
3160 PRINT "YOU DIDN'T LISTEN, DID YOU?"
3170 PRINT "A DOZEN DEADLY PURPLE CAVE CREATURES"
3180 PRINT "PULL YOU DOWN INTO THE MUCK. THEY"
3190 PRINT "HAVEN'T HAD A MEAL SINCE THE LAST"
3200 PRINT "FOOLISH EXPLORER DROPPED IN. BYE-BYE!"
3210 END
3220 PRINT "YOU JUST WALKED INTO A CORNER AND"
3230 PRINT "BANGED YOUR HEAD ON A SLIMY ROCK."
3240 PRINT
3250 PRINT "*";
3260 FOR L=1 TO 34
3270 PRINT "=";
3280 NEXT L
3290 PRINT "*"
3300 PRINT ": THE ONLY DIRECTIONS YOU CAN GIVE : "
3310 PRINT ": ARE: ";Q$;"N";Q$;"", ";Q$;"E";Q$;"", ";Q$;"S";
3320 PRINT Q$;"", OR ";Q$;"W";Q$;TAB(36);": "
3330 PRINT "*";
3340 FOR L=1 TO 34
3350 PRINT "=";
3360 NEXT L
3370 PRINT "*"
3380 PRINT
3390 RETURN
3400 REM ----- BEEP SUBROUTINE -----
3410 FOR B=1 TO 5
3420 PRINT CHR$(7);
3430 FOR LAG=1 TO 600
3440 NEXT LAG
3450 NEXT B
3460 RETURN
3470 REM --- SLIME MONSTER SUBROUTINE ---
3480 IF SLIME=1 THEN 3610
3490 IF ROOM<7 THEN 3760
3500 MAGIC$=""
3510 FOR S=1 TO 7
3520 MAGIC$=MAGIC$+CHR$(INT(RND(1)*26)+65)
3530 NEXT S
3540 GOSUB 3780
3550 PRINT "YOU SEE A WORD WRITTEN IN BLOOD ON THE"
3560 PRINT "WALL: ";MAGIC$;" REMEMBER THIS WORD--"
3570 PRINT "IT MAY JUST SAVE YOUR LIFE!!!"
3580 GOSUB 3780
3590 SLIME=1
3600 GOTO 3760
3610 IF INT(RND(1)*7)<>1 THEN 3760

```

(continued)

```

3620 GOSUB 3780
3630 PRINT "THE HORRIBLE GREEN MUCK MONSTER IS"
3640 PRINT "IN THE ROOM WITH YOU. YOU HAVE ONE"
3650 PRINT "CHANCE TO USE THE MAGIC WORD TO SAVE"
3660 PRINT "YOUR LIFE--OTHERWISE YOU'LL BE DEAD"
3670 PRINT "IN TWO SECONDS! QUICK! WHAT IS IT";
3680 INPUT TRY$
3690 IF TRY$=MAGIC$ THEN 3730
3700 PRINT "NO!! YOU'RE WRONG! THE MAGIC WORD"
3710 PRINT "WAS ";MAGIC$;". AARRRRGGGHHHH!!!!"
3720 END
3730 PRINT "THIS TIME YOU REMEMBERED. LUCKY YOU!"
3740 SLIME=0
3750 GOSUB 3780
3760 RETURN
3770 REM --- LINE DRAWING SUBROUTINE ---
3780 PRINT
3790 N=(RND(1)*26)+64
3800 FOR E=1 TO 38
3810 PRINT CHR$(N);
3820 NEXT E
3830 PRINT
3840 PRINT
3850 RETURN
3860 REM --- CHECK FOR WIN SUBROUTINE ---
3870 IF F1<>1 OR F2<>1 THEN 3980
3880 PRINT
3890 PRINT "!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!"
3900 PRINT
3910 PRINT "YOU DID IT!!!! YOU FOUND THE TREASURE"
3920 PRINT "AND THE GIANT RUBY--AND YOU MADE IT"
3930 PRINT "BACK TO ROOM 1 TO GET OUT OF THE MAZE!"
3940 PRINT
3950 PRINT "!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!"
3960 GOSUB 3410
3970 END
3980 PRINT
3990 PRINT "$$$$$$$$$$$$ INVENTORY $$$$$$$$$$$$"
4000 PRINT
4010 IF F2=1 THEN PRINT TAB(9);"-- GIANT RUBY --"
4020 IF F1=1 THEN PRINT TAB(7);"-- FABULOUS TREASURE --"
4030 IF F1=0 AND F2=0 THEN PRINT TAB(5);"-- NOTHING BUT
YOUR WITS --"
4040 PRINT
4050 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$"
4060 PRINT
4070 RETURN

```

□ If You Have . . .

APPLE II

Delete: 730

Change: 200 HOME
440 PRINT "(HIT THE RETURN KEY FOR MORE)";
460 HOME
740 HOME

ATARI

Delete: 730

Add: 91 DIM X\$(1),Q\$(1),TRY\$(11),I\$(1),MAGIC\$(11)

Change: 120 PRINT " 1--HARD"
130 PRINT " 2--MEDIUM"
140 PRINT " 3--EASY"
200 PRINT CHR\$(125)
440 PRINT "(HIT THE RETURN KEY FOR MORE)";
460 PRINT CHR\$(125)
740 PRINT CHR\$(125)
780 TR=INT(RND(0)*15)+3
790 MG=INT(RND(0)*15)+3
2370 ON INT(RND(0)*5)+1 GOTO 2400,2430,2520,2550,2580
2380 ON INT(RND(0)*6)+1 GOTO 2400,2460,2490,2520,2550,
2580
2390 ON INT(RND(0)*7)+1 GOTO 2400,2430,2460,2490,
2520,2550,2580
2580 ON INT(RND(0)*4)+1 GOTO 2590,2650,2710,2770
2600 ON INT(RND(0)*6)+1 GOTO 2610,2610,2610,2610,
2610,2630
2660 ON INT(RND(0)*6)+1 GOTO 2670,2670,2670,2670,
2670,2690
2720 ON INT(RND(0)*6)+1 GOTO 2730,2730,2730,2730,
2730,2750
2780 ON INT(RND(0)*6)+1 GOTO 2790,2790,2790,2790,
2790,2810
2860 PRINT " -----> YOU ARE IN ROOM ";ROOM;"<-----"
3130 PRINT CHR\$(253)
3320 PRINT Q\$;"", OR " ";Q\$;"W";Q\$;" :"
3340 FOR L=1 TO 32
3420 PRINT CHR\$(253)
3520 MAGIC\$(S)=CHR\$(INT(RND(0)*26)+65)
3610 IF INT(RND(0)*7)<>1 THEN 3760
3790 N=(RND(0)*26)+64
3800 FOR E=1 TO 36
4010 IF F2=1 THEN PRINT " ";"-- GIANT RUBY --"
4020 IF F1=1 THEN PRINT " ";"-- FABULOUS TREASURE --"

(continued)

```
4030 IF F1=0 AND F2=0 THEN PRINT " "; "-- NOTHING  
      BUT YOUR WITS --"
```

COMMODORE 64

Delete: 730

```
Change: 200 PRINT CHR$(147)  
440 PRINT "(HIT THE RETURN KEY FOR MORE)";  
460 PRINT CHR$(147)  
740 PRINT CHR$(147)  
780 TR=INT(RND(0)*15)+3  
790 MG=INT(RND(0)*15)+3  
2370 ON INT(RND(0)*5)+1 GOTO 2400,2430,2520,2550,  
      2580  
2380 ON INT(RND(0)*6)+1 GOTO 2400,2460,2490,2520,  
      2550,2580  
2390 ON INT(RND(0)*7)+1 GOTO 2400,2430,2460,2490,  
      2520,2550,2580  
2580 ON INT(RND(0)*4)+1 GOTO 2590,2650,2710,2770  
2600 ON INT(RND(0)*6)+1 GOTO 2610,2610,2610,2610,  
      2610,2630  
2660 ON INT(RND(0)*6)+1 GOTO 2670,2670,2670,2670,  
      2670,2690  
2720 ON INT(RND(0)*6)+1 GOTO 2730,2730,2730,2730,  
      2730,2750  
2780 ON INT(RND(0)*6)+1 GOTO 2790,2790,2790,2790,  
      2790,2810  
3320 PRINT Q$;"", OR ";Q$;"W";Q$;"      :"
```

COMMODORE VIC-20

Program will not run on unexpanded VIC-20

TEXAS INSTRUMENTS 99/4A

```
Add: 222 PRINT "PARING FOR A GREAT EXPEDI-"  
262 PRINT "HAS LED TO A DUSTY OLD BOOK"  
372 PRINT "THERE WAS"  
672 PRINT "GOOD LUCK!"  
872 GOSUB 2360  
882 GOSUB 2360  
972 GOSUB 2360  
1032 GOSUB 2360  
1052 GOSUB 2360  
1132 GOSUB 2360  
1142 GOSUB 2360  
1152 GOSUB 2360  
1232 GOSUB 2360
```

```

1242 GOSUB 2360
1302 GOSUB 2360
1412 GOSUB 2360
1422 GOSUB 2360
1502 GOSUB 2360
1482 GOSUB 2360
1572 GOSUB 2360
1582 GOSUB 2360
1672 GOSUB 2360
1682 GOSUB 2360
1752 GOSUB 2360
1782 GOSUB 2360
1852 GOSUB 2360
1932 GOSUB 2360
1942 GOSUB 2360
2022 GOSUB 2360
2032 GOSUB 2360
2052 GOSUB 2360
2142 GOSUB 2360
2212 GOSUB 2360
2222 GOSUB 2360
2292 GOSUB 2360
2322 GOSUB 2360
4012 PRINT TAB(5);"-- GIANT RUBY --"
4022 PRINT TAB(3);"-- FABULOUS TREASURE --"
4034 GOTO 4040
4036 PRINT TAB(1);"-- NOTHING BUT YOUR WITS --"

```

```

Change: 200 CALL CLEAR
210 PRINT "YOU ARE A FAMOUS EXPLORER"
220 PRINT "WHO HAS SPENT YEARS PRE-"
230 PRINT "TION TO THE ISLAND OF PHIC,"
240 PRINT "AN UNCHARTED DOT OF LAND"
250 PRINT "THOUSANDS OF MILES FROM"
260 PRINT "CIVILIZATION. YOUR RESEARCH"
270 PRINT "WITH A SECRET MAP SEWN"
280 PRINT "INSIDE THE BINDING. YOU'VE FOLLOWED THE"
290 PRINT "MAP AND HAVE TRAVELED"
300 PRINT "THROUGH THE THICK JUNGLE OF PIHC. NOW YOU"
340 PRINT "YOUR NATIVE GUIDES REFUSED"
350 PRINT "TO GO THIS FAR, BECAUSE THEY HAD HEARD THE CAVE"
360 PRINT "CONTAINED TERRIBLE SCARY"
370 PRINT "MONSTERS. THEY SAID "
400 PRINT "THAT WAS TO FIND THE GIANT"
410 PRINT "RUBY IDOL'S EYE HIDDEN IN THE"
420 PRINT "DARKEST CHAMBER OF THE CAVE"
460 CALL CLEAR
480 PRINT "GIANT RUBY WAS JUST PART OF"
490 PRINT "A VAST TREASURE. SEVERAL"
500 PRINT "OTHERS HAD FOUND THE CAVE, AND ONE HAD"
520 PRINT "THE MONSTERS DID NOT LET"
530 PRINT "THEM ESCAPE. THE EXPLORERS WERE NOT"
570 PRINT "ONE MORE THING: SOMEWHERE" (continued)

```



```

580 PRINT "IN THE MAZE OF TUNNELS IS A"
590 PRINT "MAGIC WORD WRITTEN BY AN OLD VICTIM."
620 PRINT "FROM TIME TO TIME, SO BE"
630 PRINT "CAREFUL! THE MONSTER THAT PROWLs THE
    CAVE WILL SMOTHER"
640 PRINT "YOU IN SLIME UNLESS YOU RE-"
650 PRINT "MEMBER IT!! FINALLY, AT THE"
660 PRINT "END, YOU MUST GET TO ROOM 1"
670 PRINT "TO LEAVE THE CAVE. "
730 RANDOMIZE
740 CALL CLEAR
780 TR=INT(RND*15)+3
790 MG=INT(RND*15)+3
870 IF I$<>"N" THEN 880
880 IF I$<>"S" THEN 890
970 IF I$<>"S" THEN 980
1030 IF I$<>"E" THEN 1040
1050 IF I$<>"N" THEN 1060
1130 IF I$<>"W" THEN 1040
1140 IF I$<>"N" THEN 1050
1150 IF I$<>"S" THEN 1060
1230 IF I$<>"N" THEN 1240
1240 IF I$<>"S" THEN 1250
1300 IF I$<>"E" THEN 1310
1410 IF I$<>"N" THEN 1420
1420 IF I$<>"S" THEN 1430
1480 IF I$<>"E" THEN 1490
1500 IF I$<>"N" THEN 1510
1570 IF I$<>"E" THEN 1580
1580 IF I$<>"W" THEN 1590
1670 IF I$<>"W" THEN 1680
1680 IF I$<>"N" THEN 1690
1750 IF I$<>"E" THEN 1760
1780 IF I$<>"S" THEN 1790
1850 IF I$<>"W" THEN 1850
1930 IF I$<>"E" THEN 1940
1940 IF I$<>"W" THEN 1950
2020 IF I$<>"E" THEN 2030
2030 IF I$<>"W" THEN 2040
2050 IF I$<>"S" THEN 2060
2140 IF I$<>"S" THEN 2150
2210 IF I$<>"W" THEN 2220
2220 IF I$<>"N" THEN 2230
2290 IF I$<>"E" THEN 2300
2320 IF I$<>"S" THEN 2330
2370 ON INT(RND*5)+1 GOTO 2400,2430,2520,2550,2580
2380 ON INT(RND*6)+1 GOTO 2400,2460,2490,2520,2550,
    2580
2390 ON INT(RND*7)+1 GOTO 2400,2430,2460,2490,2520,
    2550,2580
2430 PRINT "THIS WHOLE WALL IS OOZING"
2440 PRINT "ACID. YOU CAN'T GO THIS WAY."
2460 PRINT "THIS DOOR HAS SUDDENLY"
2470 PRINT "CLOSED. YOU CAN'T GO THIS WAY."
2520 PRINT "AHEAD ARE SHEETS OF UNPASSABLE FLAME."
2580 ON INT(RND*4)+1 GOTO 2590,2650,2710,2770

```

```

2600 ON INT(RND*6)+1 GOTO 2610,2610,2610,2610,2610,
2630
2660 ON INT(RND*6)+1 GOTO 2670,2670,2670,2670,2670,
2690
2720 ON INT(RND*6)+1 GOTO 2730,2730,2730,2730,2730,
2750
2730 PRINT "BUT YOU FREEZE AND THEY IG- NORE YOU."
2780 ON INT(RND*6)+1 GOTO 2790,2790,2790,2790,2790,
2810
2860 PRINT "--> YOU ARE IN ROOM ";ROOM;"<--"
2890 PRINT "** YOU'VE FOUND ALL THE TREASURE!!!**"
3060 PRINT "***** YOU DID IT!!! *****"
3070 PRINT "YOU FOUND THE TREASURE--- AND ALSO
FOUND"
3080 PRINT "THE GIANT RUBY TO LET YOU ESCAPE !!!!!"
3120 IF (I$="N") + (I$="E") + (I$="W") + (I$="S")
THEN 3390
3130 CALL SOUND(100,440,2)
3170 PRINT "A DOZEN DEADLY PURPLE CAVE CREATURES
PULL YOU DOWN"
3180 PRINT "INTO THE MUCK. THEY HAVEN'T"
3190 PRINT "HAD A MEAL SINCE THE LAST"
3220 PRINT "YOU JUST WALKED INTO A"
3230 PRINT "CORNER AND BANGED YOUR HEAD ON A SLIMY
ROCK."
3260 FOR L=1 TO 24
3320 PRINT Q$;"", OR ";Q$;"W";Q$;TAB(26)
3340 FOR L=1 TO 24
3420 CALL SOUND(100,400,2)
3520 MAGIC$=MAGIC$+CHR$(INT(RND*26)+65)
3550 PRINT "YOU SEE A WORD WRITTEN IN BLOOD ON
THE WALL:"
3560 PRINT MAGIC$;" REMEMBER THIS WORD--"
3610 IF INT(RND*7)<>1 THEN 3760
3630 PRINT "THE HORRIBLE GREEN MON-STER IS IN THE
ROOM WITH YOU."
3640 PRINT "YOU HAVE ONE CHANCE TO USE"
3650 PRINT "THE MAGIC WORD TO SAVE YOUR"
3660 PRINT "LIFE--OTHERWISE YOU'LL BE DEAD IN TWO
SECONDS! QUICK!"
3670 PRINT "WHAT IS IT"
3700 PRINT "NO!! YOU'RE WRONG! THE MAGIC WORD WAS";
MAGIC$
3710 PRINT "AARRRRGGGHHHH!!!!"
3790 N=(RND*26)+64
3800 FOR E=1 TO 26
3870 IF (F1<>1) + (F2<>1) THEN 3980
3890 PRINT "!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!"
3910 PRINT "YOU DID IT!!!! YOU FOUND THE TREASURE
AND THE GIANT"
3920 PRINT "RUBY--AND YOU MADE IT BACK"
3930 PRINT "TO ROOM 1 TO GET OUT OF THE MAZE!"
3950 PRINT "!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!"
3990 PRINT "$$$$$$$ INVENTORY $$$$$$$$" (continued)

```

```

4010 IF F2<>1 THEN 4020
4020 IF F1<>1 THEN 4030
4030 IF (F1=0) * (F2=0) THEN 4036
4050 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$$"

```

TRS-80 COLOR COMPUTER

Delete: 330,430,560,680,730,760,810,910,1000,1090,1180,1270,
1360,1450,1540,1550,1560,1570,1580,1590,1600,1610,
1620,1630,1720,1810,1900,1990,2080,2170,2260,2350,
2830,3110,3240,3330,3340,3350,3360,3370,3400,3470,
3540,3580,3620,3750,3770,3780,3790,3800,3810,3820,
3830,3840,3850,3860,3880,3940,3950,3980,4050,4060

Add: 325 PRINT "(HIT ENTER KEY FOR MORE)";
326 INPUT X\$
555 PRINT "(HIT ENTER KEY FOR MORE)";
556 INPUT X\$

Change: 200 CLS
210 PRINT "YOU ARE A FAMOUS EXPLORER WHO HAS ";
220 PRINT "SPENT YEARS PREPARING FOR A GREAT ";
230 PRINT "EXPEDITION TO THE ISLAND OF PIHC, ";
240 PRINT "AN UNCHARTED DOT OF LAND THOUSANDS ";
250 PRINT "OF MILES FROM CIVILIZATION. YOUR ";
260 PRINT "RESEARCH HAS LED TO A DUSTY OLD ";
270 PRINT "BOOK WITH A SECRET MAP SEWN INSIDE ";
280 PRINT "THE BINDING. YOU'VE FOLLOWED THE ";
290 PRINT "MAP AND HAVE TRAVELED THROUGH THE ";
300 PRINT "THICK JUNGLE OF PIHC. NOW YOU ";
310 PRINT "ARE DEEP INSIDE THE MAZE OF ";
460 CLS
740 CLS
780 TR=RND(15)+2
790 MG=RND(15)+2
2370 ON RND(5) GOTO 2400,2430,2520,2550,2580
2380 ON RND(6) GOTO 2400,2460,2490,2520,2550,2580
2390 ON RND(7) GOTO 2400,2430,2460,2490,2520,2550,
2580
2580 ON RND(4) GOTO 2590,2650,2710,2770
2600 ON RND(6) GOTO 2610,2610,2610,2610,2610,2630
2660 ON RND(6) GOTO 2670,2670,2670,2670,2670,2690
2720 ON RND(6) GOTO 2730,2730,2730,2730,2730,2750
2780 ON RND(6) GOTO 2790,2790,2790,2790,2790,2810
2860 PRINT "YOU ARE IN ROOM ";ROOM
3060 PRINT "YOU DID IT!!"
3130 SOUND 173,3
3260 FOR L=1 TO 30
3300 PRINT ": DIRECTIONS YOU CAN GIVE:"
3320 PRINT Q\$;"", OR ";Q\$;"W";Q\$;TAB(30);": "
3340 FOR L=1 TO 30
3420 SOUND 173,3
3520 MAGIC\$=MAGIC\$ + CHR\$(RND(26)+64)
3610 IF RND(7) <> 1 THEN 3760

```
3870 IF F1<>1 OR F2<>1 THEN 3990
3990 PRINT "INVENTORY:"
4010 IF F2=1 THEN PRINT TAB(5);"-- GIANT RUBY --"
4020 IF F1=1 THEN PRINT TAB(3);"-- FABULOUS
TREASURE --"
4030 IF F1=0 AND F2=0 THEN PRINT TAB(1);"--
NOTHING BUT YOUR WITS --"
```

MORE FUN! LESS TYPING!

If you just finished typing in all the programs in this book and your pet gerbil ate the disk or cassette you saved them on...

If you're already enjoying the shorter programs in this book, but your fingers tremble at the thought of typing in the longer ones...

If your secretary does all your typing, but she's on a year-long vacation in Moose Jaw, Saskatchewan...

Or if you'd just like to have all these programs handy on a disk or cassette that'll run on your machine...

You're in luck!

You can get every single program in this book (along with extra bonus programs not included in this book) on ready-to-run cassettes or disks, for only \$19.95 per book. You can also get tapes or disks for any of the *other* exciting books in this series.

(As a special bonus, order any three, and we'll throw in the fourth one *absolutely free*—so you can enjoy the entire set—and we'll pay all postage and handling!)

Programs are available in one format only (disk *or* cassette) for each specific computer, so please make sure you have the proper equipment before ordering. Remember to check off BOTH the set of programs you want AND the type of machine you have.

Hard/Soft Inc., PO Box 1277, Riverdale, NY 10471

Yes! Please send me the ready-to-run programs I've checked off below.

1. Check off which format you want:

- ATARI cassette tapes
- APPLE II/II PLUS/IIe disks
- COMMODORE 64 disks
- COMMODORE VIC-20 disks
- IBM PC/PCjr disks
- TEXAS INSTRUMENTS 99/4A cassette tapes
- TRS-80 COLOR COMPUTER cassette tapes

2. Check off which programs you are ordering:

- Computer Crazyness (\$19.95)
- Computer Monsters (\$19.95)
- Computer Olympics (\$19.95)
- Computer Space Adventures (\$19.95)
- ALL FOUR BOOKS (SPECIAL PRICE \$59.85)

**3. PRINT your name and address, and fill in the amount you are enclosing:
(Add applicable sales taxes!)**

I am enclosing \$19.95 PLUS \$2.00 postage and handling and applicable sales tax for each. Total: \$_____

SPECIAL BONUS: Check this box, enclose \$59.85, and we'll send you all FOUR sets of programs for the price of three—and we'll pay all postage! This is a savings of over \$20.00.

PRINT your name _____

PRINT your address _____

City _____ State _____ Zip _____

Make checks payable to Hard/Soft, Inc. Allow 2-8 weeks for delivery.



MONSTERS ARE INVADING YOUR COMPUTER!

Has your keyboard been attacked by monsters? Are there bats in your computer's belfry? Run the programs in **COMPUTER MONSTERS** and you'll meet a host of vampires, werewolves, mummies, and ogres – even brand-new monsters you create yourself!

COMPUTER MONSTERS is an exciting collection of games, puzzles, and tricks that turn your computer into a den of wild creatures. You'll slay dragons, match wits with trolls, and hunt for the Abominable Snowman. You may even find the mysterious monster of Loch Ness!

You'll have these monstrously wonderful programs running in minutes, and they'll keep you horrified for hours!

Other books by these authors:

**COMPUTER CRAZINESS
COMPUTER OLYMPICS
COMPUTER SPACE ADVENTURES**



ISBN 0-590-33177-9


SCHOLASTIC INC.